
1. はじめに 

本例題集では，様々な数理計画問題を紹介し，それらに対する SIMPLE での定式化の例を与

えます．扱う問題のモデルファイル等は，プロジェクトファイルとしてまとめてあります．

Visual Analytics Platform を起動し，メニューから［プロジェクト］→［インポート］

を選択し，選択画面を開きます． 

 

選択画面から以下のファイルを選びます． 

 ｛NUOPT のインストール場所｝¥SAMPLES¥例題集.prj 

なお，デフォルトの NUOPTのインストール場所は 32bit 版 Windows であれば 

 C:¥Program Files¥Mathematical Systems Inc¥NUOPT 

64bit 版 Windows であれば 

 C:¥Program Files (x86)¥Mathematical Systems Inc¥NUOPT 

となっています．また，例題集で扱う Excel ファイル等も同様に以下の場所にあります． 

  ｛NUOPT のインストール場所｝¥SAMPLES  

Visual Analytics Platform の起動方法等については「NUOPT/SIMPLEチュートリアル」

をご参考ください． 

扱う問題の構成は以下の通りです．表における ○ は，それぞれの問題が，どのような種類の

数理計画問題に属するかを表しています．例えば，ナップサック問題は混合線形整数計画問題で

す． 

LP は線形計画問題， MIP(MILP) は混合線形整数計画問題， QP は二次計画問題， NLPは

非線形計画問題，SDPは半正定値計画問題，WCSP は重み付き制約充足問題， RCPSP は資源制

約付きスケジューリング問題を意味します． 

問題 LP MIP QP NLP SDP WCSP RCPSP 

配合問題 ○       

輸送問題 ○       

多期間計画問題 ○       

包絡分析法（DEA）モデル ○       

ナップサック問題  ○      

集合被覆問題  ○      

最大流問題 ○       

最小費用流問題 ○       

多品種流問題 ○       

p メディアン問題  ○      

p センター問題  ○      

巡回セールスマン問題  ○      

割当問題  ○    ○  

二次割当問題      ○  

設備計画問題      ○  

最小二乗問題   ○     

ポートフォリオ最適化問題   ○     

ロジスティック回帰モデル    ○    

イールドカーブ推定問題    ○    

格付け推移行列推定問題    ○    

相関行列取得問題     ○   

ロバストポートフォリオ最適化問題     ○   

隣接行列（最大カット問題）  ○   ○ ○  

セミナー割当問題       ○ 

ジョブショップスケジューリング問題       ○ 

遷移確率行列 ○       

ムーア・ペンローズ一般逆行列 ○       

 

なお，数理計画問題の種類と NUOPT の製品ラインアップは以下のように対応します． 

モジュール名 LP MIP QP NLP SDP WCSP RCPSP 

LP/IP Liteモジュール ○ ○ × × × × × 

LP/IP モジュール ○ ○ × × × ○ ○ 

NLP モジュール ○ ○ ○ ○ ○ × × 

フルセット ○ ○ ○ ○ ○ ○ ○ 



 6 

 

2. 各種例題の紹介および SIMPLEでの記述例 

本章では， 1 章で紹介した各種例題に対する問題の説明および，SIMPLE での記述例を紹介

します． 

 

2.1 配合問題（SAMPLES¥例題集 1.prj 参照） 

配合問題の例として，ここでは特定の組成を持つ合金を生成する問題を扱います．この他にも

薬剤の調合や必要な栄養素を含む献立を考えるダイエット問題など配合問題として扱えるもの

は多岐にわたります． 

 

この問題を NUOPTで解くために定式化を行います．本例題は文献[1]からの引用です． 

まず，変数として市販の合金 1,2,3,…,9の混合比率，つまり混ぜ合わせる割合を，それぞれ

1 2 3 9, , , ,x x x x としましょう． 

次に，最小化するべき目的関数は，各市販の合金について「単位量当たりのコスト」と「混合

比率」の積の総和として表現することができます． 

最後に制約条件です．まず，混合比率は負の値をとれませんので，各変数に対して非負制約が

必要です．混合比率の総和は 1 ですので，その制約も加えます．また，生成する合金の組成に

ついての制約は，鉛，亜鉛，スズに対して，各市販の合金についての「構成比率」と「混合比率」

の積の総和が，それぞれ 30%，30%，40%と等しい，という形になります． 

 

（例題） 

鉛，亜鉛，スズの構成比率が，それぞれ 30%，30%，40%となるような合金を，市販の

合金を混ぜ合わせ，できるだけ安いコストで生成することを考えます．現在手に入れる

ことができる市販の合金は９種類で，それらの構成比率と単位量あたりのコストは以下

の通りです．  

 

市販の合金 1 2 3 4 5 6 7 8 9 

鉛(%) 20 50 30 30 30 60 40 10 10 

亜鉛(%) 30 40 20 40 30 30 50 30 10 

スズ(%) 50 10 50 30 40 10 10 60 80 

コスト($/lb) 7.3 6.9 7.3 7.5 7.6 6.0 5.8 4.3 4.1 

 

所望の組成を持つ合金をコストを一番安く生成するには，市販の合金をどのように混ぜ

合わせれば良いでしょうか． 

以上のことから，次のように定式化することができます． 

 

   

変数 
1x  市販の合金１の混合比率 

 2x  市販の合金２の混合比率 

 3x  市販の合金３の混合比率 

 
4x  市販の合金４の混合比率 

 5x  市販の合金５の混合比率 

 6x  市販の合金６の混合比率 

 7x  市販の合金７の混合比率 

 8x  市販の合金８の混合比率 

 9x  市販の合金９の混合比率 

   

目的関数 (最小化)  総コスト 

1 2 3 4 5 6 7 8 97.3 6.9 7.3 7.5 7.6 6.0 5.8 4.3 4.1x x x x x x x x x         

   

非負制約 1 0x    

 2 0x    

 3 0x    

 4 0x    

 5 0x    

 6 0x    

 7 0x    

 8 0x    

 9 0x    

   

混合比率の制約   

1 2 3 4 5 6 7 8 9 1x x x x x x x x x          

1 2 3 4 5 6 7 8 90.2 0.5 0.3 0.3 0.3 0.6 0.4 0.1 0.1 0.3x x x x x x x x x          

1 2 3 4 5 6 7 8 90.3 0.4 0.2 0.4 0.3 0.3 0.5 0.3 0.1 0.3x x x x x x x x x          

1 2 3 4 5 6 7 8 90.5 0.1 0.5 0.3 0.4 0.1 0.1 0.6 0.8 0.4x x x x x x x x x          

   

 

この問題は，目的関数，制約式全て線形なので，線形計画問題となります． 

定式化した結果を SIMPLEで記述すると以下のようになります． 

 

 

mixture1.smp 

// 変数 

Variable x1(name="市販の合金 1の混合比率"); 

Variable x2(name="市販の合金 2の混合比率"); 

Variable x3(name="市販の合金 3の混合比率"); 

Variable x4(name="市販の合金 4の混合比率"); 

Variable x5(name="市販の合金 5の混合比率"); 

Variable x6(name="市販の合金 6の混合比率"); 

Variable x7(name="市販の合金 7の混合比率"); 

Variable x8(name="市販の合金 8の混合比率"); 

Variable x9(name="市販の合金 9の混合比率"); 

// 目的関数 

Objective z(name="総コスト", type=minimize); 

z=7.3*x1+6.9*x2+7.3*x3+7.5*x4+7.6*x5+6.0*x6+5.8*x7+4.3*x8+4.1*x9; 

// 非負制約 

x1 >= 0; 

x2 >= 0; 

x3 >= 0; 

x4 >= 0; 

x5 >= 0; 

x6 >= 0; 

x7 >= 0; 

x8 >= 0; 

x9 >= 0; 

// 混合比率の制約 

x1+x2+x3+x4+x5+x6+x7+x8+x9 == 1; 

0.2*x1+0.5*x2+0.3*x3+0.3*x4+0.3*x5+0.6*x6+0.4*x7+0.1*x8+0.1*x9==0

.3; 

0.3*x1+0.4*x2+0.2*x3+0.4*x4+0.3*x5+0.3*x6+0.5*x7+0.3*x8+0.1*x9==0

.3; 

0.5*x1+0.1*x2+0.5*x3+0.3*x4+0.4*x5+0.1*x6+0.1*x7+0.6*x8+0.8*x9==0

.4; 

// 求解 

solve(); 

// 出力 

z.val.print(); 

より汎用的に問題を定式化すると以下のようになります． 

 

   

集合 {1,2,3,4,5,6,7,8,9}Alloy   市販の合金の種類の集合 

 { , , }Blend Lead Zinc Tin  構成金属の集合 

   

定数 , ,ijr i Alloy j Blend   
市販の合金 i における構成金属

j の比率 

 ,ic i Alloy  
市販の合金 i の単位量あたりの

コスト 

 ,jb j Blend  構成金属 j の目標比率 

   

変数 ,ix i Alloy  市販の合金 i の混合比率 

   

目的関数 (最小化) i i

i Alloy

c x


  総コスト 

   

制約 0,ix i Alloy    非負制約 

 1i

i Alloy

x


  比率の総和が 1という制約 

 ,ij i j

i Alloy

r x b j Blend


    混合比の制約 

次に，定数（構成比率，コスト，目標比率）をデータファイルから与える SIMPLE モデルを

示します．このようにモデルとデータを分離することにより，市販の合金の数や構成金属の種類

の数が変わったとしてもデータファイルを変更するだけで対応できるようになります． 

 

 

mixture2.smp 

// 集合と添字 

Set Alloy(name="市販の合金集合"); 

Element i(set=Alloy); 

Set Blend(name="構成金属集合"); 

Element j(set=Blend); 

 

// パラメータ 

Parameter r(name="構成比率", index=(i,j)); 

Parameter c(name="コスト", index=i); 

Parameter b(name="目標比率", index=j); 

 

// 変数 

Variable x(name="混合比率", index=i); 

 

// 目的関数 

Objective z(name="総コスト", type=minimize); 

z = sum(c[i]*x[i],i); 

 

// 非負制約 

x[i] >= 0; 

 

// 混合比の制約 

sum(x[i],i) == 1; 

sum(r[i,j]*x[i],i) == b[j]; 

 

// 求解 

solve(); 

 

// 出力 

z.val.print(); 

x.val.print(); 

データファイル（.dat形式）は以下のようになります． 

 

 

 

このモデルを実行すると，市販の合金 6 を 40%，市販の合金 8を 60%混ぜ合わせるのが最適

で，そのときの総コストは 4.98であることがわかります． 

data2.dat 

構成比率 = 

[1,Lead] 0.2 [1,Zinc] 0.3 [1, Tin] 0.5 

[2,Lead] 0.5 [2,Zinc] 0.4 [2, Tin] 0.1 

[3,Lead] 0.3 [3,Zinc] 0.2 [3, Tin] 0.5 

[4,Lead] 0.3 [4,Zinc] 0.4 [4, Tin] 0.3 

[5,Lead] 0.3 [5,Zinc] 0.3 [5, Tin] 0.4 

[6,Lead] 0.6 [6,Zinc] 0.3 [6, Tin] 0.1 

[7,Lead] 0.4 [7,Zinc] 0.5 [7, Tin] 0.1 

[8,Lead] 0.1 [8,Zinc] 0.3 [8, Tin] 0.6 

[9,Lead] 0.1 [9,Zinc] 0.1 [9, Tin] 0.8 

; 

 

コスト = 

[1] 7.3 

[2] 6.9 

[3] 7.3 

[4] 7.5 

[5] 7.6 

[6] 6.0 

[7] 5.8 

[8] 4.3 

[9] 4.1 

; 

 

目標比率 = 

[Lead] 0.3 

[Zinc] 0.3 

[ Tin] 0.4 

; 

2.2 輸送問題（SAMPLES¥例題集 1.prj 参照） 

輸送問題は複数の供給地から複数の需要地への物の流れ方を決める問題ということができま

す．供給地と需要地をノード，物の流れる経路をアークとすれば，輸送問題はネットワークによ

って表現できる問題の一種と捉えることができます．輸送問題に対する定式化の方法は他のネッ

トワークによって表現できる問題にも応用できます． 

 

 

この問題を NUOPTで解くために定式化を行います．本例題は文献[1]からの引用です． 

まず，変数として各工場から各店舗への輸送量を以下の図のように設定します． 

 

 

 

 

 

 

 

 

 

（例題） 

ある配送業者は二つの工場 1，2から三つの店舗 a，b，cへの製品の輸送を請け負って

いるとします．各工場，店舗について，それぞれ供給可能量と需要量が決められており，

それらを満たしつつ，最もコストがかからない製品の運び方をどのように決定すればよ

いでしょうか．各工場の供給可能量，各店舗の需要量，単位量あたりの輸送コストは以

下の通りです． 

 

工場 供給可能量 店舗 需要量 

1 250 a 200 

2 450 b 200 

  c 200 

 

単位量あたりの輸送コスト 

 a b c 

1 3.4 2.2 2.9 

2 3.4 2.4 2.5 

 

1 

2 

a 

b 

c 

x1a 

x1b 

x1c 

x2a 

 
x2b 

x2c 

1 

2 

a 

b 

c 

x1a 

x1b 

x1c 

x2a 

 
x2b 

x2c 

次に，目的関数は，工場から店舗への各経路について「単位量あたりの輸送コスト」と「輸送

量」の積の総和として表現することができます． 

最後に制約条件です．輸送量は負にはなり得ないので，各変数に対して非負制約が必要です．

さらに，各工場について工場からの輸送量の和が供給可能量以下である，各店舗について店舗へ

の輸送量の和が需要量と等しい，という制約が加わります． 

 

以上のことから，次のように定式化することができます． 

 

   

変数 1ax  工場 1から店舗 a への輸送量 

 1bx  工場 1から店舗 b への輸送量 

 1cx  工場 1から店舗 c への輸送量 

 2ax  工場 2から店舗 a への輸送量 

 2bx  工場 2から店舗 b への輸送量 

 2cx  工場 2から店舗 c への輸送量 

   

目的関数 (最小化)  総コスト 

1 1 1 2 2 23.4 2.2 2.9 3.4 2.4 2.5a b c a b cx x x x x x      

   

非負制約 1 0ax    

 1 0bx    

 1 0cx    

 2 0ax    

 2 0bx    

 2 0cx    

   

工場の生産量の制約 
1 1 1 250a b cx x x    工場 1について 

 
2 2 2 450a b cx x x    工場 2について 

   

店舗の需要量の制約 
1 2 200a ax x   店舗 aについて 

 
1 2 200b bx x   店舗 bについて 

 
1 2 200c cx x   店舗 cについて 

   

 

ネットワークで表現できる問題の多くは，上で見てきたように各アークに対して変数を定義し，

各ノードについて制約をたてることによって定式化されます． 

定式化した結果を SIMPLEで記述すると以下のようになります． 

 

 

transport1.smp 

// 変数 

Variable x1a(name="工場 1から店舗 aへの輸送量"); 

Variable x1b(name="工場 1から店舗 bへの輸送量"); 

Variable x1c(name="工場 1から店舗 cへの輸送量"); 

Variable x2a(name="工場 2から店舗 aへの輸送量"); 

Variable x2b(name="工場 2から店舗 bへの輸送量"); 

Variable x2c(name="工場 2から店舗 cへの輸送量"); 

 

// 非負制約 

x1a >= 0; 

x1b >= 0; 

x1c >= 0; 

x2a >= 0; 

x2b >= 0; 

x2c >= 0; 

 

// 工場の生産量の制約 

x1a + x1b + x1c <= 250; 

x2a + x2b + x2c <= 450; 

 

// 店舗の需要量の制約 

x1a + x2a == 200; 

x1b + x2b == 200; 

x1c + x2c == 200; 

 

// 目的関数 

Objective z(name="総コスト", type=minimize); 

z = 3.4*x1a + 2.2*x1b + 2.9*x1c + 3.4*x2a + 2.4*x2b + 2.5*x2c; 

 

// 求解 

solve(); 

 

// 出力 

z.val.print(); 

より汎用的に問題を定式化すると以下のようになります． 

 

   

集合 {1,2}Cannery   工場 

 { , , }Warehouse a b c  店舗 

   

定数 ,iupper i Cannery  工場 i の供給可能量 

 ,jdemand j Warehouse  店舗 j の需要量 

 , ,ijc i Cannery j Warehouse   
工場 i から店舗 j  への単位量あ

たりの輸送コスト 

   

変数 , ,ijx i Cannery j Warehouse   工場 i から店舗 j  への輸送量 

   

目的関数 

(最小化) 

ij ij

i Cannery j Warehouse

c x
 

   総コスト 

   

制約 

0

, ,

ijx

i Cannery j Warehouse



     
非負制約 

 ,ij i

j Warehouse

x upper i Cannery


    
工場の生産量の制約 

 

,ij j

i Cannery

x demand j Warehouse


  

 

店舗の需要量の制約 

次に，入力データとモデルを分離した SIMPLEモデルを示します．  

 

 

transport2.smp 

// 集合と添字 

Set Cannery(name="工場"); 

Element i(set=Cannery); 

Set Warehouse(name="店舗"); 

Element j(set=Warehouse); 

 

// パラメータ 

Parameter upper(name="供給可能量", index=i); 

Parameter demand(name="需要量", index=j); 

Parameter cost(name="輸送コスト", index=(i,j)); 

 

// 変数 

Variable x(name="輸送量", index=(i,j)); 

 

// 目的関数 

Objective z(name="総コスト", type=minimize); 

z = sum(cost[i,j]*x[i,j],(i,j)); 

 

// 非負制約 

x[i,j] >= 0; 

 

// 工場の生産量の制約 

sum(x[i,j],j) <= upper[i]; 

 

// 店舗の需要量の制約 

sum(x[i,j],i) == demand[j]; 

 

// 求解 

solve(); 

 

// 出力 

z.val.print(); 

x.val.print(); 

データファイル（.dat形式）は以下のようになります． 

 

 

 

このモデルを実行すると，以下のような結果が得られます．  

 

 

data2.dat 

供給可能量 = 

[1] 250 

[2] 450 

; 

 

需要量 = 

[a] 200 

[b] 200 

[c] 200 

; 

 

輸送コスト = 

[1,a] 3.4 

[1,b] 2.2 

[1,c] 2.9 

[2,a] 3.4 

[2,b] 2.4 

[2,c] 2.5 

; 

総コスト=1620 

輸送量[1,"a"]=26.4602 

輸送量[1,"b"]=200 

輸送量[1,"c"]=1.79049e-008 

輸送量[2,"a"]=173.54 

輸送量[2,"b"]=3.58099e-008 

輸送量[2,"c"]=200 

2.3 多期間計画問題（SAMPLES¥例題集 1.prj 参照） 

多期間計画問題とは，多期間にわたり期間ごとに意思決定をする問題のことをいいます．多期

間計画問題を定式化する場合は，期間ごとに変数を定義するのが一般的です． 

 

 

この問題を NUOPTで解くために定式化を行います．本例題は文献[2]からの引用です． 

まず，変数として各月における製品 I，II の生産量をそれぞれ I1 II1 I2 II2 I3 II3, , , , ,x x x x x x とし，

在庫量をそれぞれ I1 II1 I2 II2, , ,y y y y としましょう． 

次に，最小化するべき目的関数は，各生産量／在庫量とその単位量あたりのコストとの積の総

和として表現することができます． 

最後に制約条件です．まず，各変数に対して非負制約が必要です．次に，1ヶ月に利用できる

原料が決められているので，各月ごとに各原料に関する制約が必要です．さらに，各製品に対し

て在庫量は次の月に持ち越せますので，それを踏まえた出荷量の制約を加えます． 

以上のことから，次のように定式化することができます． 

（例題） 

2種類の原料 A，Bを加工して 2種類の製品 I,IIを生産している工場が，向こう３ヶ月

間の生産計画を立てようとしています．各製品を 1単位生産するために必要な原料の使

用量，各製品の生産／在庫コスト，各製品の月ごとの出荷量，各原料の月ごとの利用可

能量は以下のように与えられています． 

 

原料使用量 生産／在庫コスト 

 I II  I II 

A 2 7 生産 75 50 

B 5 3 在庫 8 7 

 

製品の出荷量 原料の利用可能量 

 I II  A B 

1 30 20 1 920 790 

2 60 50 2 750 600 

3 80 90 3 500 480 

 

各月に出荷する製品をその月中に全て生産できるとは限らないので，前の月に生産した

製品を在庫として保管して来月に出荷することも考えられます．このような状況の下で，

要求された製品の出荷量と与えられた原料の利用可能量の制約を満たしつつ総コストを

最小にするには，各月における各製品の生産量と在庫量をどのように決定すればよいで

しょうか． 

変数 I1x  製品 Iの 1ヶ月目の生産量 

 II1x  製品 IIの 1ヶ月目の生産量 

 I2x  製品 Iの 2ヶ月目の生産量 

 II2x  製品 IIの 2ヶ月目の生産量 

 I3x  製品 Iの 3ヶ月目の生産量 

 II3x  製品 IIの 3ヶ月目の生産量 

 I1y  製品 Iの 1ヶ月目の在庫量 

 II1y  製品 IIの 1ヶ月目の在庫量 

 I2y  製品 Iの 2ヶ月目の在庫量 

 II2y  製品 IIの 2ヶ月目の在庫量 

   

目的関数 (最小化)  総コスト 

I1 II1 I1 II1 I2 II2 I2 II2 I3 II375 50 8 7 75 50 8 7 75 50x x y y x x y y x x          

   

非負制約 I1 0x  , II1 0x    

 I2 0x  , II2 0x    

 I3 0x  , II3 0x    

 I1 0y  , II1 0y    

 I2 0y  , II2 0y    

   

原料の制約 I1 II12 7 920x x   原料 Aについて（1ヶ月目） 

 I1 II15 3 790x x   原料 Bについて（1ヶ月目） 

 I2 II22 7 750x x   原料 Aについて（2ヶ月目） 

 I2 II25 3 600x x   原料 Bについて（2ヶ月目） 

 I3 II32 7 500x x   原料 Aについて（3ヶ月目） 

 I3 II35 3 480x x   原料 Bについて（3ヶ月目） 

   

出荷量の制約 I1 I1 30x y   製品 Iについて（1ヶ月目） 

 II1 II1 20x y   製品 IIについて（1ヶ月目） 

 I2 I1 I2 60x y y    製品 Iについて（2ヶ月目） 

 II2 II1 II2 50x y y    製品 IIについて（2ヶ月目） 

 I3 I2 80x y   製品 Iについて（3ヶ月目） 

 II3 II2 90x y   製品 IIについて（3ヶ月目） 

問題を SIMPLE で記述すると以下のようになります．  

 

 

multiplan1.smp 

// 変数 

Variable x11(name=”製品 Iの 1ヶ月目の生産量”); 

Variable x21(name=”製品 IIの 1ヶ月目の生産量”); 

Variable x12(name=”製品 Iの 2ヶ月目の生産量”); 

Variable x22(name=”製品 IIの 2ヶ月目の生産量”); 

Variable x13(name=”製品 Iの 3ヶ月目の生産量”); 

Variable x23(name=”製品 IIの 3ヶ月目の生産量”); 

Variable y11(name=”製品 Iの 1ヶ月目の在庫量”); 

Variable y21(name=”製品 IIの 1ヶ月目の在庫量”); 

Variable y12(name=”製品 Iの 2ヶ月目の在庫量”); 

Variable y22(name=”製品 IIの 2ヶ月目の在庫量”); 

// 目的関数 

Objective z(name="総コスト", type=minimize); 

z=75*x11+50*x21+8*y11+7*y21+75*x12+50*x22+8*y12+7*y22+75*x13+50*x23; 

// 非負制約 

x11 >= 0; x21 >= 0; 

x12 >= 0; x22 >= 0; 

x13 >= 0; x23 >= 0; 

y11 >= 0; y21 >= 0; 

y12 >= 0; y22 >= 0; 

// 原料の制約 

2*x11 + 7*x21 <= 920; 

5*x11 + 3*x21 <= 790; 

2*x12 + 7*x22 <= 750; 

5*x12 + 3*x22 <= 600; 

2*x13 + 7*x23 <= 500; 

5*x13 + 3*x23 <= 480; 

// 出荷量の制約 

x11        - y11 == 30; 

x21        - y21 == 20; 

x12 + y11 - y12 == 60; 

x22 + y21 - y22 == 50; 

x13 + y12        == 80; 

x23 + y22        == 90; 

// 求解 

solve(); 

// 出力 

z.val.print(); 

より汎用的に問題を定式化すると以下のようになります． 

 

   

集合 {I, II}Product   製品 

 { , }Material A B  原料 

 {1,2,3}Period   期間 

   

定数 , ,ijuse i Product j Material   
製品 i を生産するのに必要な原

料 j の使用量 

 , Pr , itout i oduct t Period  t ヶ月目の製品 i の出荷量 

 , ,jtupper j Material t Period   t ヶ月目の原料 j の利用可能量 

 ,icostp i Product  製品 i の生産コスト 

 ,icosti i Product  製品 i の在庫コスト 

   

変数 , ,itx i Product t Period   製品 i の t ヶ月目の生産量 

 , ,ity i Product t Period   製品 i の t ヶ月目の在庫量 

   

目的関数 

(最小化) 

( )it it it it

t Period i Product

costp x costi y
 

   総コスト 

   

制約 0, ,itx i Product t Period      生産量の非負制約 

 0, ,ity i Product t Period      在庫量の非負制約 

 

, ,

ij it jt

i Product

use x upper

j Material t Period





   


 原料の使用量の制約 

 
1 1 1i i ix y out   1ヶ月目の出荷量について 

 
2 1 2 2i i i ix y y out    2ヶ月目の出荷量について 

 
3 2 3i i ix y out   3ヶ月目の出荷量について 

次に，定数をデータファイルから与える場合のモデルを示します． 

 

 

multiplan2.smp 

// 集合と添字 

Set Product(name="製品"); 

Element i(set=Product); 

Set Material(name="原料"); 

Element j(set=Material); 

OrderedSet Period(name="期間"); 

Element t(set=Period); 

// パラメータ 

Parameter use(name="原料使用量", index=(i,j)); 

Parameter out(name="出荷量", index=(i,t)); 

Parameter upper(name="取扱可能量", index=(j,t)); 

Parameter costp(name="生産コスト", index=i); 

Parameter costi(name="在庫コスト", index=i); 

// 変数 

Variable x(name="生産量", index=(i,t)); 

Variable y(name="在庫量", index=(i,t)); 

// 目的関数 

Objective z(name="総コスト", type=minimize); 

z = sum(costp[i]*x[i,t]+ costi[i]*y[i,t], (i, t)); 

// 非負制約 

x[i,t] >= 0; 

y[i,t] >= 0; 

// 原料の制約 

sum(use[i,j]*x[i,t],i) <= upper[j,t]; 

// 在庫量の制約 

x[i,t]            - y[i,t] == out[i,t], t == 1; 

x[i,t] + y[i,t-1] - y[i,t] == out[i,t], t == 2; 

x[i,t] + y[i,t-1]          == out[i,t], t == 3; 

// 求解 

solve(); 

// 出力 

z.val.print(); 

x.val.print(); 

y.val.print(); 

データファイル（.dat形式）は以下のようになります． 

 

 

 

このモデルを実行すると，以下のような結果が得られます．  

data2.dat 

原料使用量 = 

["I", "A"] 2 

["I", "B"] 5 

["II", "A"] 7 

["II", "B"] 3 

; 

 

出荷量 = 

["I", 1] 30 

["I", 2] 60 

["I", 3] 80 

["II", 1] 20 

["II", 2] 50 

["II", 3] 90 

; 

 

取扱可能量 = 

["A", 1] 920 

["A", 2] 750 

["A", 3] 500 

["B", 1] 790 

["B", 2] 600 

["B", 3] 480 

; 

 

生産コスト = 

["I"] 75 

["II"] 50 

; 

 

在庫コスト = 

["I"] 8 

["II"] 7 

; 

なお，一つ注意点があります． 

上記の求解実行において，在庫量に関する添字付きの変数を 3 ヶ月目についても用意してい

ます（在庫量["I",3]，在庫量["II",3]）．これらは，この章のはじめに定式化した際には，

用意していなかった変数です． 

しかし，これらの変数について非負制約以外の制約がないこと，および目的関数（総コスト）

は最小化についてのものであること，この両者から，在庫量["I",3]，在庫量["II",3] の求

解結果は上記のようにほぼ 0と等しい値になります． 

従って在庫量["I",3]，在庫量["II",3] を定義しても，一般性は失われません． 

総コスト=21199.2 

生産量["I",1]=38 

生産量["I",2]=67.8621 

生産量["I",3]=64.1379 

生産量["II",1]=20 

生産量["II",2]=86.8966 

生産量["II",3]=53.1034 

在庫量["I",1]=8 

在庫量["I",2]=15.8621 

在庫量["I",3]=6.15667e-07 

在庫量["II",1]=2.24177e-06 

在庫量["II",2]=36.8966 

在庫量["II",3]=7.03619e-07 

2.4 包絡分析法（DEA）モデル（SAMPLES¥例題集 1.prj 参照） 

包絡分析法（DEA；Data Envelopment Analysis）とは，複数の企業体の相対的な効率性

を測定する方法で，評価対象の具体例としては，銀行・病院があげられます． 

以下の例題では，チェーン展開している 6 店舗に対して，DEA に基づいた効率性の評価を行

います．なお，本節では[3]を参考文献としています． 

 

 

この問題の定式化は次のようになります． 

 

集合 

 

 

 

I  入力項目集合 

J  出力項目集合 

K  店舗集合 

K  

対象店舗（単一要素から

なる集合） 

定数 

 

 

, , ,　k iinD k K i I  入力データ 

, , ,　k joutD k K j J  出力データ 

変数 ,　iinW i I  入力項目に対する重み 

 ,　joutW j J  出力項目に対する重み 

目的関数 (最大化) ,
,

jk j
k j

outD outW  
対象店舗に対する出力 

制約式 ,
1,ik i

i

inD inW k K   　  対象店舗の入力条件 

（例題） 

ある会社は，以下の 6店舗を抱えている． 

 

店舗番号 1 2 3 4 5 6 

店員数 5 10 20 20 30 50 

稼働時間 24 12 12 24 12 12 

売上 2 6 10 12 12 20 

  

各店舗が，全店舗に対して相対的に効率的であるかどうかを判定せよ． 

 



 26 

 
, , ,k j j k i i

j i

outD outW inD inW

k K



 

 

 

全店舗に対する入出力

条件 

 0 ,  　iinW i I  

重みの非負条件 
 0 ,  　joutW j J  

 

以下で，本問題の定式化の説明を行います． 

DEAでは，本例題の店舗のような分析対象を DMU（Decision Making Unit）と呼びます．

全ての DMUを一度に評価することは困難なため，店舗集合 K に属するある店舗 k K K   に

着目した問題を考えます． 

対象とする問題は，効率を最大化するような問題です．この効率
k

f を以下で定義することと

します． 

,

,

jk jj

k

ik ii

outD outW
f

inD inW




 

即ち，対象店舗 k  に対する，（総出力/総入力）を効率と定め，これを最大化するような問題

を扱います．なお，この問題の変数となる各入出力項目に対する重みは，各店舗共通で与えられ

るものとします．制約条件としては，任意の店舗に関して，（総出力/総入力）が 1 以下になる

という条件が必要です．即ち， 

,

,

1,  



　
k j jj

k i ii

outD outW
k K

inD inW
 

となります．また，各入出力項目に対する重みの非負制約も考慮する必要があります．なお，こ

の問題は目的関数・制約式ともに分数式で表わされる分数計画問題となります．これらの分数関

数の分母は正であるため，制約式は両辺に分母部分をかけても一般性は失われなく，また目的関

数
k

f についても，分母部分を 1と等価であるという制約式で表わして，分子部分のみを目的関

数としても一般性は失われません． 

このようにして，等価な線形計画問題に置き換えた問題の定式化がはじめに示したものとなり

ます． 

以上が，本問題に対する定式化の説明です． 

この問題を SIMPLEで記述すると，以下のようになります． 

 

 

DEA.smp 

// 入出力項目集合 

Set I(name="入力項目集合"); Element i(set=I); 

Set J(name="出力項目集合"); Element j(set=J); 

// 店舗集合 

Set K(name="店舗集合"); Element k(set=K); 

// 対象店舗 

Set Kbar(name="対象店舗", superSet=K); Element kbar(set=Kbar); 

// 入力データ 

Parameter inD(name="入力データ", index=(k,i)); 

// 出力データ 

Parameter outD(name="出力データ", index=(k,j)); 

// 変数 

// 入力項目に対する重み 

Variable inW(name="inW", index=i); 

// 出力 

Variable outW(name="outW", index=j); 

// 目的関数 

Objective f(name="f", type=maximize); 

f = sum(outD[kbar,j]*outW[j], (kbar,j)); 

// 制約式 

sum(inD[kbar,i]*inW[i], i) == 1; 

sum(outD[k,j]*outW[j], j) <= sum(inD[k,i]*inW[i], i); 

// 非負条件 

0 <= inW[i]; 

0 <= outW[j]; 

options.method = "simplex"; // 単体法の利用 

// 求解 

solve(); 

// 結果出力 

f.val.print(); 

inW.val.print(); 

outW.val.print(); 

入力データとしては以下の 3種類（csvファイル：2種類，datファイル：1種類）を用意す

る必要があります． 

 

（csv形式ファイル） 

 

 

（dat形式ファイル） 

 

 

 

対象店舗（店舗番号：1～6）に関して逐次変化させて解いた結果を以下の表にまとめます．

なお，一部結果は小数点以下第四位を四捨五入した値で表わしています． 

 

店舗番号 1 2 3 4 5 6 

目的関数値 0.667 1 1 1 0.9 1 

入力

項目 

重み（店員数） 0.2 0.067 0.04 0.033 0.025 0.017 

重み（稼働時間） 0 0.028 0.017 0.014 0.021 0.014 

出力

項目 
重み（売上） 0.333 0.167 0.1 0.083 0.075 0.05 

 

目的関数値が 1 となる店舗（店舗番号：2，3，4，6）は効率的であるといえます．一方，1

より小さな値となる店舗（店舗番号：1，5）は，各種制約を満たすどのような重みを選択して

も効率が 1になりえないということで，非効率であるということになります． 

非効率な店舗に関しては，非効率となっている原因を形成している店舗集合を参照集合として

導きだすことも可能です．詳細については，[4] を参照ください． 

inD.csv 

入力データ,定員数,稼働時間 

1,5,24 

2,10,12 

3,20,12 

4,20,24 

5,30,12 

6,50,12 

outD.csv 

出力データ,売上 

1,2 

2,6 

3,10 

4,12 

5,12 

6,20 

Kbar.dat 

対象店舗 = 1; 

2.5 ナップサック問題（SAMPLES¥例題集 1.prj 参照） 

ナップサック問題は，ナップサックの中にいくつかの品物を詰め込み入れた品物の総価値を最

大にするという問題です．ただし，ナップサックと品物にはそれぞれ容量が与えられていて，入

れた品物の総容量がナップサックの容量を超えてはいけないという条件があります．この問題は，

組合せ最適化問題の代表的な例の一つとしてよく知られていて，プロジェクトの選択や物資の購

入などの問題に応用されています．以下は，整数ナップサック問題と呼ばれるものです．なお，

0－1ナックサック問題につきましては，本節の最後で紹介します． 

 

まず，変数は各品物を詰め込む個数です．よって，この変数は整数値しか取らないということ

になります．ただし，「-1個詰め込む」というようなありえない答えを排除する必要があります．

このため，各変数は 0 以上の値しか取らないということを制約条件として明示しておく必要が

あります．なお，「0個詰め込む」は「その品物を詰め込まない」と解釈します． 

次に，最大化することになる目的関数は詰め込んだものの総価値です．これは，各品物につい

て「1個あたりの価値」と「その品物を詰め込んだ個数」の積を求め，その総和を取ることで表

現できます． 

制約条件は，先ほど述べた変数に関するものの他に，詰め込んだ品物の総容量がナップサック

の容量を超えないというものがあります．目的関数の時と同様に考えると，各品物に関する「1

個あたりの容量」と「その品物を詰め込んだ個数」の積の総和をとると詰め込んだ品物の総容量

が得られます．よって，この総和がナップサックの容量である 65を超えないということを式で

表せばよいことになります． 

（例題） 

容量 65 のナップサックに次の表にある品物を詰め込むことにします．この時，詰め

込んだ品物の総価値を最大にするためには何をいくつ詰め込むと良いでしょうか．た

だし，同じ品物を何個詰め込んでも良いものとします． 

 

品物 1個あたりの価値 1個あたりの容量 

缶コーヒー １２０ １０ 

水入りペットボトル １３０ １２ 

バナナ ８０ ７ 

りんご １００ ９ 

おにぎり ２５０ ２１ 

パン １８５ １６ 

以上のことから，この例題は次のように定式化することが出来ました． 

 

整数変数 coffee 缶コーヒーの個数 

 water  水入りペットボトルの個数 

 banana  バナナの個数 

 apple  りんごの個数 

 _rice ball  おにぎりの個数 

 bread  パンの個数 

   

目的関数 

（最大化） 

120 130 80 100 250 _ 185coffee water banana apple rice ball bread      

 総価値を最大化する 

   

制約条件 10 12 7 9 21 _ 16 65coffee water banana apple rice ball bread       

  容量に関する制約 

 0coffee   缶コーヒーは 0個以上詰め込む 

 0water   水入りペットボトルは 0個以上詰め込む 

 0banana   バナナは 0個以上詰め込む 

 0apple   りんごは 0個以上詰め込む 

 _ 0rice ball   おにぎりは 0個以上詰め込む 

 0bread   パンは 0個以上詰め込む 

定式化した結果を SIMPLEで記述すると次のようになります． 

 
 

knapsack.smp 

//整数変数を宣言する 

IntegerVariable coffee(name= "缶コーヒーの個数"); 

IntegerVariable water(name="水入りペットボトルの個数"); 

IntegerVariable banana(name="バナナの個数"); 

IntegerVariable apple(name="りんごの個数"); 

IntegerVariable rice_ball(name="おにぎりの個数"); 

IntegerVariable bread(name="パンの個数"); 

 

//総価値を最大化する 

Objective total_value(name="総価値",type=maximize); 

total_value=120*coffee+130*water+80*banana+100*apple+ 

250*rice_ball+185*bread; 

 

//容量に関する制約 

10*coffee+12*water+7*banana+9*apple+21*rice_ball+16*bread<=65; 

 

//各品物は 0個以上詰め込む 

coffee>=0; 

water>=0; 

banana>=0; 

apple>=0; 

rice_ball>=0; 

bread>=0; 

 

//求解し結果を出力する 

solve(); 

coffee.val.print(); 

water.val.print(); 

banana.val.print(); 

apple.val.print(); 

rice_ball.val.print(); 

bread.val.print(); 

total_value.val.print(); 

このモデルを NUOPTで実行すると，最後に 

 

という表示がされます．そして，この表示から「缶コーヒーを 3 個，バナナを 2 個，そしてお

にぎりを 1個詰め込むと良い」というこの例題の答えを確認できます． 

 

ところで，このモデルについて品物の種類などを変更したい場合 SIMPLE での記述を修正す

る箇所が多く大変な手間がかかってしまいます．この対策として，ナップサックの容量，品物の

価値および品物の容量を別に用意した dat ファイルから与えることにします．このようにする

ことで，SIMPLE での記述が汎用的なものになり，品物の種類が変わったとしても dat ファイ

ルの変更のみで対応できるようになります．そのために，ここでは「品物の集合」という概念を

導入します．すると定式化は次のように書き直すことができます． 

 

集合 { , }Object  缶コーヒー水入りペットボトル,バナナ,りんご,おにぎり,パン  

  品物の集合 

   

整数変数 ,icount i Object  品物 iを詰め込む個数 

   

定数 capacity  ナップサックの容量 

 ,ivalue i Object  品物 iの 1個あたりの価値 

 ,iweight i Object  品物 iの 1個あたりの容量 

   

目的関数 

（最大化） 

i i

i Object

value count


  総価値を最大化する 

   

制約条件 i i

i Object

weight count capacity


   容量に関する制約 

 0,icount i Object    各品物は 0個以上詰め込む 

 

この定式化を SIMPLE で記述すると，以下のような簡潔なものになります．なお，品物の集

缶コーヒーの個数=3 

水入りペットボトルの個数=0 

バナナの個数=2 

りんごの個数=0 

おにぎりの個数=1 

パンの個数=0 

総価値=770 

合の具体的な要素については NUOPTでは datファイルから自動的に認識します． 

 

 

knapsack.smp 

//品物の集合を宣言する 

Set Object; 

Element i(set=Object); 

 

//データファイルから与えるパラメータを宣言する 

Parameter capacity(name="ナップサックの容量"); 

Parameter value(name="品物の価値",index= i); 

Parameter weight(name="品物の容量",index= i); 

 

//整数変数を宣言する 

IntegerVariable count(name="詰め込む個数",index= i); 

 

//総価値の最大化 

Objective total_value(name="総価値",type=maximize); 

total_value=sum(value[i]*count[i], i); 

 

//容量に関する制約 

sum(weight[i]*count[i], i)<=capacity; 

 

//各品物は 0個以上詰め込む 

count[i]>=0; 

 

//求解し結果を出力する 

solve(); 

count.val.print(); 

total_value.val.print(); 

なお，今回の例題についての datファイルは次のようになります． 

 

 

data_knapsack.dat 

ナップサックの容量=65; 

品物の価値= 

[缶コーヒー]120 

[水入りペットボトル]130 

[バナナ]80 

[りんご]100 

[おにぎり]250 

[パン]185; 

品物の容量= 

[缶コーヒー]10 

[水入りペットボトル]12 

[バナナ]7 

[りんご]9 

[おにぎり]21 

[パン]16; 

最後に，この例題では「缶コーヒーが 3 個」というように容量が許す限り同じ品物を何個で

も詰め込むことができました．それでは，各品物についてナップサックに詰め込むことができる

のは 1 個だけということにするとどのようになるでしょうか．なお，この制限を加えると 0-1

ナップサック問題という典型的な 0-1 整数計画問題になります．SIMPLE では，0-1 変数であ

るという宣言を簡単に行うことができます．具体的には，先ほど汎用化させた SIMPLE モデル

中で変数を宣言している部分を 

 

とするだけです．さらに，この宣言から各変数がとりうる値は 0 か 1 しかないということが明

らかなため，各品物は 0 個以上詰め込むという制約「count[i]>=0;」を記述する必要がなく

なります．以上の点についてモデルファイルを書き換えた(knapsack3.smp 参照)上で， 

NUOPTで実行させると， 

 

 

 

という結果が得られ，缶コーヒー以外の品物を詰め込むと良いという事がわかります． 

詰め込む個数["おにぎり"]=1 

詰め込む個数["りんご"]=1 

詰め込む個数["バナナ"]=1 

詰め込む個数["パン"]=1 

詰め込む個数["缶コーヒー"]=0 

詰め込む個数["水入りペットボトル"]=1 

総価値=745 

IntegerVariable count(name="詰め込む個数",index=i, type=binary); 

2.6 集合被覆問題(SAMPLES¥例題集 1.prj  参照) 

集合 U とその部分集合の族および各部分集合に対応するコストが与えられているものとし

ます．この時，全ての U の要素をカバーするように部分集合の族から部分集合を選び，その際

にかかるコストを最小にするという問題が集合被覆問題です．応用例には乗務員スケジューリン

グ問題などがあります．なお集合被覆問題の場合， U の各要素について複数個の部分集合でカ

バーすることを許しています．このことに関連して，複数個の部分集合でカバーすることを許さ

ない場合，集合分割問題と呼ばれ，選挙区の設定問題などに応用されています． 

 

 

 

この例題の変数は，各候補者について「採用する」もしくは「採用しない」という状態を表現

できるものとなります．よって，ここでは各候補者に対し採用する場合は 1，採用しない場合は

0を取るような変数 x
(候補者)を導入します． 

この時，目的関数はどのように表すことができるでしょうか．まず，佐藤を例に実際にかかる

（例題） 

ある企業は A，B，C，D，E，F，G の 7 つのエリアがある都市で宅配便の配達事業を始

めるため，配達員を採用することになりました．この都市には配達員の候補は 10 人い

て，各人が配達できるエリアおよび配達を依頼した際にかかるコストは次の表のように

なっています． 

 

候補者 配達可能エリア コスト 

佐藤 Ａ，Ｂ，Ｃ 200 

鈴木 Ａ，Ｄ，Ｆ 280 

高橋 Ｂ，Ｅ 175 

田中 Ｃ，Ｄ，Ｅ，Ｆ，Ｇ 560 

渡辺 Ａ，Ｆ 205 

伊藤 Ｂ，Ｄ，Ｆ 245 

山本 Ｄ 80 

中村 Ｃ，Ｇ 195 

小林 Ｃ，Ｆ，Ｇ 265 

斉藤 Ｂ，Ｅ，Ｇ 190 

 

この時，最も少ないコストで 7つのエリアすべてに配達するためには誰を採用すると良

いでしょうか．ただし，配達可能エリアの一部のみを依頼する（例：伊藤に Bと Dのみ

依頼する）ことはできないものとします． 

コストを表現します．佐藤を採用した場合（ 1x 
佐藤

の場合）には 200のコストがかかります．

一方，採用しなかった場合（ 0x 
佐藤

の場合）は佐藤については何もコストがかかりません．

言い換えると，かかったコストが 0 であるということになります．よって，先ほど導入した変

数を用い 2 つの場合をまとめると，佐藤については実際には 200x
佐藤

のコストがかかったと表

現できます．他の候補者に対しても同様に実際のコストを表し，その総和を取ると実際にかかっ

た総コストとなります．この総コストが，最小化することになる目的関数です． 

制約条件については，「各エリアに 1人以上配置されている」という条件を式で表現すること

になります．例えば，エリア A の場合佐藤・鈴木・渡辺の中から 1 人以上採用することで条件

を満たすことができます．このことを式で表すと 1x x x  渡辺鈴木佐藤
となります．他のエリ

アについても同様に考えることで制約条件を表現できます． 

以上のことから，この例題は 0-1整数計画問題として次のように定式化できます． 

 

0-1 変数 x
佐藤

　, x鈴木　, x高橋　, x田中　, x渡辺　, x
伊藤

　, x山本　, x中村　, x小林　, x
斉藤

　 

 各候補者について採用する時 1，採用しない時 0を取る変数 

   

目的関数 

（最小化） 

200 280 175 560 205 245 80 195 265 190x x x x x x x x x x        渡辺鈴木 田中 山本 小林高橋 中村佐藤 伊藤 斉藤
 

  総コストの最小化 

   

制約 1x x x  渡辺鈴木佐藤
 エリア Aで配達可能にする 

 1x x x x   高橋佐藤 伊藤 斉藤
 エリア Bで配達可能にする 

 1x x x x   田中 小林中村佐藤
 エリア Cで配達可能にする 

 1x x x x   鈴木 田中 山本伊藤
 エリア Dで配達可能にする 

 1x x x  田中高橋 斉藤
 エリア Eで配達可能にする 

 1x x x x x    渡辺鈴木 田中 小林伊藤
 エリア Fで配達可能にする 

 1x x x x   田中 小林中村 斉藤
 エリア Gで配達可能にする 

この定式化した結果についてそのまま SIMPLEで記述すると次のようになります． 

 

cover1.smp 

//変数の宣言 

IntegerVariable x_sato(name="佐藤",type=binary); 

IntegerVariable x_suzuki(name="鈴木",type=binary); 

IntegerVariable x_takahashi(name="高橋",type=binary); 

IntegerVariable x_tanaka(name="田中",type=binary); 

IntegerVariable x_watanabe(name="渡辺",type=binary); 

IntegerVariable x_ito(name="伊藤",type=binary); 

IntegerVariable x_yamamoto(name="山本",type=binary); 

IntegerVariable x_nakamura(name="中村",type=binary); 

IntegerVariable x_kobayashi(name="小林",type=binary); 

IntegerVariable x_saito(name="斉藤",type=binary); 

//目的関数の設定 

Objective total_cost(type=minimize); 

total_cost=200*x_sato+280*x_suzuki+175*x_takahashi+560*x_tanaka+ 

205*x_watanabe+245*x_ito+80*x_yamamoto+195*x_nakamura+ 

265*x_kobayashi+190*x_saito; 

//各エリアに 1人以上配置する 

x_sato+x_suzuki+x_watanabe>=1; 

x_sato+x_takahashi+x_ito+x_saito>=1; 

x_sato+x_tanaka+x_nakamura+x_kobayashi>=1; 

x_suzuki+x_tanaka+x_ito+x_yamamoto>=1; 

x_takahashi+x_tanaka+x_saito>=1; 

x_suzuki+x_tanaka+x_watanabe+x_ito+x_kobayashi>=1; 

x_tanaka+x_nakamura+x_kobayashi+x_saito>=1; 

//求解し解を出力する 

solve(); 

x_sato.val.print(); 

x_suzuki.val.print(); 

x_takahashi.val.print(); 

x_tanaka.val.print(); 

x_watanabe.val.print(); 

x_ito.val.print(); 

x_yamamoto.val.print(); 

x_nakamura.val.print(); 

x_kobayashi.val.print(); 

x_saito.val.print(); 

total_cost.val.print(); 

この記述を見て分かるように，このままですと修正が大変ですし汎用的でもありません．この

ため，汎用的になるように定式化した結果を見直すことにします． 

前節のナップサック問題の時と同様に「候補者の集合」および「エリアの集合」という概念を

導入します．また，各候補者のコストは定数として外部から与えることにします． 

さらに，制約条件についても見直します．例えば，エリア A に 1 人以上配置されているとい

う制約条件 1x x x  渡辺鈴木佐藤
については，ほかの候補者も考慮すると 

1 1 0 0 1 0 0 0 0 0 1x x x x x x x x x x         渡辺鈴木 田中 山本 小林高橋 中村佐藤 伊藤 斉藤
 

と記述できます．ここで，この式の各変数についている係数 0 または 1 を外部から定数として

与えることにします．また，他の地域についても同様に定数を導入します．すると，各エリアに

対する制約条件は全て同じ枠組みで定式化できます．その結果，SIMPLEでは一般的な形での記

述で対応でき，よりわかりやすいものになります． 

以上のことを反映し，汎用化させた結果は以下のようになります． 

 

集合 { , , , , , , }Man  佐藤 鈴木高橋田中渡辺伊藤山本,中村,小林,斉藤  

  候補者集合 

 { , , , , , , }Area A B C D E F G  エリア集合 

   

0-1 変数 ix i Man,  
候補者 iを採用する時 1， 

採用しない時 0を取る変数 

   

定数 ij, ,deliver i Man j Area   
候補者 iがエリア jに配達可能な時 1， 

不可能な時 0を取る 

 
icost i Man,  候補者 iを採用した際のコスト 

   

目的関数 

（最小化） 

i i

i Man

cost x


  総コストの最小化 

   

制約 i j 1,i

i Man

deliver x j Area


    
すべてのエリアで配達可能になる 

ようにする 

これを SIMPLE で記述すると，次のようになります．なお，制約条件についてはデータファ

イルの内容をもとに各エリアに対して自動的に生成されます． 

 

 

 

実行させる際には，次の配達可能エリアを表す csvファイル（areadata_cover2.csv）と

コストを表す datファイル（costdata_cover2.csv）を与えます． 

 

areadata_cover2.csv 

配達可能エリア,A,B,C,D,E,F,G 

佐藤,1,1,1,0,0,0,0 

鈴木,1,0,0,1,0,1,0 

高橋,0,1,0,0,1,0,0 

田中,0,0,1,1,1,1,1 

渡辺,1,0,0,0,0,1,0 

伊藤,0,1,0,1,0,1,0 

山本,0,0,0,1,0,0,0 

中村,0,0,1,0,0,0,1 

小林,0,0,1,0,0,1,1 

斉藤,0,1,0,0,1,0,1 

 costdata_cover2.dat 

コスト= 

[佐藤]200 

[鈴木]280 

[高橋]175 

[田中]560 

[渡辺]205 

[伊藤]245 

[山本]80 

[中村]195 

[小林]265 

[斉藤]190; 

cover2.smp 

//候補者集合およびエリア集合の宣言 

Set Man; 

Element i(set=Man); 

Set Area; 

Element j(set=Area); 

//パラメータおよび変数の宣言 

Parameter deliver(name="配達可能エリア",index=(i,j)); 

Parameter cost(name="コスト",index=i); 

IntegerVariable x(name="採用",index=i,type=binary); 

//総コストを最小化する 

Objective total_cost(name="総コスト",type=minimize); 

total_cost=sum(cost[i]*x[i],i); 

//各エリアに配置する 

sum(deliver[i,j]*x[i],i)>=1; 

//求解し結果を表示する 

solve(); 

x.val.print(); 

total_cost.val.print(); 

このモデルファイルを実行させることにより，佐藤・伊藤・斉藤の 3 人を採用すると良く，

この場合の総コストは 635であるということが分かります． 

 

ここで，NUOPT の解法について少し述べておきます．今回の例題では厳密解法を用い求解を

しました．一方で，0-1整数計画問題の場合には近似解法である wcspを用い近似解を求めるこ

ともできます． wcspを用いた際の近似解を求めたい場合には，SIMPLEモデル中の solve();

より前に 

 

と記述します．なお，wcspを用いる場合には終了条件に関する定数をいくつか設定することが

できます．例えば，wcspを最大で 1秒間実行し終了したい場合には，モデルファイルに 

 

と記述します．詳しいことにつきましては NUOPT/SIMPLEマニュアルを参考にしてください． 

 

最後に，この 3 人を採用した場合，エリアＢは 3 人とも担当可能になっています．このよう

なダブりを許さないように，例題に「1つのエリアはちょうど 1人で担当する」という制約を加

えてみるとどのようになるでしょうか．これは先ほどの SIMPLE モデル中の制約条件

「sum(deliver[i,j]*x[i],i)>=1;」を「sum(deliver[i,j]*x[i],i)==1;」とするこ

とで表現できます．（cover3.smp参照）実行すると，鈴木・高橋・中村を採用すると良く，こ

の場合の総コストは 650であるという結果が得られます． 

options.maxtim=1; 

options.method="wcsp"; 

2.7 最大流問題（SAMPLES¥例題集 1.prj 参照） 

この節では，有向グラフをもとにしたデータ構造であるネットワークに関する基本的な問題の

一つである最大流問題を取り上げます．最大流問題とは，ネットワーク上で始点（Source）か

ら終点（Sink）まで流すことができる量の最大値を求める問題です．ただし，各辺には流すこ

とができる量の上限が与えられています．さらに，始点と終点以外の点については，「流入して

くる総量と流出していく総量は一致する」という関係（「保存則」と言います）が成り立つ必要

があります．最大流問題は，配送や通信の分野などで応用が可能です． 

 

まず変数として，ネットワーク上で始点（X町の製粉工場）から終点（Y市のレストラン）へ

1日に輸送する小麦粉の量（今後は「総輸送量」と呼びます）と各辺に対し輸送する小麦粉の量

を用意します．こうすると，目的関数は変数「総輸送量」自身となります． 

次に，制約条件について考えます．まず，2点間の輸送量については，負の値は許さず，対応

する辺に与えられている上限を超えてはならないという制約条件が必要です．このことは，上下

限制約として表現できます．さらに，各地点に対しては保存則が成立している必要があります．

ただし，製粉工場については，各地点への輸送量の和が総輸送量と一致するということになりま

す．また，レストランについては，各地点からの輸送量の和が総輸送量と一致しなければなりま

せん．  

（例題） 

ある企業は，X町にある製粉工場で製造した小麦粉を Y市のレストランに納めています．

この際，下の図のように，製粉工場からレストランまで輸送する間にいくつかの問屋を

経由しています．また，2つの地点の間で 1日に運べる小麦粉の量の上限（kg）が図の

数字の通りに決まっています．この時，製粉工場からレストランへ 1日に納めることの

できる小麦粉は最大で何 kg でしょうか．ただし，各問屋は他の地点から輸送されてき

た小麦粉を全て他の地点に輸送するものとします． 

以上のことから，この例題は次のように定式化できます． 

 

変数 total  総輸送量 

 
XAf  製粉工場から A地点への輸送量 

 
XBf  製粉工場から B地点への輸送量 

 
ACf  A地点から C地点への輸送量 

 
BCf  B地点から C地点への輸送量 

 
BDf  B地点から D地点への輸送量 

 
CYf  C地点からレストランへの輸送量 

 
DYf  D地点からレストランへの輸送量 

   

目的関数（最大化） total  総輸送量 

   

制約条件 XA XBtotal f f   製粉工場での輸送量 

 
XA ACf f  A地点での輸送量保存則 

 
XB BC BDf f f   B地点での輸送量保存則 

 
AC BC CYf f f   C地点での輸送量保存則 

 
BD DYf f  D地点での輸送量保存則 

 
CY DYf f total   レストランでの輸送量 

 0 4XAf   製粉工場から A地点への輸送量の上下限 

 0 6XBf   製粉工場から B地点への輸送量の上下限 

 0 5ACf   A地点から C地点への輸送量の上下限 

 0 4BCf   B地点から C地点への輸送量の上下限 

 0 3BDf   B地点から D地点への輸送量の上下限 

 0 8CYf   C地点からレストランへの輸送量の上下限 

 0 5DYf   D地点からレストランへの輸送量の上下限 

これを SIMPLEで記述すると以下のようになります． 

 

maxflow1.smp 

//変数の宣言 

Variable total(name="製粉工場からレストランへの総輸送量"); 

Variable f_XA(name="製粉工場から地点 Aへの輸送量"); 

Variable f_XB(name="製粉工場から地点 Bへの輸送量"); 

Variable f_AC(name="地点 Aから地点 Cへの輸送量"); 

Variable f_BC(name="地点 Bから地点 Cへの輸送量"); 

Variable f_BD(name="地点 Bから地点 Dへの輸送量"); 

Variable f_CY(name="地点 Cからレストランへの輸送量"); 

Variable f_DY(name="地点 Dからレストランへの輸送量"); 

//総輸送量の最大化 

Objective obj(name="製粉工場からレストランへの総輸送量",type=maximize); 

obj=total; 

//各地点での輸送量 

total==f_XA+f_XB; 

f_XA==f_AC; 

f_XB==f_BC+f_BD; 

f_AC+f_BC==f_CY; 

f_BD==f_DY; 

f_CY+f_DY==total; 

//輸送量に関する上下限制約 

0<=f_XA<=4; 

0<=f_XB<=6; 

0<=f_AC<=5; 

0<=f_BC<=4; 

0<=f_BD<=3; 

0<=f_CY<=8; 

0<=f_DY<=5; 

//求解して値を出力する 

solve(); 

f_XA.val.print(); 

f_XB.val.print(); 

f_AC.val.print(); 

f_BC.val.print(); 

f_BD.val.print(); 

f_CY.val.print(); 

f_DY.val.print(); 

total.val.print(); 

このモデルファイルを実行させると， 

 

という表示がされ，1日に最大 10kg小麦粉を納めることができるという結果を確認できます． 

ここで，SIMPLEでの記述について，より簡潔な形に書き直していくことにします．そのため

に，都市の集合という概念を導入します．また，2点間の輸送量の上限値を定数として外部から

与えることにします．すると，定式化については次のように書き直すことができます． 

 

集合 { , , , , , }City A B C D X Y  都市の集合 

   

変数 total  総輸送量 

 , ,ijf i City j City   iから jへの輸送量 

   

目的関数 

（最大化） 
total  総輸送量 

   

定数 , ,ijupper i City j City   iから jへの輸送量の上限 

   

制約条件 Xj

j City

total f


   
製粉工場での輸送量 

 , \{ , }ik kj

i j

f f k City X Y     
問屋での輸送量の保存則 

 iY

i City

f total


  
レストランでの輸送量 

 0 , ,ij ijf upper i j City     iから jへの輸送量の上下限 

製粉工場から地点 Aへの輸送量=4 

製粉工場から地点 Bへの輸送量=6 

地点 Aから地点 Cへの輸送量=4 

地点 Bから地点 Cへの輸送量=3.24901 

地点 Bから地点 Dへの輸送量=2.75099 

地点 Cからレストランへの輸送量=7.24901 

地点 Dからレストランへの輸送量=2.75099 

製粉工場からレストランへの総輸送量=10 

これより，SIMPLEでの記述は次のようなものになります． 

 

なお，実行時には次の csvファイルを与えます． 

 

upperdata_maxflow2.dat 

輸送量の上限,A,B,C,D,X,Y 

A,0,0,5,0,0,0 

B,0,0,4,3,0,0 

C,0,0,0,0,0,8 

D,0,0,0,0,0,5 

X,4,6,0,0,0,0 

Y,0,0,0,0,0,0 

maxflow2.smp 

//都市の集合の宣言 

Set City; 

Element i(set=City),j(set=City),k(set=City); 

//変数の宣言 

Variable total(name="製粉工場からレストランへの総輸送量"); 

Variable f(index=(i,j)); 

//2点間の輸送量の上限をパラメータとして宣言 

Parameter upper(name="輸送量の上限",index=(i,j)); 

//総輸送量の最大化 

Objective obj(name="製粉工場からレストランへの総輸送量",type=maximize); 

obj=total; 

//各地点での輸送量 

total==sum(f["X",j],j); 

sum(f[k,j],j)==sum(f[j,k],j),k!="X",k!="Y"; 

sum(f[i,"Y"],i)==total; 

//輸送量に関する上下限制約 

0<=f[i,j]<=upper[i,j]; 

//求解して解を出力する 

solve(); 

total.val.print(); 

Vector/Matrix を使用した SIMPLE モデル 

NUOPT V13 より導入された，Vector/Matrix クラスの記述を使ったモデルを紹介します． 

このモデルでは，グラフの接続行列を利用します． 

有向グラフ  ,G V E  の接続行列とは，次のように定義されます． 

定義 接続行列 

頂点集合  1,..., nV v v  と有向辺集合  1,..., mE e e  からなる有向グラフ  ,G V E  に

対して， 

頂点 v  が 辺 e  の始点であるとき 1vei   ， 

頂点 v  が 辺 e  の終点であるとき 1vei   ， 

それ以外は             0vei   

となる n m  行列 ( )veI i  を，有向グラフ  ,G V E  の接続行列という． 

 

最大流問題では，ネットワークの終点から始点へ向かう辺 flow  を加えた次の図を考えます．

辺 flowの輸送量の上限は，他の辺の輸送用の上限の総和とします． 

 

図のグラフの接続行列を I  ，各辺の輸送量を成分とするベクトルを x  ，各辺の輸送量の

上限値を成分とするベクトルをupper とおくと， 

max

s.t ,

,

0

flowx

I 





0x

x upper

x  

という線形計画問題に帰着できます． 

定式化については以下のように書き直します． 

集合 
{ , , , , , }City A B C D X Y  都市の集合 

{ 1, 2, 3, 4, 5, 6, 7, }Edge e e e e e e e flow  辺の集合 

   

変数 ,ex e Edge  各辺の輸送量 

   

定数 
( , ) , ,I v e v City e Edge   接続行列 

,eupper e Edge  各辺の輸送量の上限 

   

目的関数（最大化） 
flowx  総輸送量 

   

制約条件 I 0x  輸送量の保存則 

 x upper  各輸送量の上限 

 0x  各輸送量の下限 

これにより，SIMPLE での記述は次のようになります． 

 

実行時の csv ファイルは以下の二つとなります． 

 

maxflow3.smp 

Set City, Edge; 

Element v(set=City), e(set=Edge); 

 

Matrix I((v,e)); 

Vector Upper(e); 

 

Variable xvar(index=e); 

Vector x(e); 

x[e] = xvar[e]; // ベクトル x に変数 xvar を詰め込む 

 

Objective total(type=maximize); 

total = xvar["flow"]; 

 

I * x == zeros(City); 

zeros(Edge) <= x; 

x <= Upper; 

 

options.method="simplex"; 

 

solve(); 

I_maxflow3.csv 

I,e1,e2,e3,e4,e5,e6,e7,flow 

A,-1,0,1,0,0,0,0,0 

B,0,-1,0,1,1,0,0,0 

C,0,0,-1,-1,0,1,0,0 

D,0,0,0,0,-1,0,1,0 

X,1,1,0,0,0,0,0,-1 

Y,0,0,0,0,0,-1,-1,1 

uperdata_maxflow3.csv 

e,Upper 

e1,4 

e2,6 

e3,5 

e4,4 

e5,5 

e6,6 

e7,7 

flow,37 

2.7.1 最小カット問題 

最大流問題を，線形計画問題の一般形に定式化したことを利用して，その双対問題である， 

最小カット問題を SIMPLE で記述してみましょう． 

 

最大流問題の制約式 I 0x  ， x upper  に対する双対変数を，それぞれ y ，z  とします． 

また，最大流問題の目的関数 flowx  は， 

1,

0 ,

flow

e

a

a e flow



   

であるようなベクトル a  を用いて a x と表すことができます． 

よって最小カット問題は， 

T

min ( )

s.t ,

0.

I

    

 



0upper z y upper z

y z a

z  

のように記述することができます． 

 

定式化については，次のようになります． 

集合 
{ , , , , , }City A B C D X Y  都市の集合 

{ 1, 2, 3, 4, 5, 6, 7, }Edge e e e e e e e flow  辺の集合 

   

変数 ,vy v City   

 ,ez e Edge   

   

定数 

( , ) , ,I v e v City e Edge   接続行列 

,eupper e Edge  各辺の輸送量の上限 

0 ( ), 1e flowa e flow a     

   

目的関数（最大化） upper z  カット総量 

   

制約条件 
TI y + z a  双対制約式 

 0z  
不等号制約式に対する 

双対変数は 0以上 

   



 51 

 

これより，SIMPLE の記述は次のようになります． 

 

入力データは，最大流問題と同じです． 

このモデルファイルを実行すると， 

 

という表示がされ，最大流問題と同じ目的関数の値を得ることができます． 

mincut.smp 

Set City, Edge; 

Element v(set=City), e(set=Edge); 

 

Matrix I((v,e)); 

Vector Upper(e); 

 

Variable yvar(index=v), zvar(index=(e)); 

Vector y(v), z(e); 

y[v] = yvar[v]; // ベクトル y に変数 yvar を詰め込む 

z[e] = zvar[e]; // ベクトル z に変数 zvar を詰め込む 

 

Objective cut(name=”カットの総量”,type=minimize); 

cut = inprod(Upper,z); 

 

Vector a(e); 

a = zeros(Edge); 

a["flow"] = 1; 

 

trans(I) * y + z >= a; 

z >= zeros(Edge) ; 

 

options.method="simplex"; 

 

solve(); 

cut.val.print(); 

カットの総量=10 

2.8 最小費用流問題(SAMPLES¥例題集 1.prj 参照) 

前節の最大流問題では，小麦粉の輸送を例にネットワークを流れる量をできるだけ多くしよう

という方針で最適化を行いました．このため，小麦粉を輸送する際にかかる燃料費などの様々な

コストを考慮していませんでした．また，製粉工場から 1 日に出荷する小麦粉の量が決まって

いるような場合には別の方針で最適化を行う必要があります． 

決まった量をネットワーク上で始点から終点まで流す際にかかる費用（コスト）を最小にしよ

うという問題のことを最小費用流問題といいます．なお，最大流問題のときと同様に，保存則が

成立する必要があります．また，「決まった量」が 1である場合には最小路問題と呼ばれ，カー

ナビゲーションシステムでのルート探索や鉄道の経路案内などに応用されています． 

 

 

定式化について，前節の最大流問題の時との違いは 

・総輸送量（前節の例題での totalという変数）が 8に決まっている． 

・目的関数が総費用の最小化に変わっている． 

の 2 点です．この 2 点以外については，前節の例題と同じ定式化となりますので必要に応じて

前節を参考にしてください．なお，総費用は各輸送ルートについて「小麦粉 1kgあたりの費用」

と「輸送した小麦粉の量」の積を求め総和をとることで求められます．  

（例題） 

ある企業は，X町の製粉工場で製造した小麦粉を Y市のレストランまで 1日に 8kg納め

ています．この際，右下の図のように，製粉工場からレストランまで輸送する間にいく

つかの問屋を経由しています．また，2つの地点の間で 1日に運べる小麦粉の量の上限

（kg）が図の数字のように決まっています．さらに，2つの地点の間で小麦粉を 1kg輸

送する毎に左下の表に書かれている費用がかかります．このとき，小麦粉を 8kg製粉工

場からレストランへ輸送する際にかかる総費用は最低いくらでしょうか．ただし，各問

屋は他の地点から輸送されてきた小麦粉を全て他の地点に輸送するものとします． 

 

輸送元 輸送先 １kgあた

りの費用 

  

 

製粉工場 X A 250  

製粉工場 X B 200  

A C 270  

B C 300  

B D 220  

C レストラン Y 190  

D レストラン Y 170  

以上のことから，この例題は次のように定式化できます． 

 

変数 XAf  製粉工場から A地点への輸送量 

 
XBf  製粉工場から B地点への輸送量 

 
ACf  A地点から C地点への輸送量 

 
BCf  B地点から C地点への輸送量 

 
BDf  B地点から D地点への輸送量 

 
CYf  C地点からレストランへの輸送量 

 
DYf  D地点からレストランへの輸送量 

   

目的関数（最小化） 250 200 270 300 220 190 170XA XB AC BC BD CY DYf f f f f f f       

  総費用 

   

制約条件 8 XA XBf f   製粉工場での輸送量 

 
XA ACf f  A地点での輸送量の保存則 

 
XB BC BDf f f   B地点での輸送量の保存則 

 
AC BC CYf f f   C地点での輸送量の保存則 

 
BD DYf f  D地点での輸送量の保存則 

 8CY DYf f   レストランでの輸送量 

 0 4XAf   製粉工場から A地点への輸送量の上下限 

 0 6XBf   製粉工場から B地点への輸送量の上下限 

 0 5ACf   A地点から C地点への輸送量の上下限 

 0 4BCf   B地点から C地点への輸送量の上下限 

 0 3BDf   B地点から D地点への輸送量の上下限 

 0 8CYf   C地点からレストランへの輸送量の上下限 

 0 5DYf   D地点からレストランへの輸送量の上下限 

 



 54 

この結果を SIMPLEで記述すると以下のようになります． 

 

 

mincost1.smp 

//変数の宣言 

Variable f_XA(name="製粉工場から地点 Aへの輸送量"); 

Variable f_XB(name="製粉工場から地点 Bへの輸送量"); 

Variable f_AC(name="地点 Aから地点 Cへの輸送量"); 

Variable f_BC(name="地点 Bから地点 Cへの輸送量"); 

Variable f_BD(name="地点 Bから地点 Dへの輸送量"); 

Variable f_CY(name="地点 Cからレストランへの輸送量"); 

Variable f_DY(name="地点 Dからレストランへの輸送量"); 

//総費用の最小化 

Objective totalcost(name="総費用",type=minimize); 

totalcost=250*f_XA+200*f_XB+270*f_AC+300*f_BC+220*f_BD+190*f_CY

+ 

170*f_DY; 

//各地点での輸送量 

8==f_XA+f_XB; 

f_XA==f_AC; 

f_XB==f_BC+f_BD; 

f_AC+f_BC==f_CY; 

f_BD==f_DY; 

f_CY+f_DY==8; 

//輸送量に関する上下限制約 

0<=f_XA<=4; 

0<=f_XB<=6; 

0<=f_AC<=5; 

0<=f_BC<=4; 

0<=f_BD<=3; 

0<=f_CY<=8; 

0<=f_DY<=5; 

//求解して値を出力する 

solve(); 

f_XA.val.print(); 

f_XB.val.print(); 

f_AC.val.print(); 

f_BC.val.print(); 

f_BD.val.print(); 

f_CY.val.print(); 

f_DY.val.print(); 

totalcost.val.print(); 

このモデルを実行させると， 

 

 

という表示がされ，結果を確認できます． 

ところで，このままの記述では都市の数が多い問題に応用することが困難です．そこで，都市

の集合という概念を導入し，必要なデータを定数として外部から与えるようにすることで大規模

な問題に対しても応用しやすいようにします．まず，この方針で定式化をしなおすと次のように

なります．なお，保存則については「他の地点からの輸送量の和」と「他の地点への輸送量の和」

の差のデータを外部から与える形式を取りました．ちなみに，この値が負の地点が製粉工場，正

の地点がレストラン，そして 0の地点が問屋ということになります． 

 

集合 { , , , , , }City A B C D X Y  都市の集合 

   

変数 , ,ijf i City j City   iから jへの輸送量 

   

定数 , ,ijupper i City j City   iから jへの輸送量の上限 

 , ,ijcost i City j City   
iから jへの輸送の際に 

かかる小麦粉 1kgあたりの費用 

 ,isupply i City  iでの流入量と流出量の差 

   

目的関数 

（最小化） 

ij ij

j City i City

cost f
 

   
総費用 

   

制約条件 
,ik kj k

i City j City

f f supply k City
 

      
各地点での輸送量 

 0 , ,ij ijf upper i j City     iから jへの輸送量の上下限 

製粉工場から地点 Aへの輸送量=2 

製粉工場から地点 Bへの輸送量=6 

地点 Aから地点 Cへの輸送量=2 

地点 Bから地点 Cへの輸送量=3 

地点 Bから地点 Dへの輸送量=3 

地点 Cからレストランへの輸送量=5 

地点 Dからレストランへの輸送量=3 

総費用=5260 

この定式化をし直した結果を SIMPLEで記述すると次のようになります． 

 

mincost2.smp 

//都市の集合の宣言 

Set City; 

Element i(set=City),j(set=City),k(set=City); 

 

//変数の宣言 

Variable f(name="輸送量",index=(i,j)); 

 

//パラメータの宣言 

Parameter upper(name="輸送量の上限",index=(i,j)); 

Parameter cost(name="輸送費用",index=(i,j)); 

Parameter supply(name="supply",index=k); 

 

//総費用の最小化 

Objective total_cost(name="総費用",type=minimize); 

total_cost=sum(cost[i,j]*f[i,j], (i,j)); 

 

//各地点での輸送量 

sum(f[i,k],i)-sum(f[k,j],j)==supply[k]; 

 

//輸送量に関する上下限制約 

0<=f[i,j]<=upper[i,j]; 

 

//求解して解を出力する 

solve(); 

total_cost.val.print(); 

なお，実行時には次の 2つの csvファイルと 1つの datファイルを与えます． 

 

upperdata_mincost2.csv 

輸送量の上限,A,B,C,D,X,Y 

A,0,0,5,0,0,0 

B,0,0,4,3,0,0 

C,0,0,0,0,0,8 

D,0,0,0,0,0,5 

X,4,6,0,0,0,0 

Y,0,0,0,0,0,0 

 costdata_mincost2.csv 

輸送費用,A,B,C,D,X,Y 

A,0,0,270,0,0,0 

B,0,0,300,220,0,0 

C,0,0,0,0,0,190 

D,0,0,0,0,0,170 

X,250,200,0,0,0,0 

Y,0,0,0,0,0, 0 

 

supply_mincost2.dat 

supply= 

[A]0  

[B]0  

[C]0  

[D]0  

[X]-8  

[Y]8; 

2.9 多品種流問題（SAMPLES¥例題集 1.prj 参照） 

今まで述べてきた最大流問題や最小費用流問題では，ネットワーク上を１種類のものしか流れ

ていませんでした．しかし，現実世界ではネットワーク上を複数のものが流れるというケースが

よくあります．多品種流問題は，複数の品物をそれぞれ始点から終点まで輸送するという場合に

利用される問題で，通信など多くの分野で応用されています． 

 

（例題） 

ある企業は，X町の製粉工場で製造したパン用の小麦粉を Y市のパン屋まで 1日に 8kg

納めています．さらに，同じ製粉工場で製造したうどん用の小麦粉を Z市のうどん屋ま

で 1 日に 13 kg 納めています．この際，下の図のように，製粉工場からパン屋，製粉

工場からうどん屋まで輸送する間にいくつかの問屋を経由しています．また，2 つの地

点の間で 1日に運べるパン用の小麦粉とうどん用の小麦粉の合計量の上限（kg）が図の

数字のように決まっています．さらに，2 つの地点の間でパン用の小麦粉やうどん用の

小麦粉を 1kg輸送する毎に下の表に書かれているだけの費用がかかります．このとき，

パン用の小麦粉とうどん用の小麦粉を決められた量だけ納める際にかかる総費用は 1日

につき最低いくらでしょうか． 

 

 

 

輸送元 輸送先 パン用の小麦粉の

輸送費 

うどん用の小麦粉

の輸送費 

製粉工場 X A 12 11 

製粉工場 X B 10 12 

A C 4 5 

A パン屋 Y 11 - 

B A 6 7 

B C 11 9 

C パン屋 Y 9 - 

C うどん屋 Z - 5 

まず，各輸送ルートについてパン用の小麦粉の輸送量とうどん用の小麦粉の輸送量が変数とし

て必要です．ただし，うどん屋にパン用の小麦粉を輸送したりパン屋にうどん用の小麦粉を輸送

したりすることはありませんので，これらに対応する変数はここでは宣言しないことにします． 

次に，目的関数については最小費用流問題の時と同様に考えることになります．ただし，パン

用の小麦粉とうどん用の小麦粉で輸送コストが異なる点には注意が必要です． 

最後に，制約条件を考えます．各輸送ルートについては問題文にあるようにパン用の小麦粉の

輸送量とうどん用の小麦粉の輸送量の和が上限を超えないという制約条件が必要です．また，保

存則については地点と商品に関し場合分けをする必要があります．さらに，「-1個輸送する」と

いうことはありえませんので変数について非負制約が必要です． 

以上のことから，次のように定式化できます． 

 

変数 ,XA XAPWheat UWheat  製粉工場から A地点への輸送量 

 ,XB XBPWheat UWheat  製粉工場から B地点への輸送量 

 ,AC ACPWheat UWheat  A地点から C地点への輸送量 

 
AYPWheat  A地点からパン屋への輸送量 

 ,BA BAPWheat UWheat  B地点から A地点への輸送量 

 ,BC BCPWheat UWheat  B地点から C地点への輸送量 

 
CYPWheat  C地点からパン屋への輸送量 

 
CZUWheat  C地点からうどん屋への輸送量 

   

目的関数 

（最小化） 

12 10 4 11

6 11 9

11 12 5

7 9 5

XA XB AC AY

BA BC CY

XA XB AC

BA BC CZ

PWheat PWheat PWheat PWheat

PWheat PWheat PWheat

UWheat UWheat UWheat

UWheat UWheat UWheat

  

  

  

    

  総輸送費 

   

制約条件 

15, 12,

18, 3,

3, 10,

10, 15

XA XA XB XB

AC AC AY

BA BA BC BC

CY CZ

PWheat UWheat PWheat UWheat

PWheat UWheat PWheat

PWheat UWheat PWheat UWheat

PWheat UWheat

   

  

   

   

 各輸送ルートの輸送量の上限を超えない 

8 ,

,

,

,

8

XA XB

AC AY XA BA

BA BC XB

CY AC BC

AY CY

PWheat PWheat

PWheat PWheat PWheat PWheat

PWheat PWheat PWheat

PWheat PWheat PWheat

PWheat PWheat

 

  

 

 

   

 パン用の小麦粉に関する保存則 

 

13 ,

,

,

,

13

XA XB

AC AY XA BA

BA BC XB

CZ AC BC

CZ

UWheat UWheat

UWheat UWheat UWheat UWheat

UWheat UWheat UWheat

UWheat UWheat UWheat

UWheat

 

  

 

 

  

 うどん用の小麦粉に関する保存則 

 

0, 0, 0, 0,

0, 0, 0,

0, 0, 0,

0, 0, 0

XA XB AC AY

BA BC CY

XA XB AC

BA BC CZ

PWheat PWheat PWheat PWheat

PWheat PWheat PWheat

UWheat UWheat UWheat

UWheat UWheat UWheat

   

  

  

    

 

 輸送量に関する非負制約 

これを SIMPLEで記述すると次のようになります． 

multiflow1.smp 

//変数の宣言 

Variable 

PWheat_XA,PWheat_XB,PWheat_AC,PWheat_AY,PWheat_BA,PWheat_BC, 

PWheat_CY, 

UWheat_XA,UWheat_XB,UWheat_AC,UWheat_BA,UWheat_BC,UWheat_CZ; 

//目的関数の宣言 

Objective total_cost(name="総輸送費",type=minimize); 

total_cost=12*PWheat_XA+10*PWheat_XB+4*PWheat_AC+11*PWheat_AY 

+6*PWheat_BA+11*PWheat_BC+9*PWheat_CY 

+11*UWheat_XA+12*UWheat_XB+5*UWheat_AC+7*UWheat_BA 

+9*UWheat_BC+5*UWheat_CZ; 

//輸送ルートごとの輸送量の上限 

PWheat_XA+UWheat_XA<=15; 

PWheat_XB+UWheat_XB<=12; 

PWheat_AC+UWheat_AC<=18; 

PWheat_AY<=3; 

PWheat_BA+UWheat_BA<=3; 

PWheat_BC+UWheat_BC<=10; 

PWheat_CY<=10; 

UWheat_CZ<=15; 

//パン用の小麦粉に関する保存則 

PWheat_XA+PWheat_XB==8; 

PWheat_AC+PWheat_AY==PWheat_XA+PWheat_BA; 

PWheat_BA+PWheat_BC==PWheat_XB; 

PWheat_CY==PWheat_AC+PWheat_BC; 

8==PWheat_AY+PWheat_CY; 

//うどん用の小麦粉に関する保存則 

UWheat_XA+UWheat_XB==13; 

UWheat_AC==UWheat_XA+UWheat_BA; 

UWheat_BA+UWheat_BC==UWheat_XB; 

UWheat_CZ==UWheat_AC+UWheat_BC; 

13==UWheat_CZ; 

//非負制約 

PWheat_XA>=0;PWheat_XB>=0;PWheat_AC>=0;PWheat_AY>=0; 

PWheat_BA>=0;PWheat_BC>=0;PWheat_CY>=0; 

UWheat_XA>=0;UWheat_XB>=0;UWheat_AC>=0;UWheat_BA>=0; 

UWheat_BC>=0;UWheat_CZ>=0;  

ここで，SIMPLEでの記述をより簡潔なものにしていくことにします．そのためには，都市の

集合の概念を導入し，さまざまなデータに対応できるようにすることが有効です．この際，輸送

量の上限値などの具体的なデータを出来るだけ外部から与えるようにしておきます． 

すると，定式化については次のように表現できます． 

集合 { , , , , , }City A B C X Y Z  都市の集合 

   

変数 , ,ijPWheat i City j City   
iからjへのパン用の

小麦粉の輸送量 

 , ,ijUWheat i City j City   
iからjへのうどん用

の小麦粉の輸送量 

   

定数 , ,ijupper i City j City   
iからjへの輸送量の

上限 

 _ , ,ijPWheat cost i City j City   

iからjへの輸送の際

にかかるパン用の小

麦粉 1kg あたりの費

用 

 _ , ,ijUWheat cost i City j City   

iからjへの輸送の際

にかかる 

うどん用の小麦粉

1kgあたりの費用 

 _ ,iPWheat supply i City  

i でのパン用の小麦

粉 1kg の流入量と流

出量の差 

 _ ,iUWheat supply i City  

i でのうどん用の小

麦粉 1kg の流入量と

流出量の差 

   

目的関数 

（最小化） 

( _ _ )ij ij ij ij

j City i City

PWheat cost PWheat UWheat cost UWheat
 

     

  総輸送費 

制約条件 , ,ij ij ijPWheat UWheat upper i j City     
iからjへの輸送量の

上限 

 

_ ,

ik kj

i City j City

k

PWheat PWheat

PWheat supply k City

 



  

 

　　　　　  

各地点でのパン用の

小麦粉の輸送量 

_ ,

ik kj

i City j City

k

UWheat UWheat

UWheat supply k City

 



  

 

　　　　　  

各地点でのうどん用

の小麦粉の輸送量 

 0 , ,ijPWheat i j City    
パン用の小麦粉の輸

送量の非負制約 

 0 , ,ijUWheat i j City    
うどん用の小麦粉の

輸送量の非負制約 

 

 

SIMPLEで記述すると次のようになり，先ほどのものより簡潔になっていることが分かります． 

 

 

multiflow2.smp 

//都市の集合の宣言 

Set City; 

Element i(set=City),j(set=City),k(set=City); 

//変数の宣言 

Variable PWheat(name="パン用の小麦粉の輸送数",index=(i,j)); 

Variable UWheat(name="うどん用の小麦粉の輸送数",index=(i,j)); 

//パラメータの宣言 

Parameter upper(name="輸送量の上限",index=(i,j)); 

Parameter PWheat_cost(name="パン用の小麦粉の輸送費用",index=(i,j)); 

Parameter UWheat_cost(name="うどん用の小麦粉の輸送費用",index=(i,j)); 

Parameter PWheat_supply(name="PWheat_supply",index=k); 

Parameter UWheat_supply(name="UWheat_supply",index=k); 

//目的関数の宣言 

Objective total_cost(name="総輸送費",type=minimize); 

total_cost=sum(PWheat_cost[i,j]*PWheat[i,j]+UWheat_cost[i,j]*UWhe

at[i,j],(i,j)); 

//輸送ルートごとの輸送量の上限 

PWheat[i,j]+UWheat[i,j]<=upper[i,j]; 

//パン用の小麦粉に関する保存則 

sum(PWheat[i,k],i)-sum(PWheat[k,j],j)== PWheat_supply[k]; 

//うどん用の小麦粉に関する保存則 

sum(UWheat[i,k],i)-sum(UWheat[k,j],j)== UWheat_supply[k]; 

//非負制約 

PWheat[i,j]>=0; 

UWheat[i,j]>=0; 

なお，実行時には次の 3つの csvファイル（輸送量の上限・パン用の小麦粉の輸送費用・う

どん用の小麦粉の輸送費用）と 1つの datファイル（PWheat_supplyおよび UWheat_supply）

を与えます． 

 

PWcost_multiflow2.csv 

パン用の小麦粉の輸送費用,A,B,C,X,Y,Z 

A,0,0,4,0,11,0 

B,6,0,11,0,0,0 

C,0,0,0,0,9,0 

X,12,10,0,0,0,0 

Y,0,0,0,0,0,0 

Z,0,0,0,0,0,0 

 upperdata_multiflow2.csv 

輸送量の上限,A,B,C,X,Y,Z 

A,0,0,18,0,3,0 

B,3,0,10,0,0,0 

C,0,0,0,0,10,15 

X,15,12,0,0,0,0 

Y,0,0,0,0,0,0 

Z,0,0,0,0,0,0 

 

UWcost_multiflow2.csv 

うどん用の小麦粉の輸送費用,A,B,C,X,Y,Z 

A,0,0,5,0,0,0 

B,7,0,9,0,0,0 

C,0,0,0,0,0,5 

X,11,12,0,0,0,0 

Y,0,0,0,0,0,0 

Z,0,0,0,0,0,0 

 supplydata_multiflow2.dat 

PWheat_supply=[A]0 [B]0 [C]0 

[X]-8 [Y]8 [Z]0; 

UWheat_supply=[A]0 [B]0 [C]0  

[X]-13 [Y]0 [Z]13; 

 

このモデルを実行した結果から，総費用が 494 になるように輸送すると最適であり，例えば

次の図のようにパン用の小麦粉とうどん用の小麦粉を輸送するとよいことが分かります． 

 

パン用の小麦粉の輸送量 うどん用の小麦粉の輸送量 

2.10  ｐ メディアン問題（SAMPLES¥例題集 2.prj 参照） 

施設をどの地点に配置すると最適なのかを求める施設配置問題は，都市計画などに応用されて

います．施設配置問題の中から，この節で p メディアン問題を，次の節で p センター問題を

取り上げます．なお，本チュートリアルでは各地点は需要を有する地点（今後は「需要点」と呼

びます）でありさらに施設を配置することが可能な地点でもあると仮定します． 

まず，メディアンについて説明します．メディアンとは各需要点から施設を配置した地点（今

後は「施設点」と呼びます）への移動距離の総和を最小にする点のことです．そして，このメデ

ィアンを求める問題のことをメディアン問題と呼びます．なお，一般にはどの程度の需要がある

かは需要点ごとに異なりますので重み付きの総移動距離を最小にすることになります．また，施

設を 1箇所ではなく p箇所に配置するような場合に pメディアン問題と呼びます．  

 

 

まず，この問題では各都市に対しデパートを出店するのかどうかを表す変数が必要であること

がわかります．しかし，これだけでは各都市からどちらの出店地が近いのかを求める必要があり，

複雑な式になってしまいます．そこで，都市 i について都市 j に配分する場合 1，そうでない

場合は 0を取るような 0-1変数 ijx を導入します．なお， 1iix  の時には都市 iにデパートを出

店し， 0iix  の時には出店しないという解釈をします． 

次に，制約条件について考えていきます．先ほど述べたように， iix は都市 iに出店するかど

（例題） 

ある企業は，A市から F市までの 6市の中から 2つの市に新たにデパートを出店しよう

と考えています．この時，出店地までの総移動距離を最小にするためにはどの市に出店

すると良いでしょうか．なお，各市の人口は左下の表のようになっています．また，2

市間の距離は右下の図のようになっています．ただし，直接結ばれていない 2都市間の

距離は他の都市を経由したときの最短距離を採用します．  

 

都市 人口 

A市 800 

B市 550 

C市 780 

D市 600 

E市 1020 

F市 360 
 

 

 



 66 

うかを表しています．この例題では 2 都市に出店しますので，
ii

i

x は 2 である必要がありま

す．また，各都市について近い方のデパートに配分しなければなりません（同距離の場合には任

意にどちらかを選択することになります）． ijx を用いて表現すると，各 i について ij

j

x が 1

であるということになります．最後に，今のままですと出店していない都市に配分されてしまう

可能性があります．これを防ぐためには， 0jjx  の場合に 1ijx  となってはいけないという制

約条件が必要です．言い換えると， ijx は jjx より大きい値をとらないということになります．よ

って，式で表すと ij jjx x となります．ただし，この制約条件は i j の場合には無意味ですか

ら i j の場合のみで十分であることに注意してください． 

目的関数は，重み付きの総移動距離ということになります．まず，都市 i から都市 j への移

動距離は 1ijx  の場合にのみ考慮する必要があります．よって，まず 2市 i,j間の距離と ijx の

積を求めます．そして，jについて和をとることで都市 iからの重みなしの移動距離が得られま

す．そして，この移動距離に重みである人口を掛けあわせることで都市 i からの重みをつけた

移動距離を得ます．重みをつけた移動距離を全ての都市について求め，和をとることで目的関数

である重み付きの総移動距離となります．なお，図より 2 都市間の距離は次の表の通りになる

ことが分かります． 

 

 Ａ市 Ｂ市 Ｃ市 Ｄ市 Ｅ市 Ｆ市 

Ａ市 0 7 2 13 19 17 

Ｂ市 7 0 8 6 12 10 

Ｃ市 2 8 0 14 20 18 

Ｄ市 13 6 14 0 7 4 

Ｅ市 19 12 20 7 0 3 

Ｆ市 17 10 18 4 3 0 

以上のことをまとめると，次のように定式化できます．  

 

0-1変数 

, , , , , , , , , , , , , , , , , ,

, , , , , , , , , , , , , , , , ,

AA AB AC AD AE AF BA BB BC BD BE BF CA CB CC CD CE CF

DA DB DC DD DE DF EA EB EC ED EE EF FA FB FC FD FE FF

x x x x x x x x x x x x x x x x x x

x x x x x x x x x x x x x x x x x x  

 0または 1をとる出店するのかと最も近い出店地はどこなのかを表す変数 

   

目 的 関 数

（最小化） 

800(7 2 13 19 17 )

550(7 8 6 12 10 )

780(2 8 14 20 18 )

600(13 6 14 7 4 )

1020(19 12 20 7 3 )

360(17 10 18 4 3 )

AB AC AD AE AF

BA BC BD BE BF

CA CB CD CE CF

DA DB DC DE DF

EA EB EC ED EF

FA FB FC FD FE

x x x x x

x x x x x

x x x x x

x x x x x

x x x x x

x x x x x

   

    

    

    

    

      

重み付き総移動距離 

   

制約条件 2AA BB CC DD EE FFx x x x x x       2都市に出店 

 1AA AB AC AD AE AFx x x x x x       A市は 1箇所に配分 

 1BA BB BC BD BE BFx x x x x x       B市は 1箇所に配分 

 1CA CB CC CD CE CFx x x x x x       C市は 1箇所に配分 

 1DA DB DC DD DE DFx x x x x x       D市は 1箇所に配分 

 1EA EB EC ED EE EFx x x x x x       E市は 1箇所に配分 

 1FA FB FC FD FE FFx x x x x x       F市は 1箇所に配分 

 , , , , ,AB BB AC CC AD DD AE EE AF FFx x x x x x x x x x       

 , , , , ,BA AA BC CC BD DD BE EE BF FFx x x x x x x x x x       

 , , , , ,CA AA CB BB CD DD CE EE CF FFx x x x x x x x x x       

 , , , , ,DA AA DB BB DC CC DE EE DF FFx x x x x x x x x x       

 , , , , ,EA AA EB BB EC CC ED DD EF FFx x x x x x x x x x       

 , , , ,FA AA FB BB FC CC FD DD FE EEx x x x x x x x x x       

 出店している都市に配分 

 

この結果をそのまま SIMPLEで記述すると次のようになります． 

median1.smp 

//変数の宣言 

IntegerVariable  

x_AA(type=binary),x_AB(type=binary),x_AC(type=binary), 

x_AD(type=binary),x_AE(type=binary),x_AF(type=binary), 

x_BA(type=binary),x_BB(type=binary),x_BC(type=binary), 

x_BD(type=binary),x_BE(type=binary),x_BF(type=binary), 

x_CA(type=binary),x_CB(type=binary),x_CC(type=binary), 

x_CD(type=binary),x_CE(type=binary),x_CF(type=binary), 

x_DA(type=binary),x_DB(type=binary),x_DC(type=binary), 

x_DD(type=binary),x_DE(type=binary),x_DF(type=binary), 

x_EA(type=binary),x_EB(type=binary),x_EC(type=binary), 

x_ED(type=binary),x_EE(type=binary),x_EF(type=binary), 

x_FA(type=binary),x_FB(type=binary),x_FC(type=binary), 

x_FD(type=binary),x_FE(type=binary),x_FF(type=binary); 

//重み付き総移動距離の最小化 

Objective total_distance(type=minimize); 

total_distance=800*(7*x_AB+2*x_AC+13*x_AD+19*x_AE+17*x_AF) 

+550*(7*x_BA+8*x_BC+6*x_BD+12*x_BE+10*x_BF) 

+780*(2*x_CA+8*x_CB+14*x_CD+20*x_CE+18*x_CF) 

+600*(13*x_DA+6*x_DB+14*x_DC+7*x_DE+4*x_DF) 

+1020*(19*x_EA+12*x_EB+20*x_EC+7*x_ED+3*x_EF) 

+360*(17*x_FA+10*x_FB+18*x_FC+4*x_FD+3*x_FE); 

//２都市に出店する 

x_AA+x_BB+x_CC+x_DD+x_EE+x_FF==2; 

//各都市を配分する 

x_AA+x_AB+x_AC+x_AD+x_AE+x_AF==1; 

x_BA+x_BB+x_BC+x_BD+x_BE+x_BF==1; 

x_CA+x_CB+x_CC+x_CD+x_CE+x_CF==1; 

x_DA+x_DB+x_DC+x_DD+x_DE+x_DF==1; 

x_EA+x_EB+x_EC+x_ED+x_EE+x_EF==1; 

x_FA+x_FB+x_FC+x_FD+x_FE+x_FF==1; 

//出店しない都市には配分しない 

x_AB<=x_BB; x_AC<=x_CC; x_AD<=x_DD; x_AE<=x_EE; x_AF<=x_FF; 

x_BA<=x_AA; x_BC<=x_CC; x_BD<=x_DD; x_BE<=x_EE; x_BF<=x_FF; 

x_CA<=x_AA; x_CB<=x_BB; x_CD<=x_DD; x_CE<=x_EE; x_CF<=x_FF; 

x_DA<=x_AA; x_DB<=x_BB; x_DC<=x_CC; x_DE<=x_EE; x_DF<=x_FF; 

x_EA<=x_AA; x_EB<=x_BB; x_EC<=x_CC; x_ED<=x_DD; x_EF<=x_FF; 

x_FA<=x_AA; x_FB<=x_BB; x_FC<=x_CC; x_FD<=x_DD; x_FE<=x_EE; 

ところで，この記述ですと都市の数が増えた場合に記述が大変であるなどの理由で汎用的では

ありません．そこで，モデルファイルを汎用的なものにしていくことにします．そのためには，

生データをできるだけ外部から与える必要があります．ここでは，人口と距離に関するデータを

データファイルから与えることにします．この時，様々なデータに対応できるように都市の集合

という概念を導入しておきます．すると，定式化について次のように書き直せます． 

  

集合 { , , , , , }City A B C D E F  都市の集合 

   

0-1変数 , ,ijx i City j City   

0 または 1 をとる出店するの

かとどこに配分されるのかを表

す変数 

   

定数 , ,ijdistance i City j City   都市 iから都市 jまでの距離 

 ,ipopulation i City  都市 iの人口 

   

目的関数

（最小化） ,

( )i ij ij

i City j City i j

population distance x
  

 
  

 
   重み付き総移動距離 

   

制約条件 
2ii

i City

x


  
2都市に出店 

 1,ij

j City

x i City


    各都市は 1箇所に配分 

 , , ,ij jjx x i j City i j     出店している都市に配分 

 

上記の式を SIMPLE で記述すると，次のようになります．なお，NUOPT での実行時には都市

の集合の具体的な要素はデータファイルから自動的に認識します．また，simple_printf()

を用い出店する都市のみを表示するように工夫しました． 

なお，実行時に次の 2都市間の距離を表す csvファイル（左）と都市の人口を表す datファ

イル（右）を与えます． 

 

distancedata_median2.csv 

2都市間の距離,A,B,C,D,E,F 

A,0,7,2,13,19,17 

B,7,0,8,6,12,10 

C,2,8,0,14,20,18 

D,13,6,14,0,7,4 

E,19,12,20,7,0,3 

F,17,10,18,4,3,0 

 populationdata_median2.dat 

都市の人口= 

["A"]800  

["B"]550  

["C"]780  

["D"]600  

["E"]1020  

["F"]360; 

 

このモデルを実行すると次の表示がされ，A市と E市に出店すると良いことがわかります． 

 

 

"A" 市に出店する． 

"E" 市に出店する． 

median2.smp 

//都市の集合の宣言 

Set City; 

Element i(set=City), j(set=City), k(set=City); 

//パラメータの宣言 

Parameter distance(name="2都市間の距離",index=(i,j)); 

Parameter population(name="都市の人口",index=i); 

//変数の宣言 

IntegerVariable x(index=(i,j),type=binary); 

//目的関数の宣言 

Objective total_distance(name="総移動距離",type=minimize); 

total_distance=sum(population[i]*sum(distance[i,j]*x[i,j],(j,i!=j

)),i); 

//制約条件 

sum(x[k,k],k)==2; 

sum(x[i,j],j)==1; 

x[i,k]<=x[k,k],i!=k; 

//求解し出店する都市を出力する 

solve(); 

simple_printf("%s 市に出店する．¥n",k,x[k,k].val==1); 

2.11  ｐ センター問題（SAMPLES¥例題集 2.prj 参照） 

前節の ｐ メディアン問題は，需要点と施設点の間の（重み付き）総移動距離を最小にすると

いう目的で作られた問題でした．このため，需要点から施設点までの平均的な移動距離を最小に

するような解を期待できます．しかし，解の中に需要点と施設点の間が極端に離れている組み合

わせが含まれている可能性があります．この点，この節で扱う p センター問題は，（重み付き）

移動距離が極端に大きい組み合わせを作らないように配置をする問題と見ることができます． 

ここで，センターとは最も遠い需要点からの（重み付き）距離を最小にする施設点のことで，

この点を求める問題のことをセンター問題といいます．また，p箇所に施設を配置するような場

合に ｐ センター問題と呼び，消防署のような施設を配置するような場合に用いられています．  

 

 

前節の例題との違いは目的関数の部分だけです．このため，前節の例題での変数と制約条件に

ついてはこの例題でもそのまま使用しますので詳細は前節を参照してください． 

目的関数については，都市から出店地までの重み付き移動距離の最大値ということになります．

まず，都市 i から都市 j までの重み付き移動距離を求めます．これは 1ijx  の時に意味を持つ

ものですから i-j 間の距離，i の人口そして ijx の 3 つを掛け合わせたものとなります．なお，

i j の時にはこの積は常に 0になりますので計算には含めないことにします．次に，今回のよ

うなミニマックス問題の定式化での一つのテクニックとして変数 v を新たに導入します．この

（例題） 

ある企業は，A市から F市までの 6市の中から 2つの市に新たにデパートを出店しよう

と考えています．この時，都市から出店地までの重み付き移動距離の最大値を最小にす

るためにはどの市に出店すると良いでしょうか．なお，各市の人口は左下の表のように

なっています．また，2 市間の距離は右下の図のようになっています．なお，直接結ば

れていない 2都市間の距離は他の都市を経由したときの最短距離を採用します． 

 

都市 人口 

A市 800 

B市 550 

C市 780 

D市 600 

E市 1020 

F市 360 

時，重み付き移動距離がそれぞれ v 以下であるという制約条件を加えることで v に重み付き移

動距離の最大値という意味を持たせることができます．よって，変数 v 自身が目的関数となり

ます． 

以上のことから，次のように定式化できます． 

 

0-1 変数 

, , , , , , , , , , , , , , , , , ,

, , , , , , , , , , , , , , , , ,

AA AB AC AD AE AF BA BB BC BD BE BF CA CB CC CD CE CF

DA DB DC DD DE DF EA EB EC ED EE EF FA FB FC FD FE FF

x x x x x x x x x x x x x x x x x x

x x x x x x x x x x x x x x x x x x  

 0または 1をとる出店するのかと最も近い出店地はどこなのかを表す変数 

連続変数 v  重み付き移動距離の最大値 

   

目 的 関 数

（最小化） 
v  重み付き移動距離の最大値 

  

制約条件 800 7 , 800 2 , 800 13 , 800 19 , 800 17 ,AB AC AD AE AFv x v x v x v x v x           

 550 7 , 550 8 , 550 6 , 550 12 , 550 10 ,BA BC BD BE BFv x v x v x v x v x           

 780 2 , 780 8 , 780 14 , 780 20 , 780 18 ,CA CB CD CE CFv x v x v x v x v x           

 600 13 , 600 6 , 600 14 , 600 7 , 600 4 ,DA DB DC DE DFv x v x v x v x v x           

 1020 19 , 1020 12 , 1020 20 , 1020 7 , 1020 3 ,EA EB EC ED EFv x v x v x v x v x           

 360 17 , 360 10 , 360 18 , 360 4 , 360 3FA FB FC FD FEv x v x v x v x v x           

 各重み付き移動距離は最大値を越えない 

 2AA BB CC DD EE FFx x x x x x       2都市に出店 

 1AA AB AC AD AE AFx x x x x x       A市は 1箇所に配分 

 1BA BB BC BD BE BFx x x x x x       B市は 1箇所に配分 

 1CA CB CC CD CE CFx x x x x x       C市は 1箇所に配分 

 1DA DB DC DD DE DFx x x x x x       D市は 1箇所に配分 

 1EA EB EC ED EE EFx x x x x x       E市は 1箇所に配分 

 1FA FB FC FD FE FFx x x x x x       F市は 1箇所に配分 

 , , , , ,AB BB AC CC AD DD AE EE AF FFx x x x x x x x x x      

 , , , , ,BA AA BC CC BD DD BE EE BF FFx x x x x x x x x x      

 , , , , ,CA AA CB BB CD DD CE EE CF FFx x x x x x x x x x      

 , , , , ,DA AA DB BB DC CC DE EE DF FFx x x x x x x x x x      

 , , , , ,EA AA EB BB EC CC ED DD EF FFx x x x x x x x x x      

 , , , ,FA AA FB BB FC CC FD DD FE EEx x x x x x x x x x      

 出店している都市に配分 

 

この結果を SIMPLEで記述すると次のようになります． 

center1.smp 

//変数の宣言 

IntegerVariable  

x_AA(type=binary),x_AB(type=binary),x_AC(type=binary), 

x_AD(type=binary),x_AE(type=binary),x_AF(type=binary), 

x_BA(type=binary),x_BB(type=binary),x_BC(type=binary), 

x_BD(type=binary),x_BE(type=binary),x_BF(type=binary), 

x_CA(type=binary),x_CB(type=binary),x_CC(type=binary), 

x_CD(type=binary),x_CE(type=binary),x_CF(type=binary), 

x_DA(type=binary),x_DB(type=binary),x_DC(type=binary), 

x_DD(type=binary),x_DE(type=binary),x_DF(type=binary), 

x_EA(type=binary),x_EB(type=binary),x_EC(type=binary), 

x_ED(type=binary),x_EE(type=binary),x_EF(type=binary), 

x_FA(type=binary),x_FB(type=binary),x_FC(type=binary), 

x_FD(type=binary),x_FE(type=binary),x_FF(type=binary); 

Variable v; 

//重み付き移動距離の最大値の最小化 

Objective max_distance(type=minimize); 

max_distance=v; 

v>=800*7*x_AB;  v>=800*2*x_AC;  v>=800*13*x_AD;  v>=800*19*x_AE; 

v>=800*17*x_AF; v>=550*7*x_BA;  v>=550*8*x_BC;  v>=550*6*x_BD; 

v>=550*12*x_BE; v>=550*10*x_BF; v>=780*2*x_CA;  v>=780*8*x_CB; 

v>=780*14*x_CD; v>=780*20*x_CE; v>=780*18*x_CF; v>=600*13*x_DA; 

v>=600*6*x_DB;  v>=600*14*x_DC; v>=600*7*x_DE;  v>=600*4*x_DF; 

v>=1020*19*x_EA;v>=1020*12*x_EB;v>=1020*20*x_EC;v>=1020*7*x_ED; 

v>=1020*3*x_EF; v>=360*17*x_FA; v>=360*10*x_FB; v>=360*18*x_FC; 

v>=360*4*x_FD;  v>=360*3*x_FE; 

//出店と配分に関する制約条件 

x_AA+x_BB+x_CC+x_DD+x_EE+x_FF==2; 

x_AA+x_AB+x_AC+x_AD+x_AE+x_AF==1; 

x_BA+x_BB+x_BC+x_BD+x_BE+x_BF==1; 

x_CA+x_CB+x_CC+x_CD+x_CE+x_CF==1; 

x_DA+x_DB+x_DC+x_DD+x_DE+x_DF==1; 

x_EA+x_EB+x_EC+x_ED+x_EE+x_EF==1; 

x_FA+x_FB+x_FC+x_FD+x_FE+x_FF==1; 

x_AB<=x_BB; x_AC<=x_CC; x_AD<=x_DD; x_AE<=x_EE; x_AF<=x_FF; 

x_BA<=x_AA; x_BC<=x_CC; x_BD<=x_DD; x_BE<=x_EE; x_BF<=x_FF; 

x_CA<=x_AA; x_CB<=x_BB; x_CD<=x_DD; x_CE<=x_EE; x_CF<=x_FF; 

x_DA<=x_AA; x_DB<=x_BB; x_DC<=x_CC; x_DE<=x_EE; x_DF<=x_FF; 

x_EA<=x_AA; x_EB<=x_BB; x_EC<=x_CC; x_ED<=x_DD; x_EF<=x_FF; 

x_FA<=x_AA; x_FB<=x_BB; x_FC<=x_CC; x_FD<=x_DD; x_FE<=x_EE; 

このままでは大変分かりにくいですので，前節と同様に SIMPLE での記述を簡潔なものにし

ていくことにします．この際，定式化については次のようなものにします． 

 

集合 { , , , , , }City A B C D E F  都市の集合 

   

0-1 変数 , ,ijx i City j City   
0または1をとる出店するのかと最も近い 

出店地はどこなのかを表す変数 

連続変数 v  重み付き移動距離の最大値 

   

定数 , ,ijdistance i City j City   都市 iから都市 jまでの距離 

 ,ipopulation i City  都市 iの人口 

   

目 的 関 数

（最小化） 
v  重み付き移動距離の最大値 

   

制約条件 , , ,i ij ijv population distance x i j City i j       

  各重み付き移動距離は最大値を越えない 

 
2ii

i City

x


  
2都市に出店 

 
1,ij

j City

x i City


    
各都市は 1箇所に配分 

 , , ,ij jjx x i j City i j     出店している都市に配分 

すると，SIMPLEでは次のような簡潔な記述になります． 

 

 

実行する際には，前節の例題と同様に次の 2都市間の距離を表す csvファイル（左）と各都

市の人口を表す datファイル（右）を与えます． 

 

distancedata_center2.dat 

２都市間の距離,A,B,C,D,E,F 

A,0,7,2,13,19,17 

B,7,0,8,6,12,10 

C,2,8,0,14,20,18 

D,13,6,14,0,7,4 

E,19,12,20,7,0,3 

F,17,10,18,4,3,0 

 populationdata_center2.dat 

都市の人口= 

["A"]800  

["B"]550  

["C"]780  

["D"]600  

["E"]1020  

["F"]360; 

 

このモデルを実行すると，前節の p メディアン問題の時の結果とは異なり，A 市と F 市に出

店すると良いことがわかります． 

center2.smp 

//都市の集合の宣言 

Set City; 

Element i(set=City),j(set=City),k(set=City); 

//パラメータの宣言 

Parameter distance(name="２都市間の距離",index=(i,j)); 

Parameter population(name="都市の人口",index=i); 

//変数の宣言 

IntegerVariable x(index=(i,j),type=binary); 

Variable v; 

//目的関数の宣言 

Objective max_distance(type=minimize); 

max_distance=v; 

//制約条件 

v>=population[i]*distance[i,j]*x[i,j]; 

sum(x[k,k],k)==2; 

sum(x[i,j],j)==1; 

x[i,k]<=x[k,k],i!=k; 

//求解し出店する都市を出力する 

solve(); 

simple_printf("%s 市に出店する．¥n", k,x[k,k].val==1); 

2.12 巡回セールスマン問題（SAMPLES¥例題集 2.prj 参照） 

巡回セールスマン問題とは，セールスマンがある都市から出発し，全ての都市を訪問して，出

発地点に帰還する場合，どのような順番で都市を回るのが最短経路であるか，という問題です．

この問題は最後に訪れる都市が出発地点でなくてもよいという設定もありますが，ここでは，最

後に訪れる都市は出発地点であるという設定の問題を紹介します．また，一度訪れた都市を経由

して別の都市に移動するような場合は考えないこととします． 

 

 

この問題を NUOPT で解くために定式化を行います．本例題の定式化は文献 [4] からの引用

です．“すべての都市を訪問して出発地点に帰還する”ということは，一筆書きの経路を考える

ため，経路のどの都市を出発地点に解釈してもよいことになります．ここでは仮に都市 A を出

発地点と解釈することにします． 

変数について考えましょう．どの都市からどの都市に移っていくかを決定する問題ですので，

ある都市から別のある都市に移動する場合に 1 をとり，移動しない場合に 0 をとるような 0-1

変数 , , , , , , , , , , ,AB AC AD BC BD CD BA CA DA CB DB DCx x x x x x x x x x x x を導入します．例えば， ABx は都市

Aから都市 Bに移動する場合に 1をとり，移動しない場合には 0をとります． 

次に，最小化すべき目的関数は経路長になりますので，各都市間の距離を用いて， 

6 5 5 7 4 3 6 5 5 7 4 3AB AC AD BC BD CD BA CA DA CB DB DCx x x x x x x x x x x x            

と表現することができます． 

（例題） 

次のグラフは都市 A，B，C，Dと，そのリンクの距離関係を表した図です． 

 

ある都市から出発し，全ての都市を訪問して，出発地点に帰還する場合，どのような順

番で都市を回ると最短経路となるでしょうか． 

最後に制約条件です．巡回セールマン問題特有の制約，“すべての都市を訪問して帰還する”

を定式化する必要があります．この制約は，“各都市から別都市の 1つに移動する”，“各都市に

別都市の 1つから移動する”という制約を含みます．例えば，都市 Aについては， 

1AB AC ADx x x   ：都市 Aから別都市の 1つに移動する 

1BA CA DAx x x   ：都市 Aに別都市の 1つから移動する 

という制約が必要です． 

上記の制約だけでは不十分であり，もう少し制約を加える必要があります．なぜならば，上記

の場合には， 1, 1AB BA CD DCx x x x    となるような解，つまり一筆書きで全都市を訪問す

ることができないような経路が許されてしまうからです．巡回セールスマン問題では，一筆書き

の閉路をツアーと呼び，「都市 A→都市 B→都市 A」のような，すべての都市を訪問していない

ツアーを特にサブツアーと呼びます．巡回セールスマン問題では，上記制約に加えて“すべての

サブツアーを排除する”という制約が必要になります．この制約を表現するために，中間変数と

呼ばれる変数 , ,B C Dy y y を導入し，都市 B，都市 C，都市 Dが関わるツアーを排除するという制

約を導入します．この制約は， 

3 2B C BCy y x    ・・・(1) 

3 2B D BDy y x    ・・・(2) 

3 2C B CBy y x    ・・・(3) 

3 2C D CDy y x    ・・・(4) 

2 2D B DBy y x    ・・・(5) 

3 2D C DCy y x    ・・・(6) 

と表現することができます．上記制約から，例えば，(1)，(3)の制約の両辺を足し合わせると， 

3 3 4BC CBx x   

となり， 1BC CBx x  ，すなわち「都市 B→都市 C→都市 B」のようなサブツアーを排除し，ま

た，(1)，(４)，(5)の制約の両辺を足し合わせると， 

3 3 3 6BC CD DBx x x    

となり， 1BC CD DBx x x   ，すなわち，「都市 B→都市 C→都市 D→都市 B」のようなサブツ

アーを排除することができます．都市 A に関する情報が式に現れないのは，解となるツアーま

でもを排除しないようにするためです．つまり，都市 A を含むツアーの内，都市 A 以外で構成

されるツアーをすべて排除することを表現しています． 

ところで中間変数 , ,B C Dy y y は，訪れる都市の順番と考えられます．例えば，都市 B→都市 C

なる経路を持つ，すなわち 1BCx  と仮定します．制約式(1)は 1C By y   を示します．また，

(1)~(6)がすべて同一の形をしていることから，max( ; ) min( ; ) 2c cy c City y c City     も

わかるため，等号 1C By y  が成立することが導かれます．このようにして， cy （ c  は B, C, 

D のいずれか）はステップ幅１をもって全順序が与えられます． 次の制約式 (7)，(8)，(9)

を導入することで， cy は１から始まるように調整できます．このようにして全順序と，都市 A

を出発してから再び都市 Aに戻ってくるまでの巡回順を一致させることが可能です． 

1 3By   ・・・(7) 

1 3Cy   ・・・(8) 

1 3Dy   ・・・(9) 



 79 

 

以上のことから，次のように定式化することができます． 

 

0-1変数 ABx  都市Aから都市Bへ移動するか否か 

 ACx  都市Aから都市Cへ移動するか否か 

 ADx  都市Aから都市Dへ移動するか否か 

 BCx  都市Bから都市Cへ移動するか否か 

 BDx  都市Bから都市Dへ移動するか否か 

 CDx  都市Cから都市Dへ移動するか否か 

 BAx  都市Bから都市Aへ移動するか否か 

 CAx  都市Cから都市Aへ移動するか否か 

 DAx  都市Dから都市Aへ移動するか否か 

 CBx  都市Cから都市Bへ移動するか否か 

 DBx  都市Dから都市Bへ移動するか否か 

 DCx  都市Dから都市Cへ移動するか否か 

中間変数 By  都市 Bを巡回する順番 

 Cy  都市 Cを巡回する順番 

 Dy  都市 Dを巡回する順番 

目的関数(最小化) 6 5 5 7 4 3AB AC AD BC BD CDx x x x x x      

 6 5 5 7 4 3BA CA DA CB DB DCx x x x x x       

  経路長 

制約 1AB AC ADx x x    都市Aから別都市の 1つに移動する 

 1BA BC BDx x x    都市Bから別都市の 1つに移動する 

 1CA CB CDx x x    都市Cから別都市の 1つに移動する 

 1DA DB DCx x x    都市Dから別都市の 1つに移動する 

 1BA CA DAx x x    都市Aに別都市の 1つから移動する 

 1AB CB DBx x x    都市Bに別都市の 1つから移動する 

 1AC BC DCx x x    都市Cに別都市の 1つから移動する 

 1AD BD CDx x x    都市Dに別都市の 1つから移動する 

 3 2B C BCy y x    サブツアー排除制約 

 3 2B D BDy y x    サブツアー排除制約 

 3 2C B CBy y x    サブツアー排除制約 

 3 2C D CDy y x    サブツアー排除制約 

 3 2D B DBy y x    サブツアー排除制約 

 3 2D C DCy y x    サブツアー排除制約 

 1 3By   都市 Bを訪れる順番の上下限 

 1 3Cy   都市 Cを訪れる順番の上下限 

 1 3Dy    都市 Dを訪れる順番の上下限 

この問題は，変数に 0-1 変数と連続変数が混ざっており，目的関数，制約式全てが線形なの

で，混合整数線形計画問題となります．定式化した結果を SIMPLE で記述すると以下のように

なります． 

 

 

TSP1.smp 

//変数 

IntegerVariable x_AB(name="AtoB",type=binary); 

IntegerVariable x_AC(name="AtoC",type=binary); 

IntegerVariable x_AD(name="AtoD",type=binary); 

IntegerVariable x_BC(name="BtoC",type=binary); 

IntegerVariable x_BD(name="BtoD",type=binary); 

IntegerVariable x_CD(name="CtoD",type=binary); 

IntegerVariable x_BA(name="BtoA",type=binary); 

IntegerVariable x_CA(name="CtoA",type=binary); 

IntegerVariable x_DA(name="DtoA",type=binary); 

IntegerVariable x_CB(name="CtoB",type=binary); 

IntegerVariable x_DB(name="DtoB",type=binary); 

IntegerVariable x_DC(name="DtoC",type=binary); 

//目的関数 

Objective route(name="経路長",type=minimize); 

route=6*x_AB+5*x_AC+5*x_AD+7*x_BC+4*x_BD+3*x_CD+6*x_BA 

+5*x_CA+5*x_DA+7*x_CB+4*x_DB+3*x_DC; 

//中間変数 

Variable y_B; 

Variable y_C; 

Variable y_D; 

// 各都市から別都市の 1つに移動する 

x_AB+x_AC+x_AD==1; 

x_BA+x_BC+x_BD==1; 

x_CA+x_CB+x_CD==1; 

x_DA+x_DB+x_DC==1; 

// 各都市に別都市の 1つから移動する 

x_BA+x_CA+x_DA==1; 

x_AB+x_CB+x_DB==1; 

x_AC+x_BC+x_DC==1; 

x_AD+x_BD+x_CD==1;  

// サブツアーを排除する 

y_B-y_C+3*x_BC<=2; 

y_B-y_D+3*x_BD<=2; 

y_C-y_B+3*x_CB<=2; 

y_C-y_D+3*x_CD<=2; 

y_D-y_B+3*x_DB<=2; 

y_D-y_C+3*x_DC<=2; 

//各都市を通る順番は出発地点以外の都市数以下 

1<=y_B<=3; 

1<=y_C<=3; 

1<=y_D<=3; 

//求解 

solve(); 

//出力 

x_AB.val.print(); 

x_AC.val.print(); 

x_AD.val.print(); 

x_BC.val.print(); 

x_BD.val.print(); 

x_CD.val.print(); 

x_BA.val.print(); 

x_CA.val.print(); 

x_DA.val.print(); 

x_CB.val.print(); 

x_DB.val.print(); 

x_DC.val.print(); 

より汎用的に問題を定式化すると以下のようになります． 

 

   

集合 { , , , }City A B C D  都市の集合 

 _ { , , }City B C D  出発地点以外の都市の集合 

   

定数 1 2 , 1 , 2c cdis c City c City   都市 1c と都市 2c の距離 

 num  集合 _City の要素数 

   

0-1変数 1 2 , 1 , 2c cx c City c City   
都市 1c から都市 2c に移動す

るか否か 

中間変数 , _cy c City  都市 c を通る順番 

目的関数 (最小化) 1 2 1 2

1 , 2

c c c c

c City c City

dis x
 

  経路長 

   

制約 1 2

2

1, 1c c

c City

x c City


    
都市 1c から別都市の 1 つに移

動 

 1 2

1

1, 2c c

c City

x c City


    
都市 2c に別都市の 1 つから移

動 

 
1 2 1 2 1, 1 _, 2 _, 1 2c c c cy y num x num c City c City c c       

 

  サブツアーを排除 

 1 3, _cy c City      都市 c  を通る順番の上下限 

次に，定数（各都市間の距離）をデータファイルから与える SIMPLE モデルを示します．こ

のようにモデルとデータを分離することにより，都市数が変わったとしてもデータファイルを変

更するだけで対応できるようになります． 

 

 

TSP2.smp 

//集合と添え字 

Set City(name="都市"); 

Element c1(set=City); 

Element c2(set=City); 

Set City_(name="出発地点を除く都市"); 

Element c_(set=City_); 

Element c1_(set=City_); 

Element c2_(set=City_); 

//パラメータ 

Parameter dis(name="距離",index=(c1,c2)); 

Parameter num = City_.card(); 

//変数 

IntegerVariable x(name="移動",index=(c1,c2), type=binary); 

Variable y(name="順番",index=c_); 

//目的関数 

Objective route(name="経路長",type=minimize); 

route = sum(x[c1,c2]*dis[c1,c2], ((c1,c2),c1!=c2)); 

// 各都市から 1つの別都市へ移動する 

sum(x[c1,c2],(c2,c1!=c2))==1; 

// 各都市に 1つの別都市から移動する 

sum(x[c1,c2],(c1,c1!=c2))==1; 

// サブツアーを排除する 

y[c1_] - y[c2_] + num*x[c1_,c2_] <= num-1, c1_!=c2_; 

// 順番の上下限制約 

  1 <= y[c_] <= num; 

//求解 

solve(); 

//出力 

x.val.print(); 

y.val.print(); 

route.val.print(); 

データファイル（.dat形式）は以下のようになります． 

 

 

このモデルを実行すると， [ ] 3, [ ] 1, [ ] 2y B y C y D    となり，都市 A→都市 C→都市 D→

都市 B→都市 Aの順番で移動するのが最適解となります．その経路長は 18であることがわかり

ます． 

data.dat 

"都市" = A B C D ; 

"出発地点を除く都市" = B C D ; 

 

"距離" =  

  [A,B] 6  [A,C] 5  [A,D] 5 

  [B,C] 7  [B,D] 4 

  [C,D] 3 

  [B,A] 6  [C,A] 5  [D,A] 5 

  [C,B] 7  [D,B] 4 

  [D,C] 3 

  ; 

2.13 割当問題（SAMPLES¥例題集 2.prj 参照） 

割当問題は現在多くの業務に利用されています．ここでは基本的な割当問題の例題を取り上げ，

割当問題の定式化のテクニック，モデル化の例，大規模な問題に対するアプローチ法を紹介しま

す． 

 

2.13.1  割当問題とは 

割当問題とは，集合 A の要素を集合 B の要素のどれに割り当てるかを決定する問題です． 

 

 

また，割当問題はマス目を埋める問題に置き換えることができます．上記の図を，マス目を埋

めるイメージで表すと以下の図のようになります． 

 

 

 

 

 

集合 A 集合 B 

a1 

a2 

a3 

 

・・・  

b1 

b2 

b4 

b3 

 

・・・  

  a1 a2 a3     

b1  ○         

b2      ○     

b3           

b4    ○       

            

            

 

 

割当問題はマスの埋め方のルールを制約条件として与えることによって，個々の問題に対応す

る定式化を行うことができます．割当問題の応用例として，例えば以下のものがあります． 

 

 人員配置問題 

 シフトスケジューリング問題 

 配車計画問題 

 訪問スケジューリング決定問題 

 マーケットテリトリー決定問題 

 

2.13.2  基礎的なマス埋め割当問題 

ここではまず，マス目を埋めるだけの問題を考えていきます． 

 

 

（例題） 

４×４のマス目があります．このマスに石を置きたいのですが，全ての縦横に関して

石の数が２つになるように配置します．さてどのように置けば良いでしょうか． 

  1 2 3 4 

1         

2         

3         

4         

 

 

横に 2 個ずつ 

縦に 2 個ずつ 

この問題を定式化するためには， 0-1 変数を用いる必要があります．この問題の場合以下の

図のように，各マスに対して 0-1 変数を対応させます． 

 

  1 2 3 4 

1 x11 x12 x13 X14 

2 x21 x22 x23 X24 

3 x31 x32 x33 X34 

4 x41 x42 x43 X44 

 

定式化をすると以下のようになります． 

 

0-1 変数 

11 12 13 14, , ,x x x x  

21 22 23 24, , ,x x x x  

31 32 33 34, , ,x x x x  

41 42 43 44, , ,x x x x  

それぞれのマスに石を置くならば 1 置かない

ならば 0 ． 

   

目的関数  この問題には目的関数はない 

   

制約条件 11 12 13 14 2x x x x     １行目を横に見たときに石を２つ置く 

 
21 22 23 24 2x x x x     ２行目を横に見たときに石を２つ置く 

 
31 32 33 34 2x x x x     ３行目を横に見たときに石を２つ置く 

 
41 42 43 44 2x x x x     ４行目を横に見たときに石を２つ置く 

 
11 21 31 41 2x x x x     １列目を縦に見たときに石を２つ置く 

 
12 22 32 42 2x x x x     ２列目を縦に見たときに石を２つ置く 

 
13 23 33 43 2x x x x     ３列目を縦に見たときに石を２つ置く 

 
14 24 34 44 2x x x x     ４列目を縦に見たときに石を２つ置く 

定式化した結果を SIMPLE で記述すると以下のようになります． 

 

fieldassign1.smp 

IntegerVariable x_11(type = binary); 

IntegerVariable x_21(type = binary); 

IntegerVariable x_31(type = binary); 

IntegerVariable x_41(type = binary); 

 

IntegerVariable x_12(type = binary); 

IntegerVariable x_22(type = binary); 

IntegerVariable x_32(type = binary); 

IntegerVariable x_42(type = binary); 

 

IntegerVariable x_13(type = binary); 

IntegerVariable x_23(type = binary); 

IntegerVariable x_33(type = binary); 

IntegerVariable x_43(type = binary); 

 

IntegerVariable x_14(type = binary); 

IntegerVariable x_24(type = binary); 

IntegerVariable x_34(type = binary); 

IntegerVariable x_44(type = binary); 

 

x_11 + x_12 + x_13 + x_14 == 2; 

x_21 + x_22 + x_23 + x_24 == 2; 

x_31 + x_32 + x_33 + x_34 == 2; 

x_41 + x_42 + x_43 + x_44 == 2; 

 

x_11 + x_21 + x_31 + x_41 == 2; 

x_12 + x_22 + x_32 + x_42 == 2; 

x_13 + x_23 + x_33 + x_43 == 2; 

x_14 + x_24 + x_34 + x_44 == 2; 

 

// 以下は出力用のプログラム 

solve(); 

simple_printf("%d %d %d %d¥n",x_11,x_12,x_13,x_14); 

simple_printf("%d %d %d %d¥n",x_21,x_22,x_23,x_24); 

simple_printf("%d %d %d %d¥n",x_31,x_32,x_33,x_34); 

simple_printf("%d %d %d %d¥n",x_41,x_42,x_43,x_44); 

 



 89 

 

このモデルを NUOPTで実行すると，最後に 

 

という表示がされ，この例題の答えを確認できます．どの縦横にも二箇所ずつ石が置いてある

（1 と表示されている）のが分かります． 

1 0 1 0 

1 1 0 0 

0 1 0 1 

0 0 1 1 

さて，次にこの問題を SIMPLE の添字機能を用いてモデル化してみます．定式化は以下のよ

うに変更します． 

 

集合 {1,2,3,4}I   行 

 {1,2,3,4}J   列 

   

0-1 変数 , ,ijx i I j J   
マス ( , )i j に石を置くならば 1ijx  ，置かないなら

ば 0ijx  とする． 

   

目的関数  この問題には目的関数はない 

   

制約条件 
2,ij

i

x j J    各列 j  は横に見たときに石を２つ置く 

 2,ij

j

x i I    各行 i  は縦に見たときに石を２つ置く 

   

 

 

添字を用いることにより，以下のように簡単にモデル記述することができます． 

 

 

fieldassign2.smp 

Set I = "1 2 3 4"; 

Set J = "1 2 3 4"; 

Element i(set = I); 

Element j(set = J); 

IntegerVariable x(type = binary,index = (i,j)); 

sum(x[i,j],i) == 2; 

sum(x[i,j],j) == 2; 

 

// 以下は出力用のプログラム 

solve(); 

simple_printf("%d %d %d %d¥n", 

x[i,j],x[i,j+1],x[i,j+2],x[i,j+3],j == 1); 

2.13.3 仕事割当問題 

ここでは，仕事を人に効率よく割り当てる問題を取り上げます．次の例題を考えます． 

 

（例題）「安藤」「佐藤」「鈴木」「山本」「渡辺」 の 5 人に仕事を割り当てます．仕事

は 「接客」「厨房」「レジ打ち」「発注」「ごみ捨て」「買出し」「掃除」「仕込み」の 8 つ

です．各人を仕事に割り当てるにはコストがかかり，それは個人・仕事によって決まり

ます．また，各人はそれぞれの仕事に対して熟練度があり，熟練度が高いほどコストが

かかる傾向があります．以下は熟練度とコストをまとめたものです． 

 

熟練度 安藤 佐藤 鈴木 山本 渡辺 

接客 -1 3 -2 3 -4 

厨房 5 -2 3 -4 5 

レジ打ち 0 3 -2 3 -1 

発注 -3 -1 1 1 2 

ごみ捨て 1 -1 3 1 -1 

買出し -1 -1 1 0 2 

掃除 2 -2 2 -3 4 

仕込み 5 -2 0 1 5 

 

コスト 安藤 佐藤 鈴木 山本 渡辺 

接客 570 1400 520 1410 450 

厨房 1800 1000 1700 1050 2300 

レジ打ち 800 1500 500 1500 600 

発注 500 600 1000 1000 1200 

ごみ捨て 800 600 1200 800 600 

買出し 600 600 800 700 1300 

掃除 1200 500 1200 500 1300 

仕込み 1500 1000 1200 1200 1500 

また，以下の点を守らなくてはいけません． 

 各人に割り振る仕事は，最大で３つまでとする． 

 「接客」「厨房」「レジ打ち」「掃除」「仕込み」は２人を割り当てる． 

 「発注」「ごみ捨て」「買出し」は１人を割り当てる． 

 「接客」「厨房」は別の人が担当する． 

 各仕事において，仕事を担当する人の熟練度の和を，その仕事のクオリティーとする． 

 各仕事のクオリティーは負になってはいけない． 

 

このとき，コストの合計を最小にするような割り当て方を求めて下さい． 

この問題も前節の問題と同様に以下のようなマス目に○をつける問題として考えることがで

きます． 

 

 
 

以上を踏まえて，以下のように定式化することができます． 

 

集合 
JOB ｛接客，厨房，レジ打ち，発注，ごみ捨て，買出し，掃除，仕込み｝ 

仕事の集合 

 
PEOPLE ｛安藤，佐藤，鈴木，山本，渡辺｝ 

人の集合 

   

0-1 変数 , ,jpx j JOB p PEOPLE   

仕事 j を人 p に割り当てるならば

1jpx  ，そうでないならば 0jpx  とす

る 

定数 , ,jpcost j JOB p PEOPLE   仕事 j を人 p に割り当てる際のコスト 

 , ,jpjyukuren j JOB p PEOPLE   仕事 j を人 p が行う際の熟練度 

   

目的関数 

（最小化） ,

jp jp

j p

cost x  
コストの総和 

   

横に見た場合必ず 

○が １つか 2つある． 

縦に見た場合，○

が最大 3つある． 

 

各人，接客，

厨房，につい

ては最大１つ

までしか○が

付かない． 

制約条件 
2jp

p

x j   , ｛接客，厨房，レジ打ち，掃除，仕込み｝ 

 「接客」「厨房」「レジ打ち」「掃除」「仕込み」は２人を割り当てる． 

 1,jp

p

x j   ｛発注，ごみ捨て，買出し｝ 

 「発注」「ごみ捨て」「買出し」は１人を割り当てる． 

 3,jp

j

x p PEOPLE    
各人には，最大３つまでの仕事を割り当

てることができる 

 

0,jp jp

p

jyukuren x j JOB   

 

各仕事のクオリティーが負になっては

いけない 

 
, { }

1,jp

j j

x p PEOPLE


  
接客,厨房

 

 

 

 

接客，厨房は違う人が担当する（同じ人が接客と厨房を兼ねない）． 

今回は定数に関してはファイルからデータを与えてみます．モデル部分は以下のようになりま

す． 

 

 

jobassign1.smp 

Set Job; 

Set People; 

Element j(set = Job); 

Element p(set = People); 

 

IntegerVariable x(type = binary, index = (j,p)); 

Parameter cost(index = (j,p), name = "コスト"); 

Parameter jyukuren(index = (j,p), name = "熟練度"); 

 

sum(x[j,p],p) == 2, 

 j == "接客" || j == "厨房" || j == "レジ打ち" || 

 j == "掃除" || j == "仕込み"; 

sum(x[j,p],p) == 1, 

  j == "発注" || j == "ごみ捨て" || j == "買出し" ; 

sum(x[j,p],j) <= 3; 

sum(x[j,p]*jyukuren[j,p],p) >= 0; 

sum(x[j,p],(j, j == "接客" || j == "厨房")) <= 1; 

 

Objective total_cost(type = minimize, name = "総コスト"); 

total_cost = sum(cost[j,p]*x[j,p],(j,p)); 

 

// 以下出力用プログラム 

solve(); 

total_cost.val.print(); 

simple_printf("%s,%s¥n",j,p,x[j,p].val == 1); 

simple_printf("¥n"); 

simple_printf("%sのクオリティーは  %d¥n", 

j,sum(jyukuren[j,p]*x[j,p].val,p)); 

以下が csv 形式の入力ファイルです．コストに関してと熟練度に関しての二種類のファイル

を用意する必要があります． 

 

 

jyukuren.dat 

熟練度,安藤,佐藤,鈴木,山本,渡辺 

接客,-1,3,-2,3,-4 

厨房,5,-2,3,-4,5 

レジ打ち,0,3,-2,3,-1 

発注,-3,-1,1,1,2 

ごみ捨て,1,-1,3,1,-1 

買出し,-1,-1,1,0,2 

掃除,2,-2,2,-3,4 

仕込み,5,-2,0,1,5 

cost.dat 

コスト,安藤,佐藤,鈴木,山本,渡辺 

接客,570,1400,520,1410,450 

厨房,1800,1000,1700,1050,2300 

レジ打ち,800,1500,500,1500,600 

発注,500,600,1000,1000,1200 

ごみ捨て,800,600,1200,800,600 

買出し,600,600,800,700,1300 

掃除,1200,500,1200,500,1300 

仕込み,1500,1000,1200,1200,1500 

このモデルを NUOPTで実行すると，最後に 

 

 

という表示がされ，この例題の答えを確認できます． 

 

 

（例題）先ほどの問題に以下の条件を付け加えて下さい． 

 

安藤に「接客」をさせることはできない 

佐藤に「厨房」をさせることはできない 

鈴木に「レジ打ち」をさせることはできない 

山本に「発注」をさせることはできない 

渡辺に「ごみ捨て」をさせることはできない 

総コスト=13380 

"ごみ捨て","安藤" 

"レジ打ち","佐藤" 

"レジ打ち","渡辺" 

"仕込み","山本" 

"仕込み","鈴木" 

"厨房","佐藤" 

"厨房","鈴木" 

"接客","安藤" 

"接客","山本" 

"掃除","安藤" 

"掃除","佐藤" 

"買出し","山本" 

"発注","鈴木" 

 

"ごみ捨て"のクオリティーは  1 

"レジ打ち"のクオリティーは  2 

"仕込み"のクオリティーは  1 

"厨房"のクオリティーは  1 

"接客"のクオリティーは  2 

"掃除"のクオリティーは  0 

"買出し"のクオリティーは  0 

"発注"のクオリティーは  1 

さて，この問題ですが通常に考えると先ほどの問題に以下の制約を付け加えることで解くこと

ができます．また記述する位置は x の宣言以降，solve() の手前であればどこでもかまいま

せん． 

 

上記記述を追加し，モデル（jobassign2.smp）を実行すると結果は以下のように変わりま

す． 

 

 

結果を見ると，総コストが多くなっていることが分かります． 

総コスト=13470 

"ごみ捨て","安藤" 

"レジ打ち","佐藤" 

"レジ打ち","渡辺" 

"仕込み","山本" 

"仕込み","鈴木" 

"厨房","安藤" 

"厨房","山本" 

"接客","佐藤" 

"接客","鈴木" 

"掃除","安藤" 

"掃除","佐藤" 

"買出し","山本" 

"発注","鈴木" 

 

"ごみ捨て"のクオリティーは  1 

"レジ打ち"のクオリティーは  2 

"仕込み"のクオリティーは  1 

"厨房"のクオリティーは  1 

"接客"のクオリティーは  1 

"掃除"のクオリティーは  0 

"買出し"のクオリティーは  0 

"発注"のクオリティーは  1 

x["接客,安藤"] == 0;  

x["厨房,佐藤"] == 0;  

x["レジ打ち,鈴木"] == 0;  

x["発注,山本"] == 0;  

x["ごみ捨て,渡辺"] == 0; 

以上の考え方は以下の図のように変数を準備し，色が付いているところが 0 であるという制

約を設けたと考えることができます．（なお図中のレジはレジ打ち，ごみはごみ捨て，買出は買

出し，仕込は仕込みです） 

 

  安藤 佐藤 鈴木 山本 渡辺 

接客 x[接客，安藤] x[接客，佐藤] x[接客，鈴木] x[接客，山本] x[接客，渡辺] 

厨房 x[厨房，安藤] x[厨房，佐藤] x[厨房，鈴木] x[厨房，山本] x[厨房，渡辺] 

レジ打ち x[レジ，安藤] x[レジ，佐藤] x[レジ，鈴木] x[レジ，山本] x[レジ，渡辺] 

発注 x[発注，安藤] x[発注，佐藤] x[発注，鈴木] x[発注，山本] x[発注，渡辺] 

ごみ捨て x[ごみ，安藤] x[ごみ，佐藤] x[ごみ，鈴木] x[ごみ，山本] x[ごみ，渡辺] 

買出し x[買出，安藤] x[買出，佐藤] x[買出，鈴木] x[買出，山本] x[買出，渡辺] 

掃除 x[掃除，安藤] x[掃除，佐藤] x[掃除，鈴木] x[掃除，山本] x[掃除，渡辺] 

仕込み x[仕込，安藤] x[仕込，佐藤] x[仕込，鈴木] x[仕込，山本] x[仕込，渡辺] 

 

ここで，上の図の色のついている部分は 1 になることがないので，はじめから変数を準備す

る必要がないと考えることが出来ます．特に大規模の割当問題においては，無駄な変数を準備す

ることにより不用意に問題の規模を大きくしてしまうと，計算のパフォーマンスを著しく低下さ

せてしまう要因になります．ここでは明らかな制約に関しては「制約を付け加えることなく，そ

のような選択肢を元々準備しない」という方法を紹介します．以下の図のようなイメージです． 

 

  安藤 佐藤 鈴木 山本 渡辺 

接客  x[接客，佐藤] x[接客，鈴木] x[接客，山本] x[接客，渡辺] 

厨房 x[厨房，安藤]  x[厨房，鈴木] x[厨房，山本] x[厨房，渡辺] 

レジ打ち x[レジ，安藤] x[レジ，佐藤]  x[レジ，山本] x[レジ，渡辺] 

発注 x[発注，安藤] x[発注，佐藤] x[発注，鈴木]  x[発注，渡辺] 

ごみ捨て x[ごみ，安藤] x[ごみ，佐藤] x[ごみ，鈴木] x[ごみ，山本]  

買出し x[買出，安藤] x[買出，佐藤] x[買出，鈴木] x[買出，山本] x[買出，渡辺] 

掃除 x[掃除，安藤] x[掃除，佐藤] x[掃除，鈴木] x[掃除，山本] x[掃除，渡辺] 

仕込み x[仕込，安藤] x[仕込，佐藤] x[仕込，鈴木] x[仕込，山本] x[仕込，渡辺] 

 

定式化する際に，上記を表す集合 JPPair を準備する必要があります． 

 ( , )j p JPPair j p   を に割り当てることができる 

以下が JPPair を用いた定式化です． 



 99 

 

集合 
JOB ｛接客，厨房，レジ打ち，発注，ごみ捨て，買出し，掃除，仕込み｝ 

仕事の集合， j JOB  

 
PEOPLE ｛安藤，佐藤，鈴木，山本，渡辺｝ 

人の集合， p PEOPLE  

 ( , )j p JPPair j p   を に割り当てることができる 

   

0-1 変数 , ( , )jpx j p JPPair  
仕事 j を人 p に割り当てるならば

1jpx  ，そうでないならば 0jpx  とする 

定数 , ( , )jpcost j p JPPair  仕事 j を人 p に割り当てる際のコスト 

 , ( , )jpjyukuren j p JPPair  仕事 j を人 p が行う際の熟練度 

   

目的関数 

（最小化） , ,( , )

jp jp

j p j p JPPair

cost x


  
コストの総和 

   

制約条件 
,( , )

2jp

p j p JPPair

x


 , j ｛接客，厨房，レジ打ち，掃除，仕込み｝ 

 「接客」「厨房」「レジ打ち」「掃除」「仕込み」は２人を割り当てる． 

 
,( , )

1,jp

p j p JPPair

x j


   ｛発注，ごみ捨て，買出し｝ 

「発注」「ごみ捨て」「買出し」は１人を割り当てる 

 
,( , )

3jp

j j p JPPair

x


  

各人には，最大３つまで仕事を割り当てることができる 

 
,( , )

0jp jp

p j p JPPair

jyukuren x


   

各仕事のクオリティーが負になってはいけない 

 
( , ), { , },( , )

1jp

j p j j p JPPair

x
 


接客 厨房

 

接客，厨房は違う人が担当する． 

モデルは以下のようになります． 

 

 

上記記述内の JPPair には無駄な [j,p] のペアを入れてはいけないので，入力ファイルも

それに伴い以下のように変わります． 

jobassign3.smp 

Set Job; 

Set People; 

Element j(set = Job); 

Element p(set = People); 

 

Set JPPair(dim = 2,superSet = (Job,People)); 

IntegerVariable x(type = binary, index = JPPair); 

Parameter cost(index = JPPair, name = "コスト"); 

Parameter jyukuren(index = JPPair, name = "熟練度"); 

 

sum(x[j,p],(p,(j,p)<JPPair)) == 2, 

  j == "接客" || j == "厨房" || j == "レジ打ち" || 

  j == "掃除" || j == "仕込み"; 

sum(x[j,p],(p,(j,p)<JPPair)) == 1, 

  j == "発注" || j == "ごみ捨て" || j == "買出し" ; 

sum(x[j,p],(j,(j,p)<JPPair)) <= 3; 

sum(x[j,p]*jyukuren[j,p],(p,(j,p)<JPPair)) >= 0; 

sum(x[j,p],(j,(j,p)<JPPair,j == "接客" || j == "厨房")) <= 1; 

 

Objective total_cost(type = minimize, name = "総コスト"); 

total_cost = sum(cost[j,p]*x[j,p],(j,p,(j,p)<JPPair)); 

 

// 以下出力用プログラム 

solve(); 

total_cost.val.print(); 

x.val.print(); 

jp.csv 

j,p,熟練度,コスト 

接客,佐藤,3,1400 

接客,鈴木,-2,520 

接客,山本,3,1410 

接客,渡辺,-4,450 

厨房,安藤,5,1800 

厨房,鈴木,3,1700 

厨房,山本,-4,1050 

厨房,渡辺,5,2300 

レジ打ち,安藤,0,800 

レジ打ち,佐藤,3,1500 

レジ打ち,山本,3,1500 

レジ打ち,渡辺,-1,600 

発注,安藤,-3,500 

発注,佐藤,-1,600 

発注,鈴木,1,1000 

発注,渡辺,2,1200 

ごみ捨て,安藤,1,800 

ごみ捨て,佐藤,-1,600 

ごみ捨て,鈴木,3,1200 

ごみ捨て,山本,1,800 

買出し,安藤,-1,600 

買出し,佐藤,-1,600 

買出し,鈴木,1,800 

買出し,山本,0,700 

買出し,渡辺,2,1300 

掃除,安藤,2,1200 

掃除,佐藤,-2,500 

掃除,鈴木,2,1200 

掃除,山本,-3,500 

掃除,渡辺,4,1300 

仕込み,安藤,5,1500 

仕込み,佐藤,-2,1000 

仕込み,鈴木,0,1200 

仕込み,山本,1,1200 

仕込み,渡辺,5,1500 

またこのモデルを制約充足問題ソルバ wcsp を用いる場合にはモデルを以下のように変更す

る必要があります． 

 

大規模な割当問題を解く際の問題規模に対する対応として上記で紹介した手法を試してみて

下さい． 

jobassign4.smp 

options.method = "wcsp";// 解法に wcsp を用いる指定 

options.maxtim = 10;// 計算時間の設定 

 

Set Job; 

Set People; 

Element j(set = Job); 

Element p(set = People); 

 

Set JPPair(dim = 2,superSet = (Job,People)); 

IntegerVariable x(type = binary,index = JPPair); 

Parameter cost(index = JPPair,name = "コスト"); 

Parameter jyukuren(index = JPPair,name = "熟練度"); 

 

sum(x[j,p],(p,(j,p)<JPPair)) == 2, 

  j == "接客" || j == "厨房" || 

  j == "レジ打ち" || j == "掃除" || j == "仕込み"; 

 

// wcsp 特有の関数 selection  

selection(x[j,p],(p,(j,p)<JPPair)), 

  j == "発注" || j == "ごみ捨て" || j == "買出し"; 

   

sum(x[j,p],(j,(j,p)<JPPair)) <= 3; 

sum(x[j,p]*jyukuren[j,p],(p,(j,p)<JPPair)) >= 0; 

sum(x[j,p],(j,(j,p)<JPPair,j == "接客" || j == "厨房")) <= 1; 

 

Objective total_cost(type = minimize,name = "総コスト"); 

total_cost = sum(cost[j,p]*x[j,p],(j,p,(j,p)<JPPair)); 

 

// 以下出力用プログラム 

solve(); 

total_cost.val.print(); 

x.val.print(); 

2.13.4  施設配置問題 

ここでは，どこに施設を配置するかを決定する問題を取り上げます．次の例題を考えます． 

 

（例題） 施設を候補地のどこに建てるかを決定します．施設として緊急時の収容所を考

えます．つまり，火事での火傷や脳卒中の患者の受け入れ等を考慮するため，エリアの

合計人数を考慮するのではなく，各エリアの対象予想数をカバーできればよく，例えば

施設 1 がエリア A，B，C を担当するときには，施設 1 の容量に関する制約は A，B，

C の合計人数ではなく，A，B，C エリアのそれぞれの人数をカバーできればよいという

ことになります．これは，異なるエリアで同時に事象が起こりえないモデルとして考え

ることを意味します．問題を以下に整理します． 

 

 緊急時の収容を考慮する（異なる地域を同時に収容しない） 

 各施設は受け入れ容量を持っている 

 各地域は最寄の施設に対応づけられる 

 容量をオーバーする場合は距離に比例するコストを支払えば 3 人まで超過すること

ができる 

 施設を建設するにはコストがかかる 

 施設の建設候補地は 7 地点である 

 地域は 5 × 5 の 25 地点である 

 

このとき，容量超過コストと施設建設コストの和を最小化するような建設場所を決定し

たいとします． 

以上を踏まえて，以下のように定式化することができます．但し M は大きな定数とします． 

 

集合 
I ｛1 ... 7 ｝ 

施設の集合 

 
J ｛1 ... 25｝ 

地域の集合 

   

0-1 変数 ,ix i I  
施設 i を建設するならば 1ix  ，そうで

ないならば 0ix  とする 

離散変数 ,jy j J  
エリア j に対しての超過人数．0 ... 3 

の値をとる 

定数 , ,ijdist i I j J   
施設 i から地域 j までの距離．今回は座

標を与え直線距離をモデル内で求める 

 jnum  地域 j で発生する緊急患者数 

 
icapa  施設 i が受け入れることのできる人数 

 
icost  施設 i の建設コスト 

   

目的関数 

（最小化） 

i i

i

cost x   

  min 1 ,j ij i

j

y dist M x i   

コストの総和 

   

制約条件 下記参照 

  

 

制約式は「各地域は最寄の施設に収容される」ということと，「施設の収容には容量があり，

3 人までは容量の超過が認められる」ということです．これは WCSP 特有の関数である 

argmin 関数を用いることによって１つの制約式で表現することができます．これは実際にモ

デルで確認をして下さい． 

今回は定数に関してはファイルからデータを与えてみます．モデル部分は以下のようになりま

す． 

InstitutionSetting1.smp  

// 施設の集合 

  Set I; 

  Element i(set = I); 

  // 地域の集合 

  Set J; 

  Element j(set = J); 

  IntegerVariable x(index = i,type = binary,name = "x"); 

 

  Set S = "0 ... 3"; // 離散変数用の集合 

  DiscreteVariable y(dom = S,index = j,name = "y"); 

 

  // dist はモデル内で計算する 

  Parameter dist(index = (i,j),name = "dist"); 

 

  // 実際には以下の座標を入力する 

  Parameter pos_xi(index = i,name = "pos_xi"); 

  Parameter pos_xj(index = j,name = "pos_xj"); 

  Parameter pos_yi(index = i,name = "pos_yi"); 

  Parameter pos_yj(index = j,name = "pos_yj"); 

  // 距離は直線距離を考える 

  dist[i,j] = sqrt(pow((pos_xi[i] - pos_xj[j]),2) 

     +pow((pos_yi[i] - pos_yj[j]),2)); 

 

  Parameter num(index = j,name = "num"); 

  Parameter n(index = (i,j),name = "n"); 

  Parameter capa(index = i,name = "capa"); 

  Parameter cost(index = i,name = "cost"); 

  Parameter M; 

  M = 100; 

 

// 施設の容量超過分のための計算 

  n[i,j] = capa[i] - num[j]; 

   

  Objective total_cost(type = minimize); 

  total_cost = sum(cost[i]*x[i],i) +  

sum(y[j]*min(dist[i,j]+ M*(1 - x[i]),i),j); 

 

// 制約式，超過人数 y についての記述 

  n[argmin((x[i] - 1)*M + dist[i,j],i),j] + y[j] >= 0; 

 

  options.maxtim = 5; 

  options.method = “wcsp”; 

  solve(); 

以下が csv 形式の入力ファイルです．施設に関してと地域に関しての二種類のファイルを用

意する必要があります． 

 

 

 

 

num.csv 

j,pos_xj,pos_yj,num 

1,0,0,3 

2,0,5,6 

3,0,10,1 

4,0,15,7 

5,0,20,3 

6,5,0,5 

7,5,5,8 

8,5,10,12 

9,5,15,3 

10,5,20,4 

11,10,0,5 

12,10,5,9 

13,10,10,2 

14,10,15,1 

15,10,20,11 

16,15,0,3 

17,15,5,4 

18,15,10,5 

19,15,15,9 

20,15,20,1 

21,20,0,5 

22,20,5,7 

23,20,10,8 

24,20,15,2 

25,20,20,11 

capa_cost.csv 

i,pos_xi,pos_yi,capa,cost 

1,13,18,6,24 

2,14,4,9,20 

3,19,15,7,41 

4,3,10,5,13 

5,18,11,7,23 

6,7,5,10,18 

7,6,1,7,20 

このように，WCSP 特有の関数を用いると煩雑な制約式でも簡単に記述できる場合があります．

次にこの問題で「最寄の施設まで 10.5 以上距離がある地域は 10 地域までしか許さない」と

いう制約を記述することを考えてみます．これは WCSP 用の count 関数を用いれば簡単に記

述することができます．この場合どの施設がどのエリアを担当するという変数 z を加えて以下

のように記述することができます．(InstitutionSetting2.smp参照) 

 

ここで semiHardConstraint を用いたのは万が一 10 地域以内でおさまらなかった場合

に「なるべく 10 地域以内で」ということを表現するためです． 

  IntegerVariable z(index = (i,j),type = binary); 

  selection(z[i,j],i); 

  Element ii(set = I); 

  sum(dist[i,j]*z[i,j],i) <= min(dist[ii,j]+(1 - x[ii])*M,ii); 

  semiHardConstraint(); 

  count( 10.5 <= z[i,j]*dist[i,j],(i,j)) <= 10; 

2.14 二次割当問題（SAMPLES\例題集 2.prj 参照） 

二次割当問題とは，1957 年 Koopmans と Beckmann により考案された，目的関数が二次

式となる割当問題です．NUOPT V13 より導入された Vector/Matrix クラスの表記を用いて，

次の例題を考えてみます． 

 

この問題は，工場間のフロー行列を ( )i jF f ，地区間の距離行列を ( )k lD d  とすると， 

( ) ( )min
n

i j i j
S

i j

f d 

   (1)  

と表現されます．ここで， nS  は n  次の置換群 ( この問題では 5n   ) です． 

 ここで，置換 nS  に対し，次のような変数 i kx   

1 if ( )

0 otherwise.
i k

i k
x

 
 


 

を成分に持つn n 行列 X を用いることで， (1) における ( ) ( )i jd  は 

T

( ) ( )( )i jd XDX    

と表すことができます． 

（例題） 

工場 1，..，5 を地区 I，..，V に配置することを考えます．ただし，各工場間で

の物資を輸送する量，頻度などのフロー量，および各地区間の距離が，以下のように

与えられているとします． 

 

各工場のフロー 各地域間の距離 

 1 2 3 4 5  I II III IV V 

1 0 15 12 8 7 I 0 4 7 6 10 

2 15 0 14 2 4 II 4 0 2 5 4 

3 12 14 0 6 6 III 7 2 0 8 7 

4 8 2 6 0 11 IV 6 5 8 0 12 

5 7 4 6 11 0 V 10 4 7 12 0 

 

物資の輸送コストを（フロー）×（距離）で与えるとき，最もコストがかからない配

置を求めなさい． 

よって (1) は，次のような整数計画問題に書き換えることができます． 

  
   

,

m i n T

i j i j
i j

F XDX f d
 

 
 
 
  

s.t. 1 ,  1,2,..., ,

1 ,  1,2,..., ,

{0,1},  , 1,2,..., .

ik

i

ik

k

ik

x k n

x i n

x i k n

 

 

 





 

 

以上より，定式化は次のようになります．  

   

集合 
Fac  工場の集合 

Loc 地区の集合 

定数 

, ,i jF i j Fac  各工場間のフロー 

, ,m nD m n Loc  各地区間の距離 

   

変数 , ,i mX i Fac m Loc   
工場 i を地区m に配置する場合 1 

それ以外は 0 

   

目的関数 (最小化) 
T( )F XDX  は要素ごとの積の和 

   

制約条件 T

Xe e

X e e



  e は全成分 1のベクトル 

定式化した結果を SIMPLEで記述すると以下のようになります． 

 

 

V12 までの SIMPLE では，この目的関数を 

 

と記述します．Vector/Matrix クラスにより，より簡潔に記述できることがわかります． 

QAP.smp 

// 集合と添字 

Set Fac,Loc; 

Element i(set=Fac),j(set=Fac),m(set=Loc),n(set=Loc); 

 

// パラメータ 

Matrix F((i,j)),D((m,n)); 

 

// 変数 

IntegerVariable x(index=(i,m),type=binary); 

Matrix X((i,m)); 

X[i,m] = x[i,m];// 変数を行列に詰める 

 

// 目的関数 

Objective total_cost(type=minimize); 

total_cost = inprod(F,X*D*trans(X)); 

 

// 制約条件 

X*ones(Loc) == ones(Fac); 

trans(X)*ones(Fac)== ones(Loc); 

 

options.method = “wcsp”; 

options.maxtim = 10; 

 

solve(); 

simple_printf(“工場 %d を地区 %s に配置する¥n”,i,m,X[i,m].val == 1); 

total_cost = sum(f[i,j]*d[k,l]*x[i,k]*x[j,l],(i,j,k,l)); 

 以下が csv 形式の入力ファイルです． 

 

  

このモデルを実行すると次のような結果が得られます． 

 

F.csv 

F,1,2,3,4,5 

1,0,15,12,8,7 

2,15,0,14,2,4 

3,12,14,0,6,6 

4,8,2,6,0,11 

5,7,4,6,11,0 

D.csv 

D,I,II,III,IV,V 

I,0,4,7,6,10 

II,4,0,2,5,4 

III,7,2,0,8,7 

IV,6,5,8,0,12 

V,10,4,7,12,0 

工場 1を地区 II に配置する 

工場 2を地区 V に配置する 

工場 3を地区 III に配置する 

工場 4を地区 IV に配置する 

工場 5を地区 I に配置する 

2.15 設備計画問題（SAMPLES¥例題集 2.prj 参照） 

設備計画問題として以下の例題では，電力を購入して生産を行っている工場が，自家発電用の

発電機を導入する問題を考えます． 

 

 

この問題の定式化は，以下のようになります． 

（例題） 

電力を購入して生産を行っているある工場が，自家発電用の発電機の導入を計画してい

ます．ただしその際，購入したスポット電力と自家発電の電力の和が，想定されるすべて

の時刻で電力需要を上回らなければなりません．なお，購入を想定するスポット電力商品

は次の 3 つとします．スポット電力ですので，時刻毎に購入する商品を変更することが

可能です．ただし，各時刻において必ず１つの商品を選択して購入しなければなりません． 

 

商品 1 2 3 

出力電力(kWh) 5 10 20 

単位時間あたりのコスト 10 20 30 

  

また，新たに導入する候補となる発電機は，以下の 6 種類とします．発電機を導入し

ますと，以下のような定常的な出力が得られるものとします． 

 

発電機 1 2 3 4 5 6 

出力電力(kWh) 5 6 8 10 15 20 

導入コスト 100 120 150 160 280 300 

 

時刻 t = 1,…,24 について電力需要予想 tD  は以下のように与えられているものと

します．スポット電力購入コストと発電機導入コストの和を最小化するような，発電機の

導入方法およびスポット電力の購入方法を求めて下さい．  

 

t 1 2 3 4 5 6 7 8 9 10 11 12 

tD   10 10 12 12 15 20 30 30 35 40 50 56 

t 13 14 15 16 17 18 19 20 21 22 23 24 

tD   52 47 40 35 42 50 48 40 30 20 15 12 

集合 Product  商品集合 

 Dynamo  発電機集合 

 Time 時刻集合 

   

定数 icostS i Product,　  
各商品に対する単位時間

あたりのコスト 

 jcostP j Dynamo,　  
各発電機に対する導入コ

スト 

 
ip i Product,　  

各商品の単位時間あたり

の出力電力 

 jq j Dynamo,　  
各発電機の単位時間あた

りの出力電力 

 
tD t Time,　  時刻ごとの電力需要 

   

0-1整数変数 
,iu i Product t Time t,　 　

 

各時刻ごとに各商品につ

いて，利用するならば 1，

利用しないならば 0 

 jx j Dynamo,　  

各発電機に関して，導入す

るならば 1，しないならば

0 

   

目的関数 (最小化) 

 

 

,

t

i i j j

i t j

costS u costP x   

 

総コスト（スポット電力購入コ

ストと発電機導入コストの和）

の最小化 

   

制約条件 
t

i

i

u t Time  =1,  
各時刻における商品の使

用は 1つのみ 

 

max{ }tt i i
t

i

j j

j

D p u

q x

 

 



  

すべての時刻において，購

入したスポット電力と自

家発電の出力電力の和が

電力需要を上回らないと

いけない 

定式化のポイントとしましては，すべての時刻における電力の総和と電力需要に関する制約式

を，時刻ごとに設けるのでなく，時刻に関する max 関数を用いて表現する点にあります．その

ため制約充足問題ソルバ wcspを用いて解くべく，SIMPLEでは以下のように記述します．なお，

max関数に関しましては，NUOPT_SIMPLE_マニュアルの記述もあわせてご覧下さい． 

 

 

FPP.smp 

// 商品集合 

Set Product; 

Element i(set=Product); 

// 時刻集合 

Set Time; 

Element t(set=Time); 

// 発電機集合 

Set Dynamo; 

Element j(set=Dynamo); 

 

// コスト 

// 各商品に対する単位時間あたりのコスト 

Parameter costS(name="costS", index=i); 

// 各発電機に対する導入コスト 

Parameter costP(name="costP", index=j); 

 

// 出力電力 

// 各商品の単位時間あたりの出力電力 

Parameter p(name="p", index=i); 

// 各発電機の単位時間あたりの出力電力 

Parameter q(name="q", index=j); 

 

// 時刻ごとの電力需要 

Parameter D(name="D", index=t); 

 

// 変数 

// 各時間ごとに，どの商品を利用するか 

IntegerVariable u(name=”商品”, index=(i,t), type=binary); 

// どの発電機を導入するか 

IntegerVariable x(name=”発電機”, index=j, type=binary); 

データファイル（dat 形式）は以下のようになります． 

 

data.dat 

costS = [1] 10 [2] 20 [3] 30; 

costP = [1] 100 [2] 120 [3] 150 [4] 160 [5] 280 [6] 300; 

 

p = [1] 5 [2] 10 [3] 20; 

q = [1] 5 [2] 6 [3] 8 [4] 10 [5] 15 [6] 20; 

 

D = [1] 10 [2] 10 [3] 12 [4] 12 [5] 15 [6] 20 [7] 30 [8] 30  

[9] 35 [10] 40 [11] 50 [12] 56 [13] 52 [14] 47 [15] 40 [16] 35  

[17] 42 [18] 50 [19] 48 [20] 40 [21] 30 [22] 20 [23] 15 [24] 12; 

// 目的関数（総コストの最小化） 

Objective cost(name="総コスト", type=minimize); 

cost = sum(costS[i]*u[i,t], (i,t)) + sum(costP[j]*x[j], j); 

 

// 制約式 

// 各時間における商品の使用は 1 つのみ 

selection(u[i,t], i); 

// すべての時刻において，購入したスポット電力と自家発電の出力電力の和が 

// 電力需要を上回らないといけない 

max(D[t] - sum(p[i]*u[i,t], i), t) <= sum(q[j]*x[j], j); 

 

// 最大求解時間の設定 

options.maxtim = 5; 

 

// 求解 

solve(); 

 

// 結果の出力 

cost.val.print(); 

x.val.print(); 

u.val.print(); 

 



 116 

 

このモデルを実行すると，以下のような解が得られ，発電機の導入方法および電力の購入方法

が求まりました． 

 

 

 

 

商品[1,1]=1 

商品[1,2]=1 

商品[1,3]=1 

商品[1,4]=1 

商品[1,5]=1 

商品[1,6]=1 

商品[1,7]=1 

商品[1,8]=1 

商品[1,9]=1 

商品[1,10]=1 

商品[1,11]=0 

商品[1,12]=0 

商品[1,13]=0 

商品[1,14]=0 

商品[1,15]=1 

商品[1,16]=1 

商品[1,17]=1 

商品[1,18]=0 

商品[1,19]=0 

商品[1,20]=1 

商品[1,21]=1 

商品[1,22]=1 

商品[1,23]=1 

商品[1,24]=1 

商品[2,1]=0 

商品[2,2]=0 

商品[2,3]=0 

商品[2,4]=0 

商品[2,5]=0 

商品[2,6]=0 

商品[2,7]=0 

商品[2,8]=0 

商品[2,9]=0 

商品[2,10]=0 

商品[2,11]=0 

商品[2,12]=0 

商品[2,13]=0 

商品[2,14]=1 

商品[2,15]=0 

商品[2,16]=0 

商品[2,17]=0 

商品[2,18]=0 

商品[2,19]=1 

商品[2,20]=0 

商品[2,21]=0 

商品[2,22]=0 

商品[2,23]=0 

商品[2,24]=0 

商品[3,1]=0 

商品[3,2]=0 

商品[3,3]=0 

商品[3,4]=0 

商品[3,5]=0 

商品[3,6]=0 

商品[3,7]=0 

商品[3,8]=0 

商品[3,9]=0 

商品[3,10]=0 

商品[3,11]=1 

商品[3,12]=1 

商品[3,13]=1 

商品[3,14]=0 

商品[3,15]=0 

商品[3,16]=0 

商品[3,17]=0 

商品[3,18]=1 

商品[3,19]=0 

商品[3,20]=0 

商品[3,21]=0 

商品[3,22]=0 

商品[3,23]=0 

商品[3,24]=0 

総コスト=950 

発電機[1]=0 

発電機[2]=0 

発電機[3]=1 

発電機[4]=1 

発電機[5]=0 

発電機[6]=1 

2.16 最小二乗問題（SAMPLES¥例題集 2.prj 参照） 

最小二乗問題は，科学や工学の分野でよく利用される基本的な問題です．この問題の典型的な

例としては，曲線の当てはめがあります．実験により得られたデータをモデル曲線に当てはめ定

数を推定していくことを考えます．すると，一般に各観測点においてデータとモデルの間には誤

差が発生してしまいます．この時，「各観測点における誤差の二乗和を最小にする」という基準

を採用しなるべく良い曲線に当てはめようとすると最小二乗問題となります． 

 

 

この例題において，変数となるのは推定する a,b,c の 3 つです．また，ある計測時刻 t での

データ値 ( )y t とモデルから得られる値 ( )f t との誤差は
2 ( )at bt c y t   となります．よって，

各計測時刻について誤差の二乗を求め，その総和をとることで最小化する目的関数が得られます． 

（例題） 

Aさんは，実験開始から t秒後の yという値を計測するという実験を行いました．その

結果をまとめると，下の表のようになりました．A さんはこの結果をプロットしたとこ

ろ，
2( )f t at bt c   という二次関数モデルに当てはめられることに気付きました．

そこで，各計測時刻でのデータ値とモデルから得られる値との誤差の二乗和が最小にな

るように a,b,cを推定して下さい． 

 

t y  t y 

0.5 1.22470  3.0 3.22933 

1.0 0.87334  3.5 4.44744 

1.5 0.99577  4.0 6.13509 

2.0 1.34215  4.5 8.14697 

2.5 2.12172  5.0 10.50759 

以上のことから，この例題は次のように定式化できます． 

 

変数 a 2次の項の係数 

 b 1次の項の係数 

 c 定数項 

   

目的関数 

（最小化） 

2 2 2 2

2 2 2 2

2 2 2 2

2 2 2 2

2

(0.5 0.5 1.22470) (1.0 1.0 0.87334)

(1.5 1.5 0.99577) (2.0 2.0 1.34215)

(2.5 2.5 2.12172) (3.0 3.0 3.22933)

(3.5 3.5 4.44744) (4.0 4.0 6.13509)

(4.5 4

a b c a b c

a b c a b c

a b c a b c

a b c a b c

a

      

       

       

       

  2 2 2.5 8.14697) (5.0 5.0 10.50759)b c a b c       

 各計測時刻における誤差の二乗和 

 

ここで，この結果を SIMPLEで何も工夫せずに記述すると次のようになります． 

 

leastsquare1.smp 

//変数の宣言 

Variable a(name="２次の項の係数"); 

Variable b(name="１次の項の係数"); 

Variable c(name="定数項"); 

//誤差の二乗和の最小化 

Objective err(type=minimize); 

err=(0.5*0.5*a+0.5*b+c-1.2247)*(0.5*0.5*a+0.5*b+c-1.2247) 

+(1.0*1.0*a+1.0*b+c-0.87334)*(1.0*1.0*a+1.0*b+c-0.87334) 

+(1.5*1.5*a+1.5*b+c-0.99577)*(1.5*1.5*a+1.5*b+c-0.99577) 

+(2.0*2.0*a+2.0*b+c-1.34215)*(2.0*2.0*a+2.0*b+c-1.34215) 

+(2.5*2.5*a+2.5*b+c-2.12172)*(2.5*2.5*a+2.5*b+c-2.12172) 

+(3.0*3.0*a+3.0*b+c-3.22933)*(3.0*3.0*a+3.0*b+c-3.22933) 

+(3.5*3.5*a+3.5*b+c-4.44744)*(3.5*3.5*a+3.5*b+c-4.44744) 

+(4.0*4.0*a+4.0*b+c-6.13509)*(4.0*4.0*a+4.0*b+c-6.13509) 

+(4.5*4.5*a+4.5*b+c-8.14697)*(4.5*4.5*a+4.5*b+c-8.14697) 

+(5.0*5.0*a+5.0*b+c-10.50759)*(5.0*5.0*a+5.0*b+c-10.50759); 

//求解し結果を表示する 

solve(); 

a.val.print(); 

b.val.print(); 

c.val.print(); 

これでは，何回も実験を行うような場合モデルファイルの修正に大変な手間がかかってしまい

効率的ではありません．そこで，SIMPLEの特長を生かし様々な測定結果に対応できるように定

式化から見直していきます． 

まず，ある計測時刻 tでのデータ値 ( )y t とモデルから得られる値 ( )f t との誤差を ( )e t と表す

ことにします．すると，先ほどの議論から
2( ) ( )e t at bt c y t    となります．これより，目

的関数は
2( )e t を tについて和をとったものと言えます．SIMPLEにおいては，Expressionを

用いることで数式を宣言できます．よって， ( )e t  を Expression で記述することでモデルフ

ァイルをより簡潔なものにできます．また，二乗和を求める部分についても観測時間の集合を導

入することで sum()を用い簡単に記述できます．最後に，各計測時刻 tにおける ( )y t の値は直

接モデルファイルに記述するのではなく csvファイルから与えることにします． 

以上のことから，次のように定式化しなおすことができます．  

 

集合 {0.5,1.0,1.5,2.0,2.5,3.0,3.5,4.0,4.5,5.0}ObserveTime   

  観測時間の集合 

   

変数 a 2次の項の係数 

 b 1次の項の係数 

 c 定数項 

   

定数 ( ),y t t ObserveTime  時刻 t における観測値 

   

目的関数 

（最小化） 

2 2 2( ) ( ( ))
t ObserveTime t ObserveTime

e t at bt c y t
 

      

 各観測時刻における誤差の二乗和 

この結果を SIMPLE で記述すると次のようになり，先ほどのものに比べ汎用性が高くさらに

見やすいものになっています．  

 

 

なお，実行させる際には次のような csvファイルを与えます． 

 

data_leastsquare2.csv 

t,y 

0.5,1.22470 

1.0,0.87334 

1.5,0.99577 

2.0,1.34215 

2.5,2.12172 

3.0,3.22933 

3.5,4.44744 

4.0,6.13509 

4.5,8.14697 

5.0,10.50759 

leastsquare2.smp 

//観測時間の集合の宣言 

Set ObserveTime; 

Element t(set=ObserveTime); 

//観測値をパラメータとして宣言 

Parameter y(index=t); 

//変数の宣言 

Variable a(name="２次の項の係数"); 

Variable b(name="１次の項の係数"); 

Variable c(name="定数項"); 

//誤差を Expressionを用いて表現する 

Expression e(index=t); 

e[t]=a*t*t+b*t+c-y[t]; 

//誤差の二乗和の最小化 

Objective err(type=minimize); 

err=sum(e[t]*e[t],t); 

//求解して結果を出力する 

solve(); 

a.val.print(); 

b.val.print(); 

c.val.print(); 

このモデルを実行すると，最後に 

 

 

という表示がされます．このことから，
2( ) 0.644402 -1.47656 1.76057f t t t  という二次関

数モデルが推定できたことになります． 

２次の項の係数=0.644402 

１次の項の係数=-1.47656 

定数項=1.76057 

2.17 ポートフォリオ最適化問題（SAMPLES¥例題集 2.prj参照）） 

ここでは，ポートフォリオが与える収益率の分布を評価する統計量として平均と分散を用いる

マルコビッツモデルの例を示します．具体的にはポートフォリオのもたらす収益率の変動の大き

さ（リスク）を分散で計測することにし，それを最小化するような資産配分を求めます． 

 

 

（例題） 

銘柄 1，2に対する収益率のサンプル 20期分が以下の図のように得られているとします．  

 

-0.1

-0.05

0

0.05

0.1

0.15

0.2

0.25

1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20

サンプル

収
益
率

銘柄１

銘柄２

 

 

この収益率のサンプルを用いてポートフォリオの収益率の分散が最小となる投資配分を

求めます．ただし，空売りはできないものとし，またポートフォリオの収益率の期待値

は 0以上とします． 

この問題を定式化すると以下のようになります． 

 

   

集合 {1,2}Asset   銘柄の集合 

 {1,2,3, ,20}Sample   サンプルの集合 

   

定数 , ,tjr t Sample j Asset   
サンプル t における銘柄 j の収

益率 

 

/

,

j tj

t Sample

r r Sample

j Asset









 

銘柄 j の平均収益率 

   

変数 ,jx j Asset  銘柄 j の組入比率 

目的関数 (最小化) 
2 /t

t Sample

dev Sample


  
ポートフォリオが与える期待収

益率の分散（リスク） 

 

,

t tj j

j Asset

dev r x rp

t Sample



 

 



 

サンプル t における平均からの

ぶれ 

 j j

j Asset

rp r x


   ポートフォリオの期待収益率 

   

制約条件 0,jx j Asset    非負制約（空売り禁止） 

 1j

j Asset

x


  組入比率の総和は 1 

 0rp   
ポートフォリオが与える期待収

益率は 0以上 

   

 

この問題は目的関数が二次であり制約式は全て線形ですので二次計画問題になります． 

定式化した結果を SIMPLEで記述すると以下のようになります． 

 

portfolio1.smp 

// 集合と添字 

Set Asset(name="銘柄"); 

Element j(set=Asset); 

Set Sample(name="サンプル"); 

Element t(set=Sample); 

 

// パラメータ 

Parameter r(name="収益率", index=(t,j)); 

Parameter rb(name="平均収益率", index=j); 

rb[j] = sum(r[t,j],t)/Sample.card();// Sample.card()：集合 Sampleの

要素数 

 

// 変数 

Variable x(name="組入比率", index=j); 

 

// 式 

Expression rpb(name="期待収益率"); 

rpb = sum(rb[j]*x[j],j); 

Expression dev(name="偏差", index=t); 

dev[t] = sum(r[t,j]*x[j],j)-rpb; 

 

// 目的関数 

Objective V(name="リスク"); 

V = sum(dev[t]*dev[t],t)/Sample.card(); 

 

// 制約 

x[j] >= 0.0; 

sum(x[j],j) == 1.0; 

rpb >= 0.0; 

 

// 求解 

solve(); 

 

// 出力 

V.val.print(); 

rpb.val.print(); 

x.val.print(); 



 125 

 

 

データファイル（.dat形式）は以下のようになります． 

 

 

このモデルを実行すると以下のような解が得られます． 

 

リスク=0.000951734 

期待収益率=0.0199126 

組入比率[1]=0.800874 

組入比率[2]=0.199126 

data.dat 

収益率 = 

[ 1,1]  0.0000 [ 1,2] 0.100 

[ 2,1]  0.0162 [ 2,2] 0.161 

[ 3,1]  0.0307 [ 3,2] 0.197 

[ 4,1]  0.0419 [ 4,2] 0.192 

[ 5,1]  0.0485 [ 5,2] 0.148 

[ 6,1]  0.0498 [ 6,2] 0.084 

[ 7,1]  0.0458 [ 7,2] 0.026 

[ 8,1]  0.0368 [ 8,2] 0.000 

[ 9,1]  0.0238 [ 9,2] 0.016 

[10,1]  0.0082 [10,2] 0.068 

[11,1] -0.0082 [11,2] 0.132 

[12,1] -0.0238 [12,2] 0.184 

[13,1] -0.0368 [13,2] 0.200 

[14,1] -0.0458 [14,2] 0.174 

[15,1] -0.0498 [15,2] 0.116 

[16,1] -0.0485 [16,2] 0.052 

[17,1] -0.0419 [17,2] 0.008 

[18,1] -0.0307 [18,2] 0.003 

[19,1] -0.0162 [19,2] 0.039 

[20,1]  0.0000 [20,2] 0.100 

; 

Vector/Matrix を使用した SIMPLE モデル 

NUOPT V13 より導入された，Vector/Matrix クラスの記述を使ったモデルを紹介します． 

 

実行時には，同じデータファイルを与えます．

portfolio2.smp 

Set Asset(name=”銘柄”), Sample(name=”サンプル”); 

Element j(set=Asset), t(set=Sample); 

 

// パラメータ行列，ベクトル 

Matrix R(name=”収益率”,(Sample,Asset)),Rbar((Sample,Asset)); 

Vector r(Asset); 

r[j] = inprod(ones(Sample),R.col(j))/Sample.card(); 

Rbar[t,j] = R[t,j] – r[j]; 

 

// 変数，変数ベクトル 

Variable x(name=”組入比率”,index=j); 

Vector xv(Asset), 

dev(Sample);// 偏差ベクトル 

xv[j] = x[j];// 変数 x をベクトル xv に詰め込む 

dev = Rbar*xv; 

 

Expression rpb(name=”期待収益率”); 

rpb = inprod(r,xv); 

 

Objective V(name=”リスク”,type=minimize); 

V = inprod(dev,dev)/Sample.card(); 

 

xv >= 0.0; 

inprod(ones(Asset),xv) == 1.0; 

rpb >= 0.0; 

 

solve(); 

 

V.val.print(); 

rpb.val.print(); 

x.val.print(); 

2.18 ロジスティック回帰モデル（SAMPLES¥例題集 2.prj参照） 

本章では，ロジスティック回帰モデルによるパラメータ推定方法を，アヤメの識別問題を例に

紹介します． 

 

この問題の定式化は次のようになります． 

 

集合 I  サンプル集合 

 

 

 

J  説明変数集合 

定数 , , ,　i jd i I j J  計測データ 

 

 
,　ik i I  

アヤメの種類 

（0：Iris setosa， 

1：Iris versicolor） 

   

変数 

 

 

0a  

,　ja j J  
推定したいパラメータ 

 

,　ix i I （中間変数） 

0
 

  
 

i j ij

j

x a a d  
 

目的関数 (最大化)    log 1 exp  i i i

i

k x x  対数尤度 

   

（例題） 

同じアヤメ科の Iris setosaと Iris versicolorに関して，がくの長さ，がくの幅，

花弁の長さ，花弁の幅に関する各々50 個のデータが存在する．これらの 4 つを説明変数

として，Iris setosaと Iris versicolorを判別するようなロジスティック回帰モデ

ルを構築せよ． 

サンプル集合の要素 i I に対し，ロジスティック関数 ( )iL x は，以下のような式で表わすこ

とができます． 

exp( )
( )

1 exp( )




i
i

i

x
L x

x
 

ただし
ix は， 0 i j ij

j

x a a d で表わされるものとします． 

次に，ロジスティック関数を用いた 0, ja a  のパラメータ推定方法の説明を行います．対象デ

ータ全てに対する尤もらしさを考え，それが最大となるような 0, ja a  を求めます．即ち，目的

関数 f を 

 
1

( ) 1 ( )


 
i

i
kk

i i

i

f L x L x  

で定義し，これを最大化する問題を考えることにより， 0, ja a  の推定を行います． 

なお，このままでは目的関数の形状が複雑なため，目的関数 f に対して対数をとった f  

  

 

  

log

log ( ) (1 ) log 1 ( )

exp( ) 1
log (1 ) log

1 exp( ) 1 exp( )

1
log exp( ) log

1 exp( )

log 1 exp( )

i i i i

i

i
i i

i i i

i i

i i

i i i

i

f f

k L x k L x

x
k k

x x

k x
x

k x x



   

 
   

  

 
  

 

  









　

  

　　

　　

　　  

で目的関数を置きなおしても一般性は失われない性質を利用して，対象となる問題の置き換え

を行います． 

以上が定式化の説明となります． 

次に，アヤメの計測データですが，これは [5] の文献にあるデータ（Iris setosa，Iris 

versicolor，各々50個）を使用します． 

具体的には，以下のような 2種類の csvファイルを用意します． 

（csv形式ファイル；data.csv） 

                       （左段の続き）            （中段の続き） 

 

計測データ ,がくの長

さ,がくの幅,花弁の長

さ,花弁の幅 

1,5.1,3.5,1.4,0.2 

2,4.9,3,1.4,0.2 

3,4.7,3.2,1.3,0.2 

4,4.6,3.1,1.5,0.2 

5,5,3.6,1.4,0.2 

6,5.4,3.9,1.7,0.4 

7,4.6,3.4,1.4,0.3 

8,5,3.4,1.5,0.2 

9,4.4,2.9,1.4,0.2 

10,4.9,3.1,1.5,0.1 

11,5.4,3.7,1.5,0.2 

12,4.8,3.4,1.6,0.2 

13,4.8,3,1.4,0.1 

14,4.3,3,1.1,0.1 

15,5.8,4,1.2,0.2 

16,5.7,4.4,1.5,0.4 

17,5.4,3.9,1.3,0.4 

18,5.1,3.5,1.4,0.3 

19,5.7,3.8,1.7,0.3 

20,5.1,3.8,1.5,0.3 

21,5.4,3.4,1.7,0.2 

22,5.1,3.7,1.5,0.4 

23,4.6,3.6,1,0.2 

24,5.1,3.3,1.7,0.5 

25,4.8,3.4,1.9,0.2 

26,5,3,1.6,0.2 

27,5,3.4,1.6,0.4 

28,5.2,3.5,1.5,0.2 

29,5.2,3.4,1.4,0.2 

30,4.7,3.2,1.6,0.2 

31,4.8,3.1,1.6,0.2 

32,5.4,3.4,1.5,0.4 

33,5.2,4.1,1.5,0.1 

34,5.5,4.2,1.4,0.2 

35,4.9,3.1,1.5,0.2 

36,5,3.2,1.2,0.2 

37,5.5,3.5,1.3,0.2 

38,4.9,3.6,1.4,0.1 

39,4.4,3,1.3,0.2 

40,5.1,3.4,1.5,0.2 

41,5,3.5,1.3,0.3 

42,4.5,2.3,1.3,0.3 

43,4.4,3.2,1.3,0.2 

44,5,3.5,1.6,0.6 

45,5.1,3.8,1.9,0.4 

46,4.8,3,1.4,0.3 

47,5.1,3.8,1.6,0.2 

48,4.6,3.2,1.4,0.2 

49,5.3,3.7,1.5,0.2 

50,5,3.3,1.4,0.2 

51,7,3.2,4.7,1.4 

52,6.4,3.2,4.5,1.5 

53,6.9,3.1,4.9,1.5 

54,5.5,2.3,4,1.3 

55,6.5,2.8,4.6,1.5 

56,5.7,2.8,4.5,1.3 

57,6.3,3.3,4.7,1.6 

58,4.9,2.4,3.3,1 

59,6.6,2.9,4.6,1.3 

60,5.2,2.7,3.9,1.4 

61,5,2,3.5,1 

62,5.9,3,4.2,1.5 

63,6,2.2,4,1 

64,6.1,2.9,4.7,1.4 

65,5.6,2.9,3.6,1.3 

66,6.7,3.1,4.4,1.4 

67,5.6,3,4.5,1.5 

68,5.8,2.7,4.1,1 

69,6.2,2.2,4.5,1.5 

70,5.6,2.5,3.9,1.1 

71,5.9,3.2,4.8,1.8 

72,6.1,2.8,4,1.3 

73,6.3,2.5,4.9,1.5 

74,6.1,2.8,4.7,1.2 

75,6.4,2.9,4.3,1.3 

76,6.6,3,4.4,1.4 

77,6.8,2.8,4.8,1.4 

78,6.7,3,5,1.7 

79,6,2.9,4.5,1.5 

80,5.7,2.6,3.5,1 

81,5.5,2.4,3.8,1.1 

82,5.5,2.4,3.7,1 

83,5.8,2.7,3.9,1.2 

84,6,2.7,5.1,1.6 

85,5.4,3,4.5,1.5 

86,6,3.4,4.5,1.6 

87,6.7,3.1,4.7,1.5 

88,6.3,2.3,4.4,1.3 

89,5.6,3,4.1,1.3 

90,5.5,2.5,4,1.3 

91,5.5,2.6,4.4,1.2 

92,6.1,3,4.6,1.4 

93,5.8,2.6,4,1.2 

94,5,2.3,3.3,1 

95,5.6,2.7,4.2,1.3 

96,5.7,3,4.2,1.2 

97,5.7,2.9,4.2,1.3 

98,6.2,2.9,4.3,1.3 

99,5.1,2.5,3,1.1 

100,5.7,2.8,4.1,1.3 

（csv形式ファイル；kind.csv）    （左段の続き）      （中段の続き） 

 

i,種類 

1,1 

2,1 

3,1 

4,1 

5,1 

6,1 

7,1 

8,1 

9,1 

10,1 

11,1 

12,1 

13,1 

14,1 

15,1 

16,1 

17,1 

18,1 

19,1 

20,1 

21,1 

22,1 

23,1 

24,1 

25,1 

26,1 

27,1 

28,1 

29,1 

30,1 

31,1 

32,1 

33,1 

34,1 

35,1 

36,1 

37,1 

38,1 

39,1 

40,1 

41,1 

42,1 

43,1 

44,1 

45,1 

46,1 

47,1 

48,1 

49,1 

50,1 

51,0 

52,0 

53,0 

54,0 

55,0 

56,0 

57,0 

58,0 

59,0 

60,0 

61,0 

62,0 

63,0 

64,0 

65,0 

66,0 

67,0 

68,0 

69,0 

70,0 

71,0 

72,0 

73,0 

74,0 

75,0 

76,0 

77,0 

78,0 

79,0 

80,0 

81,0 

82,0 

83,0 

84,0 

85,0 

86,0 

87,0 

88,0 

89,0 

90,0 

91,0 

92,0 

93,0 

94,0 

95,0 

96,0 

97,0 

98,0 

99,0 

100,0 

以上をもとに SIMPLEで記述すると，以下のようになります． 

 

このモデルを実行すると，以下のような解が得られます． 

 

LogisticRegression.smp 

// サンプル集合 

Set I(name="サンプル集合"); Element i(set=I); 

// 説明変数集合 

Set J(name="説明変数集合"); Element j(set=J); 

// データ 

Parameter d(name="計測データ", index=(i,j)); 

// アヤメの種類 

// 0 : Iris setosa, 1 : Iris versicolor 

Parameter k(name="種類", index=i); 

// 推定したいパラメータ 

Variable a(name="a", index=j); 

Variable a0(name="a0"); 

// 中間変数 

Variable x(name="x", index=i); 

x[i] == a0 + sum(a[j] * d[i,j], j); 

 

// 対数尤度（目的関数） 

Objective f(name=”対数尤度", type=maximize); 

f = sum(k[i]*x[i] - log(1+exp(x[i])), i); 

 

//求解 

solve(); 

 

// 結果の標準出力 

a0.val.print(); 

a.val.print(); 

a0=-3.80539 

a["がくの長さ"]=6.94731 

a["がくの幅"]=6.08239 

a["花弁の長さ"]=-14.0887 

a["花弁の幅"]=-19.5508 

2.19 イールドカーブ推定問題（SAMPLES¥例題集 3.prj 参照） 

イールドカーブとは，償還期間の異なる利回りをもつ債権等について，利回りと償還期間の相

関を描いた曲線（縦軸：利回り，横軸：償還期間）のことをいいます．ここでは，この利回りが

スポットレート（現時点から将来のある時点まで保有される資産にかかる金利のこと）である場

合を対象とします．なお，イールドカーブの形状には順イールド（右上がりの曲線）と逆イール

ド（右下がりの曲線）があり，前者は短期金利よりも長期金利のほうが高くなることを意味し，

後者は逆に短期金利よりも長期金利のほうが低くなることを意味しています． 

以下の例題では，実際の観測点からスポットレートを推定する問題を考え，ひいては，イール

ドカーブの推定を行います． 

 

（例題） 

期間（償還期間）が 1～10期まであり，各々に対するスポットレート tr  を用いて，関

数 Sが 

1

1

( ; , , ) 1/
t

t i

i

S t r r d


  ， 1/(1 0.01 )i

i id r    

のように定義されているとする．観測結果 ( , ( ))t S t の組 

( , ) {1, ,60}i it S i  

が与えられているとき， 

  
2

( )i i

i

S t S  

を最小化するようなスポットレートを推定せよ． 

この問題を定式化すると以下のようになります． 

なお，求めるスポットレートの単位は％（パーセント）とし，非負であるものとします． 

 

 

   

集合 Term 期間（償還期間）集合 

 

 

 

Point  

観測点の集合 

定数 itvalue i Point,　  観測点が何期目か 

 

 

iSvalue i Point,　  
観測値 

   

変数 tr t Term,　  t  期に対するスポットレート 

   

目的関数 (最小化) 

 

 

 

 
2

1

1

( ) ,

( ; , , )

1/ (1/ (1 0.01 ) )

i i

i

t

t
i

i

i

S tvalue Svalue

S t r r

r






 



  

理論値と観測値の誤差の二乗和 

 

 

 

   

制約条件 0 tr t Term  ,  スポットレートの非負条件 

   

 

 

この問題は目的関数が非線形ですので，非線形計画問題になります． 



 134 

 

定式化した結果を SIMPLEで記述すると以下のようになります． 

 

YieldCurvre.smp 

// スポットレートの設定 

Set Term(name=”期間”); // 期間集合 

Term = "1 .. 10"; 

Element t(set=Term); 

Variable r(name="スポットレート", index=t); 

 

// 観測点 

Set Point; 

Element i(set=Point); 

Parameter tvalue(name="tvalue", index=i); // 何期目か 

Parameter Svalue(name="Svalue", index=i); // 観測値(S) 

 

// 関数設定 

Expression d(index=t); 

d[t] = 1 / pow(1+0.01*r[t], t); 

// tvalue[i] 期における S の理論値 

Expression S(index=i); 

S[i] = 1 / sum(d[t], (t, t<=tvalue[i])); // tvalue[i] 期までの和 

Expression diff(index=i); 

diff[i] = S[i] - Svalue[i]; 

 

// 目的関数 

Objective err(name="理論値と観測値の誤差の二乗和", type=minimize); 

err = sum(pow(diff[i], 2), i); 

 

// 制約条件 

0 <= r[t]; // スポットレートの非負条件 

 

// 求解 

solve(); 

 

// 結果の標準出力 

err.val.print(); 

simple_printf("¥n"); 

r.val.print(); 

データファイル（tSvalue.csv）は以下のようになります． 

 

                         （左段の続き）           （中段の続き） 

 

21,4,0.272115 

22,4,0.272244 

23,4,0.272818 

24,4,0.271190 

25,5,0.222502 

26,5,0.223564 

27,5,0.222312 

28,5,0.223726 

29,5,0.222489 

30,5,0.223251 

31,6,0.190145 

32,6,0.190989 

33,6,0.191429 

34,6,0.191156 

35,6,0.191077 

36,6,0.191451 

37,7,0.168971 

38,7,0.167869 

39,7,0.168913 

40,7,0.168672 

41,7,0.167762 

42,7,0.167502 

43,8,0.152239 

44,8,0.150571 

45,8,0.151664 

46,8,0.151251 

47,8,0.152121 

48,8,0.151681 

49,9,0.138188 

50,9,0.138396 

51,9,0.139097 

52,9,0.139389 

53,9,0.138134 

54,9,0.138999 

55,10,0.128923 

56,10,0.129079 

57,10,0.129807 

58,10,0.129427 

59,10,0.128610 

60,10,0.129465 

i,tvalue,Svalue 

1,1,1.020476 

2,1,1.020582 

3,1,1.021692 

4,1,1.020304 

5,1,1.021171 

6,1,1.020387 

7,2,0.520158 

8,2,0.518884 

9,2,0.519507 

10,2,0.518841 

11,2,0.519271 

12,2,0.519520 

13,3,0.355121 

14,3,0.353832 

15,3,0.355088 

16,3,0.353742 

17,3,0.354565 

18,3,0.354443 

19,4,0.271774 

20,4,0.271318 

このモデルを実行すると，以下のような解が得られます． 

 

 

この節の最後に，実際に推定したスポットレートをもとに，Excel 上でイールドカーブを描

きます．そのために，スポットレートを Excel連係機能により，Excel上に表示します． 

モデルの結果出力の部分を以下のように変更する必要があります． 

 

結果は Excel上に，例えば下記のように表示されます． 

 

 

// 結果出力 

err.val.dump(); 

r.val.dump(); 

理論値と観測値の誤差の二乗和=1.74889e-05 

 

スポットレート[1]=2.07687 

スポットレート[2]=2.82648 

スポットレート[3]=3.7392 

スポットレート[4]=3.94718 

スポットレート[5]=4.37854 

スポットレート[6]=4.91763 

スポットレート[7]=5.05042 

スポットレート[8]=5.5363 

スポットレート[9]=5.76144 

スポットレート[10]=5.74989 

2.20 格付け推移行列推定問題(SAMPLES¥例題集 3.prj 参照) 

格付け（rating）とは，企業の発行する社債の元本，利息の支払い能力をランク形式で表示

したものです．格付け会社は独自の調査結果のもと，A，B，C や ＋，－ などの記号を用いて

対象社債のリスク度合いを示します．格付けは，社債の購入者側からすると購入判断時の評価指

標になりますし，発行体となる企業側からすると，資金調達の際の利回り決定の基準となります． 

格付け推移行列とは，ある格付け評価を受けている企業が，一定期間後にどのような格付けと

なるかについての確率を表す行列のことをいいます． 

以下では，この格付け推移行列の推定に関する例題を考えます． 

 

 

 

  

（例題） 

格付けとして，｛AAA, AA, A, BBB, BB, B, CCC, CC, C｝ の 9種類があるものと 

して，格付けに関する 1年後の推移行列 0Q  が以下のように与えられているものとする． 

 

 

 

0Q  を基に，1ヶ月単位期間の格付け推移行列 Q  を推定せよ． 

推移確率行列 Q  はマルコフ過程に従うものとします．そのとき，題意より行列 
12Q  は行

列 0Q  と本来一致するはずです．従って，行列 Q  を推定するとは，
12

0 F
Q Q  を最小にす

るような Q  を求めるということになります．なお 
F

・  は，Frobenius（フロベニウス）ノ

ルムのことで，行列の各成分の二乗和の平方根を表すものとします． 

また，格付け推移行列の性質として，行列 Q  の各行の和は 1 で各成分は非負であるものと

します． 

以上を踏まえると，定式化は以下のようになります． 

 

 

   

順序集合 Rating  格付けの集合 

   

定数 
0 ,ijq i j Rating,　  

1年後の格付け推移行列（ 0Q ）

の各要素 

   

変数 ,ijq i j Rating,  
1 ヶ月単位の格付け推移行列

（Q）の各要素 

   

目的関数 (最小化) 
2

12

0 F
Q Q  

行列 0Q と
12Q の差のフロベニ

ウスノルムの二乗 

   

制約条件 
1,ij

j Rating

q i Rating


    行列 Q  の各行の和は 1 

 

 
0 1, ,ijq i j Rating     行列 Q  の各要素条件 

   

 

 

この問題は目的関数が非線形ですので，非線形計画問題になります． 

定式化した結果を SIMPLEで記述すると以下のようになります． 

なお，行列 
12Q  に関しては， 

12 8 4 2 2 2 2 2( ) ( )Q Q Q Q Q Q Q       

で表現することにより，演算の回数を節約することができます． 

また，局所解に陥ることを回避するため，および収束までの反復回数を軽減するため，行列 Q  

の初期値として，対角成分に 0.9 を与え，非対角要素の上限値を 0.05としておきます． 

 

Rating1.smp 

// 集合・要素・変数・パラメータ 

 OrderedSet Rating; // 格付け集合 

 Element i(set=Rating), j(set=Rating), k(set=Rating); 

 Variable q(name="Q", index=(i,j)); // １ヶ月単位の格付け推移行列の各要素 

 Parameter q0(name="Q0", index=(i,j)); // １年後の格付け推移行列の各要素 

 // 定式の定義に利用する演算の回数を節約する記法 

 Expression q2(name="q2", index=(i,j)); 

 Expression q4(name="q4", index=(i,j)); 

 Expression q8(name="q8", index=(i,j)); 

 Expression q12(name="q12", index=(i,j)); 

 q2[i,j] = sum(q[i,k]*q[k,j], k); 

 q4[i,j] = sum(q2[i,k]*q2[k,j], k); 

 q8[i,j] = sum(q4[i,k]*q4[k,j], k); 

 q12[i,j] = sum(q8[i,k]*q4[k,j], k); 

 Expression diff(name="diff", index=(i,j)); 

 diff[i,j] = q0[i,j] - q12[i,j]; 

 // 目的関数 

 Objective diffnrm(name="行列 Q0 と 行列 Q の 12 乗 の差ノルムの二乗", 

 type=minimize); 

 diffnrm = sum(pow(diff[i,j], 2), (i,j)); 

 // 制約条件 

 sum(q[i,j], j) == 1; // 行列 Q について，各行の和は１ 

 0 <= q[i,i] <= 1; // 行列 Q の対角要素条件 

 0 <= q[i,j] <= 0.05, i != j; // 行列 Q の非対角要素条件 

 // 初期値設定 

 q[k,k]=0.9; 

 // 求解 

 solve(); 

データファイル（csv 形式）は以下のようになります． 

 

 

このモデルを実行すると以下のような解が得られます． 

Q0.csv 

Q0,AAA,AA,A,BBB,BB,B,CCC,CC,C 

AAA,0.9651,0.0349,0,0,0,0,0,0,0 

AA,0.0356,0.9382,0.0262,0,0,0,0,0,0 

A,0.0014,0.0433,0.9364,0.0186,0.0003,0,0,0,0 

BBB,0,0,0.0352,0.9456,0.0154,0.0038,0,0,0 

BB,0,0,0,0.1078,0.872,0.0049,0.0153,0,0 

B,0,0,0,0,0.0747,0.8762,0.041,0.0081,0 

CCC,0,0,0,0,0,0.0278,0.9654,0.0068,0 

CC,0,0,0,0,0,0,0.0083,0.9675,0.0242 

C,0,0,0,0,0,0,0,0.0266,0.9734 

// 結果出力 

diffnrm.val.print(); 

simple_printf("¥n"); 

simple_printf("Q"); 

simple_printf(",%s", k); 

simple_printf("¥n"); 

for( k=Rating.first(); k<Rating; k=Rating.next(k) ){ 

  simple_printf("%s", k); 

  simple_printf(",%7.5f", q[k,j]); 

  simple_printf("¥n"); 

} 

下記では，このモデルを Excel連係機能を用いて Excel上に表示させます．そのために， 

モデルの結果出力の部分を以下のように変更する必要があります． 

Excel 連係の詳細につきましては，「NUOPT Excel 連係マニュアル」をご覧下さい．結果は

Excel上に，例えば下記のように表示されます． 

 

 

 

// 結果出力 

diffnrm.val.dump(); 

q.val.dump(); 

行列 Q0 と 行列 Q の 12 乗 の差ノルムの二乗=0.000154701 

 

Q,"AAA","AA","A","BBB","BB","B","CCC","CC","C" 

"AAA",0.99680,0.00285,0.00006,0.00004,0.00007,0.00006,0.00004,0.00004,0.00004 

"AA",0.00298,0.99447,0.00218,0.00006,0.00008,0.00008,0.00005,0.00005,0.00005 

"A",0.00007,0.00371,0.99435,0.00152,0.00010,0.00009,0.00006,0.00006,0.00006 

"BBB",0.00007,0.00006,0.00289,0.99505,0.00158,0.00013,0.00006,0.00009,0.00007 

"BB",0.00008,0.00008,0.00005,0.00966,0.98839,0.00033,0.00123,0.00010,0.00008 

"B",0.00008,0.00006,0.00008,0.00003,0.00684,0.98880,0.00348,0.00056,0.00008 

"CCC",0.00005,0.00005,0.00006,0.00005,0.00006,0.00235,0.99688,0.00046,0.00005 

"CC",0.00005,0.00005,0.00006,0.00006,0.00008,0.00008,0.00059,0.99708,0.00194 

"C",0.00004,0.00004,0.00004,0.00005,0.00007,0.00007,0.00004,0.00210,0.99754 

Vector/Matrix を用いた SIMPLE モデル 

NUOPT V13 より導入された，Vector/Matrix クラスを用いたモデルを紹介します． 

 

 

実行時には，同じデータファイルを与えます． 

Rating2.smp 

Set Rating; 

Element i(set=Rating), j(set=Rating); 

 

Matrix Q0((i,j));  

Variable q(name="Q", index=(i,j)); 

Matrix Q((i,j)); Q[i,j] = q[i,j]; 

 

Matrix Q2((i,j)), Q4((i,j)), Q8((i,j)), Q12((i,j)); 

Q2 = Q*Q; 

Q4 = Q2*Q2; 

Q8 = Q4*Q4; 

Q12 = Q8*Q4; 

 

Matrix D((i,j)); 

D = Q0 - Q12; 

 

Objective diffnrm(type=minimize); 

diffnrm = inprod(D,D); 

 

Q*ones(Rating) == ones(Rating); 

0.00 <= q[i,i] <= 1;  

0 <= q[i,j] <= 0.05 , i!=j;  

 

q[i,i]=0.9; 

 

solve(); 

2.21 相関行列取得問題(SAMPLES¥例題集 3.prj 参照) 

複数の株価の時系列データ等，相関を持った乱数列を発生させる際に，元の相関行列内の要素

をある程度任意に変更したい場合があります．しかし，相関を持った乱数列を適切に発生させる

ためには，相関行列は半正定値である必要があります．そこで，与えられた行列に一番近く，か

つ，半正定値であるような相関行列を作成する問題を考えます． 

 

 

（例題） 

以下の対称行列 Aに対してフロベニウスノルムの意味で一番近く，かつ，半正定値であ

るような相関行列 X を求めよ． 

この問題を定式化すると以下のようになります． 

 

   

集合 {1,2, ,10}N   相関行列の行及び列の集合 

   

定数 , ,ijA i N j N   所与の行列の要素 

 minEig  相関行列の最小固有値 

   

変数 , ,ijX i N j N   相関行列の要素 

   

目的関数 (最小化) 

 
 

2

,

ij ij

i j N

X A


  
元の行列と相関行列の差のフロ

ベニウスノルムの二乗 

   

制約条件 X minEig  半正定値制約 

 1,iiX i N  　  対角要素は 1 

   

 

この問題は半正定値制約が入っているので，半正定値計画問題になります． 

定式化した結果を SIMPLEで記述すると以下のようになります． 

 

CorrelationMatrix.smp 

// 集合と添字 

OrderedSet N; 

Element i(set=N),j(set=N); 

 

// パラメータ 

Parameter A(index=(i,j)); // 与えられた行列の要素 

Parameter minEig;         // 出力される相関行列の最小固有値 

minEig = 1.0e-3; 

 

// 変数 

Variable  X(index=(i,j)); 

 

// 対称行列 

SymmetricMatrix M((i,j)); 

M[i,j] = X[i,j], i <= j; // 上三角部分のみ定義 

 

// 目的関数 

Objective diffnrm(type=minimize); // 差の行列のノルム 

diffnrm = sum((X[i,j]-A[i,j])*(X[i,j]-A[i,j]),(i,j)); 

 

// 制約条件 

M >= minEig;             // 半正定値制約 

X[j,i] == X[i,j], i < j; // Xは対称行列 

X[i,i] == 1;             // 対角要素は 1 

 

// 求解 

solve(); 

 

// 出力（CSV形式） 

simple_printf(" X"); 

simple_printf(",%5d", i); 

simple_printf("¥n"); 

for(i=N.first(); i<N; i=N.next(i)){ 

  simple_printf("%2d", i); 

  simple_printf(",%5.2f", X[i,j]); 

  simple_printf("¥n"); 

} 

データファイル（.csv形式）は以下のようになります． 

 

 

このモデルを実行すると以下のような解が得られます． 

 

 

 

 X,    1,    2,    3,    4,    5,    6,    7,    8,    9,   10 

 1, 1.00, 0.27, 0.15, 0.61, 0.65, 0.06, 0.10, 0.03, 0.57, 0.50 

 2, 0.27, 1.00, 0.10, 0.85, 0.15, 0.41, 0.00, 0.01,-0.04,-0.04 

 3, 0.15, 0.10, 1.00, 0.21, 0.01,-0.00,-0.00,-0.00, 0.01, 0.01 

 4, 0.61, 0.85, 0.21, 1.00, 0.34, 0.17, 0.20, 0.18, 0.12, 0.11 

 5, 0.65, 0.15, 0.01, 0.34, 1.00, 0.37,-0.00,-0.01, 0.10, 0.09 

 6, 0.06, 0.41,-0.00, 0.17, 0.37, 1.00, 0.00, 0.00,-0.02,-0.02 

 7, 0.10, 0.00,-0.00, 0.20,-0.00, 0.00, 1.00, 0.00,-0.00,-0.00 

 8, 0.03, 0.01,-0.00, 0.18,-0.01, 0.00, 0.00, 1.00,-0.01,-0.01 

 9, 0.57,-0.04, 0.01, 0.12, 0.10,-0.02,-0.00,-0.01, 1.00, 0.08 

10, 0.50,-0.04, 0.01, 0.11, 0.09,-0.02,-0.00,-0.01, 0.08, 1.00 

A.csv 

A,1,2,3,4,5,6,7,8,9,10 

1,1.00 ,0.17 ,0.17 ,0.90 ,0.90 ,0.00 ,0.10 ,0.00 ,0.80 ,0.70  

2,0.17 ,1.00 ,0.10 ,0.90 ,0.20 ,0.40 ,0.00 ,0.00 ,0.00 ,0.00  

3,0.17 ,0.10 ,1.00 ,0.20 ,0.00 ,0.00 ,0.00 ,0.00 ,0.00 ,0.00  

4,0.90 ,0.90 ,0.20 ,1.00 ,0.20 ,0.20 ,0.20 ,0.20 ,0.00 ,0.00  

5,0.90 ,0.20 ,0.00 ,0.20 ,1.00 ,0.40 ,0.00 ,0.00 ,0.00 ,0.00  

6,0.00 ,0.40 ,0.00 ,0.20 ,0.40 ,1.00 ,0.00 ,0.00 ,0.00 ,0.00  

7,0.10 ,0.00 ,0.00 ,0.20 ,0.00 ,0.00 ,1.00 ,0.00 ,0.00 ,0.00  

8,0.00 ,0.00 ,0.00 ,0.20 ,0.00 ,0.00 ,0.00 ,1.00 ,0.00 ,0.00  

9,0.80 ,0.00 ,0.00 ,0.00 ,0.00 ,0.00 ,0.00 ,0.00 ,1.00 ,0.00  

10,0.70 ,0.00 ,0.00 ,0.00 ,0.00 ,0.00 ,0.00 ,0.00 ,0.00 ,1.00 

2.22 ロバストポートフォリオ最適化問題（SAMPLES¥例題集 3.prj 参照） 

ポートフォリオのリスクを投資対象の収益率の分散共分散行列を用いて計測するマルコビッ

ツモデルにおいて，与えられた分散共分散行列は，しばしば不確実性を伴います．この不確実性

に対してロバストな解を得る以下の問題を考えます． 

以下の典型的な平均・分散モデルを考えます． 

max

. . 1

T T

x

T

x x x

s t x e

  


 (1) 

ここで， を期待リターン，をリターンの分散共分散行列，x をポートフォリオの重み，

をリスク回避係数とします． 

分散共分散行列に不確実性が伴うとして，以下のロバストポートフォリオ最適化問題を

考えます．なおU は，不確実性が伴うことによって取りうるに関する行列の集合とします． 

  
U

max max

. . 1

T T

x

T

x x x

s t x e

 


 



 (2) 

U として，分散共分散行列の各要素が 

      

のような区間を持つとします．このとき(2)は，新たに行列 ,U L を用いて以下のような問題に

置き換えることができます[6]．なお行列 A，B に対し， A B は，Aと Bの内積を表すものと

します． 

 max

. . 1

0
1

0, 0

T

x

T

T

x U L

s t x e

U L x

x

U L

   



 
 
 

 

 (3) 

（例題） 

分散共分散行列の各要素の下限を表す行列 sigL，上限を表す行列 sigU が以下のよう

に与えられているとき，上記(3)を解きなさい． 

 

 

 

 

 

ただし， 1  ，  0.1,0.1,0.1,0.1,0.1,0.1,0.1,0.1T  とします． 

この問題を定式化すると以下のようになります． 

 

   

集合 {1,2, ,8}A   投資対象の集合 

   

定数 , ,ijsigL i A j A   分散共分散行列の各要素の下限 

 , ,ijsigU i A j A   分散共分散行列の各要素の上限 

   リスク回避係数 

 ,i i A   各投資対象の平均収益率 

   

変数 ,ix i A  組入比率 

 , ,ijL i A j A    

 , ,ijU i A j A    

   

目的関数 (最大化) 

  

T x

sigU U sigL L



     

は要素ごとの積の和 

 

   

制約条件 

 
0

1T

U L x

x

 
 
 

 半正定値制約 

 1i

i A

x


  組入比率の総和は 1 

 0, ,ijL i A j A   　  
行列 L の各要素に関する非負条

件 

 0, ,ijU i A j A   　  
行列U の各要素に関する非負条

件 

   

 

この問題は半正定値制約が入っているので，半正定値計画問題になります． 

定式化した結果を SIMPLEで記述すると以下のようになります． 

 

RobustPortfolio.smp 

// 集合と添字 

Parameter nA; // 銘柄数 

Sequence Asset(from=1,to=nA); 

Element i(set=Asset); 

Element j(set=Asset); 

// パラメータ 

Parameter sigL(index=(i,j)); 

Parameter sigU(index=(i,j)); 

Parameter lambda; 

Parameter mu(index=i); 

// 変数 

Variable U(index=(i,j)); 

Variable L(index=(i,j)); 

Variable x(index=i); 

// 対称行列 

Sequence V(from=1,to=nA+1); 

Element v(set=V); 

Element w(set=V); 

SymmetricMatrix M((v,w));  

// Mの要素の定義（上三角部分のみ） 

M[i,j] = U[i,j] - L[i,j], i <= j; // 左上（の上三角部分） 

M[j,nA+1] = x[j];         // 右上 

M[nA+1,nA+1] = 1;         // 右下 

// 目的関数 

Objective f(type=maximize); 

f = sum(mu[i]*x[i],i) 

  - lambda*sum(sigU[i,j]*U[i,j]-sigL[i,j]*L[i,j],(i,j)); 

// 制約条件 

M >= 0; // 半正定値制約 

sum(x[i],i) == 1; 

U[i,j] == U[j,i], i > j; 

L[i,j] == L[j,i], i > j; 

U[i,j] >= 0; 

L[i,j] >= 0; 

// 求解 

solve(); 

// 出力 

x.val.print(); 

データファイルは以下の三つです． 

 

（csv形式） 

 

（csv形式） 

 

（dat形式） 

 

data.dat 

nA = 8; 

lambda = 1; 

mu = [1] 0.1 [2] 0.1 [3] 0.1 [4] 0.1 [5] 0.1 [6] 0.1 [7] 0.1 [8] 0.1; 

sigU.csv 

sigU,1,2,3,4,5,6,7,8 

1,3,1,2.5,-0.9,1,-0.4,2.5,2 

2,1,4.5,-1.4,1.2,0.5,-1.1,0.5,2.6 

3,2.5,-1.4,6,-0.5,1.2,1,0.4,-0.1 

4,-0.9,1.2,-0.5,6.5,1.6,0.5,1.3,-0.8 

5,1,0.5,1.2,1.6,5.5,-1.3,0.16,-1.9 

6,-0.4,-1.1,1,0.5,-1.3,11,-0.9,0.6 

7,2.5,0.5,0.4,1.3,0.16,-0.9,6.5,-1.2 

8,2,2.6,-0.1,-0.8,-1.9,0.6,-1.2,13.5 

sigL.csv 

sigL,1,2,3,4,5,6,7,8 

1,1.9,-1.4,-1,-1,-1,-0.5,-1,0.1 

2,-1.4,3.5,-1.5,0.5,-1.2,-1.2,0.4,0.6 

3,-1,-1.5,5,-1.125,1.1,0.9,0.3,-0.2 

4,-1,0.5,-1.125,6,1.5,0.4,1.2,-0.9 

5,-1,-1.2,1.1,1.5,4.5,-1.4,0.15,-2 

6,-0.5,-1.2,0.9,0.4,-1.4,9,-1,0.5 

7,-1,0.4,0.3,1.2,0.15,-1,5.5,-1.25 

8,0.1,0.6,-0.2,-0.9,-2,0.5,-1.25,11.5 

このモデルを実行すると以下のような解が得られます． 

 

 

 

x[1]=6.46383e-006 

x[2]=0.233187 

x[3]=0.175312 

x[4]=0.0633794 

x[5]=0.171046 

x[6]=0.131894 

x[7]=0.152881 

x[8]=0.0722949 

2.23 隣接行列（最大カット問題）（SAMPLES¥例題集 3.prj 参照） 

最大カット問題とは，重み付きグラフのノードを２つのグループに分割する問題で，グループ

分けをした際にカットされるエッジの重み総和を最大化する問題です．ここで，２つのノードが

異なるグループに属したとき，その２点をつなぐエッジがカットされると言います． 

この問題は，グラフ構造が隣接行列で与えられるという特徴を持ちます．また，数理計画法と

しては， SDP緩和問題として良質な上界値を求めることができる例題として知られています． 

 

 

この問題を NUOPT で解くために定式化を行いますが，まずは SDP 緩和問題として定式化し，

目的関数値の上界値を求めてみましょう．NUOPT V13から Vector/Matrixクラスが利用でき

るようになったため，ラプラス行列の計算を簡単に表現することができるようになりました．こ

こでラプラス行列とは，グラフの次数行列と隣接行列が与えられたとき，次数行列から隣接行列

を引いた行列で定義されます． 

 以下で紹介する定式化は参考文献[7]によります．まず，ノードの集合として N を与え，

変数として { 1,1},( )ix i N   を与えます．このとき，ノード ,i j N に対して， i jx x で ,i j

（例題） 

次のグラフは，ノードとして 1，2，3，4，5，6を持つ重み付きグラフです．エッジが

ない場合は，重みを 0とします． 

 

ノードを二つのグループに分けて，カットされるエッジの重みの総和を最大化したい場

合，どのようなグループとなるでしょうか． 

が同じグループに属していることを， i jx x で ,i j が異なるグループに属していることを表す

ことにすると，次のように定式化することができます． 

, ,

1
max (1 )

2

. . { 1,1}

ij i j

i j N i j

i

G x x

s t x i N

 



   



 

上記定式化に対し， 変数 ,ix i N を ij i iX x x で置き換える変数 , ,ijX i N j N  を導入す

ると，次の問題と同値になることが知られています． 

, ,

1
max (1 )

2

. . 0

rank( ) 1

1

ij ij

i j N i j

ii

G X

s t X

X

X i N

 





  



 

{ 1,1}ix   は， ( ) 1ii i iX x x  に対応し， ix ， jx の値の整合性は 0X ， rank( ) 1X  に対

応します．ここで制約式 rank( ) 1X   を緩和した問題が次の SDP緩和問題となります． 

, ,

1
max (1 )

2

. . 0

1

ij ij

i j N i j

ii

G X

s t X

X i N

 



  



 

上記定式化の目的関数を 

 

, , ,

1 1
(1 ) (1 )

2 4

1

4

1
diag( )

4

ij ij ij ij

i j N i j i j N

ij ii ij ij

i N j N j N

G X G X

G X G X

Ge G X

  

  

  

 
  

 

  

 

  

 

と式変形をすることで以下の定式化が得られます． 

集合 {1,2,3,4,5,6}N   ノードの集合 

   

定数 

, ,ijG i N j N   
ノード i N とノード j N

がつくるエッジの重みを表現

する隣接行列 

 
( )L diag Ge G   グラフG のラプラス行列 

（ eは単位ベクトル） 

   

変数 

, ,ijX i N j N   
緩和前ではノード i N とノ

ード j N がカットされてい

る場合に-1，カットされていな

い場合に 1をとる変数 

   

目的関数 (最大化) 
1

4
L X  

緩和前ではカットされるエッ

ジの重みの総和 

   

制約 0X  半正定値制約 

 1,iiX i N  　  対角要素は 1 

   

 

次に，行列をデータファイルから与える SIMPLE モデルを示します．以下のモデルでは不要

な変数を可能な限り消去するため，変数を上三角部分しか定義していないモデルになります． 

maxcut1.smp 

// 集合と添え字 

Set N; 

Element i(set=N); 

Element j(set=N); 

 

// 行列パラメータ 

Matrix G((i,j)); // 重み付き隣接行列 

Matrix L((i,j)); // グラフ G のラプラス行列 

L = diag(G*ones(N)) - G; 

 

Set NN(dim=2); // 上三角部分 

NN = setOf((i,j), i<=j); 

Element ij(set=NN); 

 

//変数 

Variable x(index=ij); 

SymmetricMatrix X((i,j)); 

X[i,j] = x[i,j], i<=j; 

X[j,i] = x[i,j], i<j; 

x[i,i] == 1; // 対角要素は 1 

X >= 0;      // 半正定値制約 

 

// 目的関数 

Objective obj(type=maximize); 

obj = 0.25 * inprod(L,X); 

 

// 求解 

solve(); 

 

// 出力 

obj.val.print(); 

データファイル（.csv形式）は以下のようになります． 

 

このモデルを実行すると，目的関数値 18.7437が得られます．これは最大化問題の緩和問題

ですので，本問題の上界値となります． 

次に 0-1 整数計画問題としての定式化を紹介します．SDP 緩和問題で得られた上界値の精度

を調べてみましょう． 

まず変数は，各ノードに対し 0-1 変数 , ( )ix i N を用意して「2 つのグループに分割する」

を「0 の値をとるグループ」と「1 の値をとるグループ」に対応させます．また，ノード i とノ

ード j が異なるグループのときに 1 をとり，同じグループのときに 0 をとる変数

 , ,ijy i N j N  を用意します．このとき，目的関数はカットされたエッジの重みの総和です

から， ij ij

i j

G y


 で表現されます．注意するべきこととして ix と ijy の整合性をとる制約を追加す

る必要があります．具体的に ( , , ) (0,1,1),(1,0,1),(0,0,0),(1,1,0)i j ijx x y  となる組合せを残せ

ばよいので，この組合せの補集合を考えて ( , , ) (0,1,0),(1,0,0),(0,0,1),(1,1,1)i j ijx x y  となる

組合せを排除する制約を導入します．このような制約を導入した定式化は次のようになります．  

 

   

集合 {1,2,3,4,5,6}N   ノードの集合 

   

行列 , ,ijG i N j N   

ノード i N とノード j N

がつくるエッジの重みを表現

する隣接行列 

   

0-1変数 ,ix i N  
0-1 の 2 つのグループのどち

らに入るかを表す変数 

G.csv 

G, 1, 2, 3, 4, 5, 6 

1, 0, 2, 3, 0, 1, 3  

2, 2, 0, 1, 2, 0, 0  

3, 3, 1, 0, 2, 0, 3  

4, 0, 2, 2, 0, 1, 4  

5, 1, 0, 0, 1, 0, 2  

6, 3, 0, 3, 4, 2, 0 

変数 , ,ijy i N j N   

カットされたエッジの情報を

,i jx x から得るための変数 

   

目的関数 (最大化) ij ij

i j

G y


  
カットされるエッジの重みの

総和 

   

制約 0 (1)i j ijx x y     ( , , ) (0,0,1)i j ijx x y  の排除 

 0 (2)i j ijx x y     ( , , ) (0,1,0)i j ijx x y  の排除 

 0 (3)i j ijx x y    ( , , ) (1,0,0)i j ijx x y  の排除 

 2 (4)i j ijx x y    ( , , ) (1,1,1)i j ijx x y  の排除 

   

 

上記にて変数 ijy は連続変数として定義されていますが，実際には 0-1しかとらない変数とな

ります．これは，式(1)(4)より 0ijy  が，式(2)(3)より， 1ijy  が導かれ，さらに ( , )i jx x の

組合せが (0,0),(0,1),(1,0),(1,1) しかとらないことと式(1)(2)(3)(4)から， 0 1ijy or が導

かれます．また，上記の不必要な組合せを排除する方法の代用として 0-1変数 ijz を追加して， 

2i j ij ijx x y z    

と表現する方法もあります．次に， SIMPLEモデルを示します． 以下のモデルでは不要な変数

を可能な限り消去するため，変数を上三角部分しか定義していないモデルになります． 



 159 

 

maxcut2.smp 

// 集合 

Set N; 

Element i(set=N); 

Element j(set=N); 

 

// 定数 

Matrix G((i,j)); 

 

Set NN(dim=2); // 対角成分 

NN = setOf((i,j),i<j); 

Element ij(set=NN); 

 

// 変数 

IntegerVariable x(index=i,type=binary); 

Variable y(index=ij);  

 

// x[i] x[j] y[i,j] の整合性をとる制約 

- x[i] - x[j] + y[i,j] <= 0,(i,j)<NN;  

- x[i] + x[j] - y[i,j] <= 0,(i,j)<NN;  

  x[i] - x[j] - y[i,j] <= 0,(i,j)<NN;  

  x[i] + x[j] + y[i,j] <= 2,(i,j)<NN;  

 

// 目的関数 

Objective obj(type=maximize); 

obj = sum(G[ij]*y[ij],ij); 

 

// 大規模問題には以下の近似解法を用いる 

// ※ その際，y を 0-1 整数変数にする必要がある 

// options.method="wcsp"; 

 

// 求解 

solve(); 

 

// 出力 

x.val.print(); 

obj.val.print(); 

このモデルを実行すると，目的関数値 18，ノード 1，4，5 のグループとノード 2,3,6 のグ

ループが得られます．SDP緩和問題で求められた目的関数値が 18.7437であったことを考える

と，SDP緩和問題が非常に良質な上界値を与えていることがわかります． 

上記のように SDP 緩和問題で良質な上界値を得られることがわかっていれば，上界値を用い

て WCSPアルゴリズムの精度保証に利用したり，終了条件を設定する際にも利用できます．参考

までに SDP緩和問題，単体法（＋分枝限定法），WCSPでそれぞれ解いた結果を以下に示します．

以下の表は SDP 緩和の上界値の精度と近似解法 WCSP の安定性を示す十分な結果を表していま

す． 

  SDP SIMPLEX WCSP 

Graph TIME Upper TIME Optimal TIME Lower 

G20.csv 0.2 464.39 0.2 452 0.05 452 

G30.csv 1.2 825.64 1.2 797 2.1 797 

G40.csv 6.5 1215.86 6.0 1164 1.7 1164 

G50.csv 23.1 1736.33 23.7 1651 7.6 1651 

G60.csv 55.5 2156.71 153.2 2068 27.4 2068 

# Graph データ G**.csv の「**」はノード数を表す．上記データはすべてノードに対して平

均 10のリンクが存在するように乱数を発生させたデータ． 

# WCSPアルゴリズムの設定は，最大探索時間 10秒，乱数を用いた 3回の探索試行を行った． 

2.24 セミナー割当問題 

資源制約付きスケジューリング問題の例として，本節ではセミナー会場提供会社におけるスケ

ジュール計画を以下で考えていきます． 

 

（例題 1） 

あるセミナー会場提供会社は，以下の 1～6 までのセミナーに対し，会場を提供する

ものとします． 

 

セミナー名 各セミナーのコマ数 

1 4 

2 6 

3 5 

4 3 

5 10 

6 7 

 

例題 1では，各セミナーにおいて 1日あたりに消化できるコマ数は 1つとします．別

のセミナーを同一の会場で行うことはできないものとし，1 日あたりの会場使用数の最

大を 4とするとき，これらすべてのセミナーが完了する時刻が最小となるようなスケジ

ュールを求めて下さい． 

なお，すべてのセミナーが終了するまでの日数は最大でも 20日までとします． 

この問題を定式化すると，以下のようになります．なお，セミナー会場に関しては区別がない

ので，一つの資源とみなすことに注意して下さい． 

 

セミナー集合 Se セミナー名 1～6 

   

資源 1 セミナー会場，供給量:常に 4 

   

モード 1A  セミナー1のコマ数消化方法 

処理時間:4 

必要量：資源 1を処理時間中常に 1 

2A  セミナー2のコマ数消化方法 

処理時間:6 

必要量：資源 1を処理時間中常に 1 

3A  セミナー3のコマ数消化方法 

処理時間:5 

必要量：資源 1を処理時間中常に 1 

4A  セミナー4のコマ数消化方法 

処理時間:3 

必要量：資源 1を処理時間中常に 1 

5A  セミナー5のコマ数消化方法 

処理時間:10 

必要量：資源 1を処理時間中常に 1 

6A  セミナー6のコマ数消化方法 

処理時間:7 

必要量：資源 1を処理時間中常に 1 

  

モード集合族 

 

,iMM i Se　  
セミナーiがとりうるモード集合 

アクティビティ ,iact i Se　  セミナーi をこなす作業 

処理モード： iA  

  

目的関数 (最小化) completionTime 最後の作業の完了時刻 

スケジュール期間 T 最大で 20日とする 

 

定式化した結果を SIMPLEで記述すると以下のようになります． 

SeminarAssign1.smp 

// 全体のスケジューリング期間 

Set T(name="T"); 

Element t(set=T); 

 

// 作業(セミナー)集合 

Set Se(name="Se"); 

Element se(set=Se); 

 

// モード集合 

Set M(name="M"); 

Element m(set=M); 

// 各セミナーがとりうるモード 

Set MM(name="MM", index=se); 

 

// 資源(部屋)集合 

Set Ro(name="Ro"); 

Element ro(set=Ro); 

 

// 各モード中の最大開催期間までの期間集合 

Set TT(name="TT"); 

Element tt(set=TT); 

 

// 必要資源 

ResourceRequire req(name="req", mode=M, resource=Ro, duration=TT); 

 

// 資源供給量 

ResourceCapacity cap(name="cap", resource=Ro, timeStep=T); 

 

// アクティビティ 

Activity act(name="セミナー", index=se, mode=MM[se]); 

 

// 目的関数（すべてのセミナーの完了時刻の最小化） 

Objective f(name=”最小完了日数”, type=minimize); 

f = completionTime; 

 

// 求解最大時間 

options.maxtim = 5; 

入力データ（dat形式）は，以下のようになります． 

 

Data_SeminarAssign1A.dat 

cap = [1,0] 4 [1,1] 4 [1,2] 4 [1,3] 4 [1,4] 4 [1,5] 4 [1,6] 4  

[1,7] 4 [1,8] 4 [1,9] 4 [1,10] 4 [1,11] 4 [1,12] 4  

[1,13] 4 [1,14] 4 [1,15] 4 [1,16] 4 [1,17] 4 [1,18] 4  

[1,19] 4 [1,20] 4; 

MM  = [1] A1 [2] A2 [3] A3 [4] A4 [5] A5 [6] A6; 

req = [A1,1,1] 1 [A1,1,2] 1 [A1,1,3] 1 [A1,1,4] 1  

[A2,1,1] 1 [A2,1,2] 1 [A2,1,3] 1 [A2,1,4] 1 [A2,1,5] 1  

[A2,1,6] 1  

[A3,1,1] 1 [A3,1,2] 1 [A3,1,3] 1 [A3,1,4] 1 [A3,1,5] 1  

[A4,1,1] 1 [A4,1,2] 1 [A4,1,3] 1  

[A5,1,1] 1 [A5,1,2] 1 [A5,1,3] 1 [A5,1,4] 1 [A5,1,5] 1  

[A5,1,6] 1 [A5,1,7] 1 [A5,1,8] 1 [A5,1,9] 1 [A5,1,10] 1  

[A6,1,1] 1 [A6,1,2] 1 [A6,1,3] 1 [A6,1,4] 1 [A6,1,5] 1  

[A6,1,6] 1 [A6,1,7] 1; 

 

// 求解 

solve(); 

 

// 結果の出力 

f.val.print(); 

simple_printf("セミナー[%s] = %d¥n", se, act[se].startTime); 

 

// ガントチャート出力 

Gantt g; 

g.add(act[se], se); 

g.dump(); 

なお，入力データを Excel連係機能を用いて与えることも可能です．具体的には次図のよう

なデータを Excelから NUOPTに渡します．Excel 連係機能におけるデータ受け渡し方法につ

きましては「NUOPT Excel連係マニュアル」をご覧下さい． 

 

 

結果は，以下のようになります． 

 

上記結果のセミナーの値は，各セミナーの開始日時を表しています． 

なお，ガントチャートによる解出力の可視化を行うと，次のようになります．  

最小完了日数=10 

セミナー[1] = 0 

セミナー[2] = 4 

セミナー[3] = 0 

セミナー[4] = 5 

セミナー[5] = 0 

セミナー[6] = 0 

例題 1の解の出力例： 

 

 

例題 1 では，各セミナーにおけるモードは 1 種類（一日一コマずつ消化していく）のみでし

たが，続く例題 2 では，各セミナーに対していくつかのモードを用意し，例題 1 からの結果の

推移を検証します． 

定式化に関しましては，例題 1 と同様です．ですので，モデルファイルの記述も例題 1 と同

一になります． 

例題 1と異なるのは，入力データのみです．Excel連係機能を用いて Excelから NUOPTへ

データを渡す場合には，次のようなデータを与えます． 

（例題 2） 

例題 1の条件のもと，各セミナーにおいて，コマ数の消化方法の選択肢をいくつか与

えるものとします．具体的には，以下の表の通りです． 

 

セミナー名 モード セミナーコマ数の消化方法 

1 A1_1 1-1-1-1 

2 A2_1 1-1-1-1-1-1 

 A2_2 2-2-2 

 A2_3 1-1-2-2 

3 A3_1 1-1-1-1-1 

 A3_2 2-2-1 

4 A4_1 1-1-1 

5 A5_1 1-1-1-1-1-1-1-1-1-1 

 A5_2 1-2-2-2-2-1 

 A5_3 2-2-2-2-2 

6 A6_1 1-1-1-1-1-1-1 

 A6_2 1-2-2-2 

 A6_3 2-2-2-1 

 

表の見方ですが，例えば A2_2 でしたら，セミナー2が全コマ数 6を 3日に渡って各々

2 ずつ消化する，ということになります．各セミナーは，そのセミナー特有に用意され

たモードのうちどれか一つを選択しなければなりません．例えばセミナー3 に対するモ

ードは A3_1，A3_2の 2つがあるので，それらのうちのどちらかを選択してセミナーの

コマ数を消化するということになります． 

すべてのセミナーが完了する時刻が最小となるように，適切なモードを選択してスケ

ジュールを求めて下さい． 

例題 1 と比較すると，各セミナーにおいてとりうるモードの数が増えたことがおわかりにな

ると思います． 

結果は，以下のようになります． 

 

 

最小完了日数=9 

セミナー[1] = 0 

セミナー[2] = 5 

セミナー[3] = 0 

セミナー[4] = 4 

セミナー[5] = 0 

セミナー[6] = 5 

結果から明らかなように，例題 1 に比べて，例題 2 ではセミナー完了時刻を短縮できるよう

なスケジュール計画を立てることができました．これは，各セミナーがとりうるモードの数が例

題 1と比較して増えたことによるためです． 

なお，ガントチャートによる解出力の結果は以下のようになります．  

 

例題 2の解の出力例： 

 

 

この節の最後に，今までは資源となる会場は一つのみでしたが，もう一つ資源となる会場を追

加してみましょう．セミナーを実施するメイン会場（大部屋）とサブ会場（小部屋）のようなも

のを想定していただければ，おわかりになるかと思います． 

 

定式化としては以下のように，各セミナーともに，メイン会場でセミナーを行った後にサブ会

場で行うという先行制約が加わります．なお先行関係は により表すものとします． 

（例題 3） 

例題 2の条件のもと，さらにメイン会場を用意します． 

各セミナーともにメイン会場で 1日セミナーを行った後，その他の会場（以下では， 

サブ会場と呼ぶことにします）において各個別のセミナーを行うものとします．サブ会場

でのセミナー実施方法は，例題 2と同一とします．1日あたりのサブ会場使用数の最大に

関しましても，例題 1・例題 2同様，4とします． 

メイン会場では 1日 1種類のセミナーしか行うことができず，しかもメイン会場は各 

セミナーのはじめの 1 コマのみしか利用することができないものとするとき，すべての

セミナーが完了する時刻が最小となるようなスケジュールを求めて下さい． 

資源 

0 メイン会場，供給量:常に 1 

1 サブ会場，供給量:常に 4 

  

先行制約 ,0 ,1,i iact act i Se 　  

任意のセミナー i において，メイン会場での

セミナーを終えた後にメイン以外の会場で

のセミナーに移行する 

 

上記定式化の追加を反映した SIMPLEでのモデル記述は，以下のようになります． 

 

SeminarAssign1B.smp 

// 全体のスケジューリング期間 

Set T(name="T"); 

Element t(set=T); 

 

// 作業(セミナー)集合 

Set Se(name="Se"); 

Element se(set=Se); 

 

// モード集合 

Set M(name="M"); 

Element m(set=M); 

 

// 資源(部屋)集合 

Set Ro(name="Ro"); 

Element ro(set=Ro); 

 

// 各セミナーがとりうるモード 

Set MM(name="MM", index=(se,ro)); 

 

// 各モード中の最大開催期間までの期間集合 

Set TT(name="TT"); 

Element tt(set=TT); 

 

// 必要資源 

ResourceRequire req(name=”req”, mode=M, resource=Ro, duration=TT); 

ResourceRequire cap(name=”cap”, resource=Ro, timeStep=TT); 

入力データは，Excel連係機能を用いて以下のようなデータを NUOPTに渡します． 

メイン会場に対するモードは，各セミナーともに large_room として与えていることに注意

して下さい． 

// アクティビティ 

Activity act(name="セミナー", index=(se,ro), mode=MM[se,ro]); 

// 先行制約 

act[se,ro-1] < act[se,ro], ro > 0; 

 

// 目的関数（すべてのセミナーの完了時刻の最小化） 

Objective f(name=”最小完了日数”, type=minimize); 

f = completionTime; 

 

// 求解最大時間 

options.maxtim = 5; 

 

// 求解 

solve(); 

 

// 結果の出力 

f.val.print(); 

simple_printf("セミナー[%s,%s] = %d¥n", se, ro, act[se,ro].startTime); 

 

// ガントチャート出力 

Gantt g; 

g.add(act[se,ro],se,ro); 

g.dump(); 

結果は，以下のようになります． 

 

では，この結果をガントチャートによる解出力で可視化して検証しましょう．確かに各セミナ

ーともに，メイン会場でのセミナー消化後，サブ会場でセミナーを行っているのが確認できます． 

最小完了日数=11 

セミナー[1,0] = 2 

セミナー[1,1] = 3 

セミナー[2,0] = 4 

セミナー[2,1] = 5 

セミナー[3,0] = 1 

セミナー[3,1] = 2 

セミナー[4,0] = 5 

セミナー[4,1] = 7 

セミナー[5,0] = 0 

セミナー[5,1] = 1 

セミナー[6,0] = 3 

セミナー[6,1] = 4 

例題 3の解の出力例： 

2.25 ジョブショップスケジューリング問題（SAMPLES¥例題集 3.prj 参照） 

ジョブショップスケジューリング問題とは，生産計画等の現場で現れ，仕事(ジョブ)を機械

に効率的に割り振る事で，完了時刻，納期遅れ等の最小化を目的としたスケジューリング問題で

す．ここでは，特に機械の数が複数で，仕事が複数の作業(オペレーション)から構成され，各

仕事の全ての作業が処理されると，仕事の処理が完了する直列機械(serial machines)につ

いて言及します．本問題は，作業が処理される順序によって，以下の 3つに大別されます． 

(1) オープンショップ問題（2.25.1 節） 

各作業が処理される機械の順序は決まっていない． 

(2) フローショップ問題（2.25.2節） 

各作業が処理される機械の順序が同じ． 

(3) ジョブショップ問題（2.25.3 節） 

各作業が処理される機械の順序が決められているが，これらの順序は仕事ごとに異なって

いてもよい．フローショップ問題の一般形． 

 

以下で，上記(1)～(3)の例題を紹介します．必要に応じて「NUOPT_SIMPLE_マニュアル」

の第 6章も併せてご覧下さい． 

 

2.25.1 オープンショップ問題 

 

 

（例題）オープンショップ問題 

3 つの作業から構成される，3 つの仕事があるとします．各作業を処理する機械と処理

時間は以下のように与えられています．  

 

 作業 1(機械 1) 作業 2(機械 2) 作業 3(機械 3) 

仕事 a 5 時間 8 時間 4 時間 

仕事 b 5 時間 4 時間 6 時間 

仕事 c 6 時間 5 時間 5 時間 

 

各機械が作業を処理している間は，他の作業を処理する事は出来ません．また，各作業

を処理する順番に制限はありません．この場合の最後の作業の完了時刻が最小となるよ

うにするには，各仕事をどのように機械に割り振ればいいでしょうか．なお，すべての

仕事を終えるまでの所要時間は最大で 30日までとする． 

この問題を資源制約付きスケジューリング問題で定式化すると，例えば以下の様になります． 

各仕事の各作業は同時には 2 つ以上処理出来ない事を，ダミー資源を用いて表現している事に

注意して下さい． 

 

資源 

machine1 機械 1，供給量:常に 1 

machine2 機械 2，供給量:常に 1 

machine3 機械 3，供給量:常に 1 

d1 ダミー資源 1，供給量:常に 1 

d2 ダミー資源 2，供給量:常に 1 

d3 ダミー資源 3，供給量:常に 1 

モード 

mode_a_1 
処理時間:5 

必要量：machine1,d1を処理時間中常に 1 

mode_a_2 
処理時間:8 

必要量：machine2,d1を処理時間中常に 1 

mode_a_3 
処理時間:4 

必要量：machine3,d1を処理時間中常に 1 

mode_b_1 
処理時間:5 

必要量：machine1,d2を処理時間中常に 1 

mode_b_2 
処理時間:4 

必要量：machine2,d2を処理時間中常に 1 

mode_b_3 
処理時間:6 

必要量：machine3,d2を処理時間中常に 1 

mode_c_1 
処理時間:6 

必要量：machine1,d3を処理時間中常に 1 

mode_c_2 
処理時間:5 

必要量：machine2,d3を処理時間中常に 1 

mode_c_3 
処理時間:5 

必要量：machine3,d3を処理時間中常に 1 

アクティビティ 

,1aact  
仕事 aの作業 1 

処理モード：mode_a_1 

,2aact  
仕事 aの作業 2 

処理モード：mode_a_2 

,3aact  
仕事 aの作業 3 

処理モード：mode_a_3 

,1bact  
仕事 bの作業 1 

処理モード：mode_b_1 

,2bact  
仕事 bの作業 2 

処理モード：mode_b_2 

,3bact  
仕事 bの作業 3 

処理モード：mode_b_3 

,1cact  
仕事 cの作業 1 

処理モード：mode_c_1 

,2cact  
仕事 cの作業 2 

処理モード：mode_c_2 

,3cact  
仕事 cの作業 3 

処理モード：mode_c_3 

目的関数 (最小化) completionTime 最後の作業の完了時刻 

スケジュール期間 T 最大で 30時間とする 

 

これを SIMPLE で記述すると次のようになります． 

なお，アクティビティの引数に渡すモード集合ですが，仕事・作業の各組み合わせに対してど

のモードが対応するかという集合 AvailModeを用意し，それを引数として渡します． 

 

 

 

openshop.smp 

// 作業集合 

Set J; // 仕事 

Element j(set=J); 

Set S; // 作業 

Element s(set=S); 

// モード集合 

Set M; 

Element m(set=M); 

Set AvailMode(name="AvailMode", index=(j,s)); // 各仕事のオペレーショ

ンにおいて処理されるモード 

// 資源集合 

Set R; 

Element r(set=R); 

// 作業時間集合 

Set D; // 各モードの作業時間の最大 

Element d(set=D); 

// 期間集合 

Set T; // スケジュール期間 

T = "0 .. 30"; 

Element t(set=T); 

データファイルは次のようになります． 

 

 

data.dat 

AvailMode = 

[a,1] mode_a_1 

[a,2] mode_a_2 

[a,3] mode_a_3 

[b,1] mode_b_1 

[b,2] mode_b_2 

[b,3] mode_b_3 

[c,1] mode_c_1 

[c,2] mode_c_2 

[c,3] mode_c_3 

; 

 

// アクティビティ(変数) 

Activity act(name="act", index=(j,s), mode=AvailMode[j,s]); 

 

// 定数 

// 必要資源量 

ResourceRequire req(name="req", mode=M, resource=R, duration=D); 

// 資源供給量 

ResourceCapacity cap(name="cap", resource=R, timeStep=T); 

cap[r,t] = 1; 

 

// 目的関数 

Objective f(type=minimize); 

f = completionTime; // 最後の作業の完了時刻最小化 

 

// 求解最大時間の設定 

options.maxtim = 15; 

 

// 求解 

solve(); 

 

// 結果の標準出力 

simple_printf("act[%s,%d] = %d¥n", j, s, act[j,s].startTime); 

req.dat 

req = 

[mode_a_1,machine1,1] 1 [mode_a_2,machine2,1] 1  [mode_a_3,machine3,1] 1 

[mode_a_1,machine1,2] 1 [mode_a_2,machine2,2] 1  [mode_a_3,machine3,2] 1 

[mode_a_1,machine1,3] 1 [mode_a_2,machine2,3] 1  [mode_a_3,machine3,3] 1 

[mode_a_1,machine1,4] 1 [mode_a_2,machine2,4] 1  [mode_a_3,machine3,4] 1 

[mode_a_1,machine1,5] 1 [mode_a_2,machine2,5] 1  

                             [mode_a_2,machine2,6] 1  

                             [mode_a_2,machine2,7] 1  

                             [mode_a_2,machine2,8] 1  

 

[mode_b_1,machine1,1] 1 [mode_b_2,machine2,1] 1  [mode_b_3,machine3,1] 1 

[mode_b_1,machine1,2] 1 [mode_b_2,machine2,2] 1  [mode_b_3,machine3,2] 1 

[mode_b_1,machine1,3] 1 [mode_b_2,machine2,3] 1  [mode_b_3,machine3,3] 1 

[mode_b_1,machine1,4] 1 [mode_b_2,machine2,4] 1  [mode_b_3,machine3,4] 1 

[mode_b_1,machine1,5] 1                           [mode_b_3,machine3,5] 1 

                                                          [mode_b_3,machine3,6] 1 

 

[mode_c_1,machine1,1] 1 [mode_c_2,machine2,1] 1 [mode_c_3,machine3,1] 1 

[mode_c_1,machine1,2] 1 [mode_c_2,machine2,2] 1 [mode_c_3,machine3,2] 1 

[mode_c_1,machine1,3] 1 [mode_c_2,machine2,3] 1 [mode_c_3,machine3,3] 1 

[mode_c_1,machine1,4] 1 [mode_c_2,machine2,4] 1 [mode_c_3,machine3,4] 1 

[mode_c_1,machine1,5] 1 [mode_c_2,machine2,5] 1 [mode_c_3,machine3,5] 1 

[mode_c_1,machine1,6] 1 

 

[mode_a_1,d1,1] 1 [mode_a_2,d1,1] 1  [mode_a_3,d1,1] 1 

[mode_a_1,d1,2] 1 [mode_a_2,d1,2] 1  [mode_a_3,d1,2] 1 

[mode_a_1,d1,3] 1 [mode_a_2,d1,3] 1  [mode_a_3,d1,3] 1 

[mode_a_1,d1,4] 1 [mode_a_2,d1,4] 1  [mode_a_3,d1,4] 1 

[mode_a_1,d1,5] 1 [mode_a_2,d1,5] 1  

                     [mode_a_2,d1,6] 1  

                     [mode_a_2,d1,7] 1  

                     [mode_a_2,d1,8] 1  

実行すると，各作業の開始時刻 

 

が出力されます． 

 

以下では，ガントチャートにより解の出力の可視化を行います． 

act["a",1] = 0 

act["a",2] = 5 

act["a",3] = 13 

act["b",1] = 6 

act["b",2] = 13 

act["b",3] = 0 

act["c",1] = 11 

act["c",2] = 0 

act["c",3] = 6 

[mode_b_1,d2,1] 1 [mode_b_2,d2,1] 1 [mode_b_3,d2,1] 1 

[mode_b_1,d2,2] 1 [mode_b_2,d2,2] 1 [mode_b_3,d2,2] 1 

[mode_b_1,d2,3] 1 [mode_b_2,d2,3] 1 [mode_b_3,d2,3] 1 

[mode_b_1,d2,4] 1 [mode_b_2,d2,4] 1 [mode_b_3,d2,4] 1 

[mode_b_1,d2,5] 1                       [mode_b_3,d2,5] 1 

                                            [mode_b_3,d2,6] 1 

 

[mode_c_1,d3,1] 1 [mode_c_2,d3,1] 1 [mode_c_3,d3,1] 1 

[mode_c_1,d3,2] 1 [mode_c_2,d3,2] 1 [mode_c_3,d3,2] 1 

[mode_c_1,d3,3] 1 [mode_c_2,d3,3] 1 [mode_c_3,d3,3] 1 

[mode_c_1,d3,4] 1 [mode_c_2,d3,4] 1 [mode_c_3,d3,4] 1 

[mode_c_1,d3,5] 1 [mode_c_2,d3,5] 1 [mode_c_3,d3,5] 1 

[mode_c_1,d3,6] 1 

; 

例題（オープンショップ問題）の解の出力例： 

2.25.2 フローショップ問題 

 

この問題を記述するには，各作業の処理順序を制約する以下の先行制約を加える必要がありま

す．なお，以下で用いている  は先行関係を表す記号とします． 

 

先行制約 

,1 ,2a aact act  
仕事 aの作業 1は，仕事 aの作業 2に先

行する 

,2 ,3a aact act  
仕事 aの作業 2は，仕事 aの作業 3に先

行する 

,1 ,2b bact act  
仕事 bの作業 1は，仕事 bの作業 2に先

行する 

,2 ,3b bact act  
仕事 bの作業 2は，仕事 bの作業 3に先

行する 

,1 ,2c cact act  
仕事 cの作業 1は，仕事 cの作業 2に先

行する 

,2 ,3c cact act  
仕事 cの作業 2は，仕事 cの作業 3に先

行する 

 

上記の先行制約を加えた SIMPLE のモデルは以下の様になります．尚，各仕事の各作業を同

時に 2つ以上処理出来ない事は，先行制約で表現されている事に注意して下さい． 

 

flowshop.smp 

// 作業集合 

Set J; // 仕事 

Element j(set=J); 

Set S; // 作業 

Element s(set=S); 

// モード集合 

Set M; 

Element m(set=M); 

 

（例題）フローショップ問題 

2.23.1節のオープンショップ問題の条件の下，各仕事は，作業 1, 作業 2, 作業 3 の

順で処理されなければならないとします．この場合の最後の作業の完了時刻が最小とな

るようにするには，どのように機械を割り振ればよいでしょうか． 

Set AvailMode(name="AvailMode", index=(j,s)); // 各仕事のオペレーシ

ョンにおいて処理されるモード 

// 資源集合 

Set R; 

Element r(set=R); 

// 作業時間集合 

Set D; // 各モードの作業時間の最大 

Element d(set=D); 

// 期間集合 

Set T; // スケジュール期間 

T = "0 .. 30"; 

Element t(set=T); 

 

// アクティビティ(変数) 

Activity act(name="act", index=(j,s), mode=AvailMode[j,s]); 

 

// 定数 

// 必要資源量 

ResourceRequire req(name="req", mode=M, resource=R, duration=D); 

// 資源供給量 

ResourceCapacity cap(name="cap", resource=R, timeStep=T); 

cap[r,t] = 1; 

 

// 目的関数(最後の作業の完了時刻の最小化) 

Objective f(type=minimize); 

f = completionTime; 

 

// 先行制約 

act[j,s-1] < act[j,s], 1 < s; 

 

// 求解最大時間の設定 

options.maxtim = 15; 

 

// 求解 

solve(); 

 

// 結果の標準出力 

simple_printf("act[%s,%d] = %d¥n", j, s, act[j,s].startTime); 

先行制約を設けた事により，ダミー資源を用いる必要が無くなった為，データファイルの１つ

が以下の様に修正されます． 

 

 実行すると，各作業の開始時刻が以下のように出力されます． 

req.dat 

req = 

[mode_a_1,machine1,1] 1 [mode_a_2,machine2,1] 1  [mode_a_3,machine3,1] 1 

[mode_a_1,machine1,2] 1 [mode_a_2,machine2,2] 1  [mode_a_3,machine3,2] 1 

[mode_a_1,machine1,3] 1 [mode_a_2,machine2,3] 1  [mode_a_3,machine3,3] 1 

[mode_a_1,machine1,4] 1 [mode_a_2,machine2,4] 1  [mode_a_3,machine3,4] 1 

[mode_a_1,machine1,5] 1 [mode_a_2,machine2,5] 1  

                             [mode_a_2,machine2,6] 1  

                             [mode_a_2,machine2,7] 1  

                             [mode_a_2,machine2,8] 1  

 

[mode_b_1,machine1,1] 1 [mode_b_2,machine2,1] 1  [mode_b_3,machine3,1] 1 

[mode_b_1,machine1,2] 1 [mode_b_2,machine2,2] 1  [mode_b_3,machine3,2] 1 

[mode_b_1,machine1,3] 1 [mode_b_2,machine2,3] 1  [mode_b_3,machine3,3] 1 

[mode_b_1,machine1,4] 1 [mode_b_2,machine2,4] 1  [mode_b_3,machine3,4] 1 

[mode_b_1,machine1,5] 1                              [mode_b_3,machine3,5] 1 

                                                          [mode_b_3,machine3,6] 1 

 

[mode_c_1,machine1,1] 1 [mode_c_2,machine2,1] 1 [mode_c_3,machine3,1] 1 

[mode_c_1,machine1,2] 1 [mode_c_2,machine2,2] 1 [mode_c_3,machine3,2] 1 

[mode_c_1,machine1,3] 1 [mode_c_2,machine2,3] 1 [mode_c_3,machine3,3] 1 

[mode_c_1,machine1,4] 1 [mode_c_2,machine2,4] 1 [mode_c_3,machine3,4] 1 

[mode_c_1,machine1,5] 1 [mode_c_2,machine2,5] 1 [mode_c_3,machine3,5] 1 

[mode_c_1,machine1,6] 1 

;  

以下では，2.25.1節同様，結果のガントチャート出力を行います． 

 

例題（フローショップ問題）の解の出力例： 

 

act["a",1] = 11 

act["a",2] = 16 

act["a",3] = 24 

act["b",1] = 6 

act["b",2] = 11 

act["b",3] = 16 

act["c",1] = 0 

act["c",2] = 6 

act["c",3] = 11 

2.25.3 ジョブショップ問題 

 

この問題は，先行制約が以下の様に変更されます． 

 

先行制約 ,1 ,3a aact act  仕事 aの作業 1は，仕事 aの作業 3に先行

する 

,3 ,2a aact act  仕事 aの作業 3は，仕事 aの作業 2に先行

する 

,1 ,2b bact act  仕事 bの作業 1は，仕事 bの作業 2に先行

する 

,2 ,3b bact act  仕事 bの作業 2は，仕事 bの作業 3に先行

する 

,2 ,1c cact act  仕事 cの作業 2は，仕事 cの作業 1に先行

する 

,1 ,3c cact act  仕事 cの作業 1は，仕事 cの作業 3に先行

する 

 

上記の先行制約をデータから与えられるように SIMPLE のモデルとデータを修正します． 

（例題）ジョブショップ問題 

2.23.2 節のオープンショップ問題の条件の下，各仕事は以下の順に処理されなければ

ならないものとします． 

 

 作業 1(機械 1) 作業 2(機械 2) 作業 3(機械 3) 

仕事 a 1 3 2 

仕事 b 1 2 3 

仕事 c 2 1 3 

 

この場合の最後の作業の完了時刻が最小となるようにするには，どのように機械を割り

振ればよいでしょうか． 

jobshop.smp 

// 作業集合 

Set J; // 仕事 

Element j(set=J); 

Set S; // 作業 

Element s(set=S); 

// モード集合 

Set M; 

Element m(set=M); 

Set AvailMode(name="AvailMode", index=(j,s)); // 各仕事のオペレーショ

ンにおいて処理されるモード 

// 資源集合 

Set R; 

Element r(set=R); 

// 作業時間集合 

Set D; // 各モードの作業時間の最大 

Element d(set=D); 

// 期間集合 

Set T; // スケジュール期間 

T = "0 .. 30"; 

Element t(set=T); 

 

// アクティビティ(変数) 

Activity act(name="act", index=(j,s), mode=AvailMode[j,s]); 

 

// 定数 

// 必要資源量 

ResourceRequire req(name="req", mode=M, resource=R, duration=D); 

// 資源供給量 

ResourceCapacity cap(name="cap", resource=R, timeStep=T); 

cap[r,t] = 1; 

 

// 目的関数(最後の作業の完了時刻の最小化) 

Objective f(type=minimize); 

f = completionTime; 

データファイルの１つが以下の様に修正されます． 

 

// 先行制約 

Set Prec(name="Prec", dim=3); 

Element u(set=S); 

Element v(set=S); 

act[j,u] < act[j,v], (j,u,v)<Prec; 

// 求解最大時間の設定 

options.maxtim = 15; 

 

// 求解 

solve(); 

 

// 結果の標準出力 

simple_printf("act[%s,%d] = %d¥n", j, s, act[j,s].startTime); 

 

data.dat 

AvailMode = 

[a,1] mode_a_1 

[a,2] mode_a_2 

[a,3] mode_a_3 

[b,1] mode_b_1 

[b,2] mode_b_2 

[b,3] mode_b_3 

[c,1] mode_c_1 

[c,2] mode_c_2 

[c,3] mode_c_3 

; 

 

Prec = 

a 1 3 

a 3 2 

b 1 2 

b 2 3 

c 2 1 

c 1 3 

; 

実行すると，各作業の開始時刻が以下のように出力されます． 

 

 

2.25.2節同様，結果のガントチャート出力を以下のように行います． 

 

例題（ジョブショップ問題）の解の出力例： 

 

act["a",1] = 0 

act["a",2] = 14 

act["a",3] = 5 

act["b",1] = 5 

act["b",2] = 10 

act["b",3] = 14 

act["c",1] = 10 

act["c",2] = 0 

act["c",3] = 20 

2.25.4 リスケジューリング問題 

実際の現場では，ある時機械が故障した等の突発事故が起こる事がしばしばあります．その場

合，過去のスケジュールを固定し，条件を変更した後，再度スケジュールを組み直します．この

ような問題をリスケジューリング問題と呼ぶ事にします． 

 

この問題を記述するには，過去のスケジュールを固定する為，アクティビティ固定関数

（fixActivity）を使用する必要があります．また，未来のスケジュールが過去に来ないよう

にする為，全ての作業に先行する sourceActivity(自動で定義されている)という先行制約

を記述する必要があります． 

以上のことを踏まえると，制約としては新たに以下のものが加わります． 

（例題）リスケジューリング問題 

2.23.3節のジョブショップ問題を解いた結果，各作業の開始時刻は， 

 

 作業 1(機械 1) 作業 2(機械 2) 作業 3(機械 3) 

仕事 a 0 14 5 

仕事 b 5 10 14 

仕事 c 10 0 20 

 

となりました．このスケジュールに基づいて機械を運転していた所，10 時間が過ぎた

所で，機械 2が故障し，5時間停止する事になりました．この場合の最後の作業の完了

時刻が最小となるようにはどのようにスケジュールを組み直せばよいでしょうか． 

アクティビティの開

始時刻固定 

,1( . )afixActivity act startTime  仕事a の作業1 の開始時刻を 0 に固定

する 

,3( . )afixActivity act startTime  仕事a の作業3 の開始時刻を 5 に固定

する 

,1( . )bfixActivity act startTime  仕事b の作業1 の開始時刻を 5 に固定

する 

,2( . )cfixActivity act startTime  仕事c の作業2 の開始時刻を 0 に固定

する 

アクティビティのモ

ード固定 

,1( )afixActivity act  仕事 a の作業 1 のモードを mode_a_1

に固定する 

,3( )afixActivity act  仕事 a の作業 3 のモードを mode_a_3

に固定する 

,1( )bfixActivity act  仕事 b の作業 1 のモードを mode_b_1

に固定する 

,2( )cfixActivity act  仕事 c の作業 2 のモードを mode_c_2

に固定する 

先行制約 ,2 ,10asourceActivity act 　 仕事 aの作業 2 は，時刻 10 以前には処

理されない 

,2 ,10bsourceActivity act 　 仕事 bの作業 2 は，時刻 10 以前には処

理されない 

,3,10bsourceActivity act 　 仕事 bの作業 3 は，時刻 10 以前には処

理されない 

,1,10csourceActivity act 　 仕事 cの作業 1 は，時刻 10 以前には処

理されない 

,3,10csourceActivity act 　 仕事 cの作業 3 は，時刻 10 以前には処

理されない 

 

また，機械 2は，時刻 10 から 5時間停止するので，資源が以下の様に修正されます． 

 

資源 machine1 機械 1，供給量:常に 1 

machine2 
機械 2，供給量:時刻 10以上 15未満 0，そ

の他常に 1 

machine3 機械 3，供給量:常に 1 

 

これを SIMPLE で記述すると以下の様になります． 

jobShopRsch.smp 

// 作業集合 

Set J; // 仕事 

Element j(set=J); 

Set S; // 作業 

Element s(set=S); 

// モード集合 

Set M; 

Element m(set=M); 

Set AvailMode(name="AvailMode", index=(j,s)); // 各仕事のオペレーション

において処理されるモード 

// 資源集合 

Set R; 

Element r(set=R); 

// 作業時間集合 

Set D; // 各モードの作業時間の最大 

Element d(set=D); 

// 期間集合 

Set T; // スケジュール期間 

T = "0 .. 30"; 

Element t(set=T); 

// アクティビティ(変数) 

Activity act(name="act", index=(j,s), mode=AvailMode[j,s]); 

 

// 必要資源量（定数） 

ResourceRequire req(name="req", mode=M, resource=R, duration=D); 

// 資源供給量（定数） 

ResourceCapacity cap(name="cap", resource=R, timeStep=T); 

cap[r,t] = 1; 

cap["machine2", t] = 0, 10<=t<=14; // 故障に対応 

 

// 目的関数(最後の作業の完了時刻の最小化) 

Objective f(type=minimize); 

f = completionTime; 

 

// 先行制約 

Set Prec(name="Prec", dim=3); 

Element u(set=S); 

Element v(set=S); 

act[j,u] < act[j,v], (j,u,v)<Prec; 

データファイルの１つが以下の様に修正されます． 

/* --- リスケジューリングの為の制約 --- */ 

// 過去(0 .. 10) のスケジュールを固定 

Set FixAct(name="FixAct", dim=2); 

Parameter fixTime(name="fixTime", index=(j,s)); 

act[j,s].startTime = fixTime[j,s], (j,s) < FixAct; 

Parameter fixMode(name="fixMode", index=(j,s)); 

act[j,s] = fixMode[j,s], (j,s) < FixAct; 

fixActivity(act[j,s].startTime, (j,s) < FixAct); 

fixActivity(act[j,s], (j,s) < FixAct); 

// 過去のジョブ以外はステップ 10 以前に来てはならない 

Set NotFixAct(name="NotFixAct", dim=2); 

NotFixAct = setOf((j,s), (j < J, s < S)) - FixAct; 

sourceActivity < act[j,s], (j,s) < NotFixAct, 10; 

/* ------------------------------------ */ 

 

options.maxtim = 15; 

solve(); 

 

simple_printf("act[%s,%d] = %d¥n", j, s,  act[j,s].startTime); 

実行すると，各作業の開始時刻が以下のように出力されます． 

data.dat 

AvailMode = 

[a,1] mode_a_1 

[a,2] mode_a_2 

[a,3] mode_a_3 

[b,1] mode_b_1 

[b,2] mode_b_2 

[b,3] mode_b_3 

[c,1] mode_c_1 

[c,2] mode_c_2 

[c,3] mode_c_3 

; 

 

Prec = 

a 1 3 

a 3 2 

b 1 2 

b 2 3 

c 2 1 

c 1 3 

; 

 

FixAct = 

a 1 

a 3 

b 1 

c 2 

; 

 

fixTime = [a,1] 0 [a,3] 5 [b,1] 5 [c,2] 0; 

fixMode = [a,1] mode_a_1 [a,3] mode_a_3 [b,1] mode_b_1 [c,2] mode_c_2; 

2.25.3節同様，結果のガントチャート出力を以下のように行います． 

 

例題（リスケジューリング問題）の解の出力例： 

 

 

act["a",1] = 0 

act["a",2] = 19 

act["a",3] = 5 

act["b",1] = 5 

act["b",2] = 15 

act["b",3] = 21 

act["c",1] = 10 

act["c",2] = 0 

act["c",3] = 16 

機械 2故障中 

スケジューリ

ング固定 リスケジューリング 

2.26 遷移確率行列 

ここでは，遷移確率行列を用いた問題を紹介します．NUOPT V13 から導入された

Vector/Matrixクラスを利用すると，遷移確率行列を用いた問題が記述し易くなります． 

 

 

 

この問題を NUOPTで解くために定式化を行います． 

変数は，小学校 2から小学校 1への調整人数 x と小学校 2から小学校 3への調整人数 y とし

ます．目的関数は調整人数の最小化ですので x y となります．また，現実問題を考慮すると変

数は整数値をとるが，ここでは連続変数として扱い，得られた解の値を整数値に丸めて取り扱う

ことにします． 

次に制約条件を考えます．まず，調整人数についての制約として，非負制約，および調整人数

の総和が小学校 2 の入学予定人数を超えてはならないという制約があります．また，各遷移確

（例題） 

ある地域には小学校が 3校，中学校が 2校，高等学校が 2校ある．それぞれの学校に

は入学者の人数制限が決められており，どの学校へ進学するかの遷移確率がわかってい

るとする． 

いま，ある年度の小学校入学予定者人数を予測すると，小学校 2の入学者上限を大幅

に超えてしまったため，小学校 2から小学校 1，小学校 3への入学者人数の調整を行う

ことにした．この年度の児童が与えられた遷移確率に従って中学校，高等学校に進学す

ると仮定したとき，入学者上限制約を満たし，かつ最小の人数調整で解決したい場合，

どのような人数調整を行えばよいか． 

率に従って進学をした場合にそれぞれの学校の入学人数制限を満たすという制約があります． 

 以上をふまえて次のように定式化されます． 

   

集合 { 1, 2, 3}E E E E  小学校の集合 

 { 1, 2}J J J  中学校の集合 

 { 1, 2}H H H  高等学校の集合 

   

定数 , ,ejEJ e E j J   
各小学校から各中学校への遷移確

率行列 

 , ,jhJH j J h H   
各中学校から各高等学校への遷移

確率行列 

 ,eflow e E  
各小学校への入学予定人数ベクト

ル 

 ,eEup e E  各小学校の入学者上限ベクトル 

 ,jJup j J  各中学校の入学者上限ベクトル 

 ,hHup h H  各高等学校の入学者上限ベクトル 

   

変数 x  小学校 2から小学校 1への調整人数 

 y  小学校 2から小学校 3への調整人数 

 ,eX e E  
人数調整後の各小学校への入学人

数ベクトル 

 

1 1

2 2

3 3

E E

E E

E E

X flow x

X flow x y

X flow y

 

  

   

 

   

目的関数 (最小化) x y  調整人数の最小化 

   

制約 , 0x y   非負制約 

 
2Ex y flow   

調整人数の総和は小学校 2への入学

予定人数を超えてはならない 

 X Eup  各小学校の入学人数の制限 

 ( )T TX EJ Jup   各中学校の入学人数の制限 

 ( )T TX EJ JH Hup    各高等学校の入学人数の制限 

 



 197 

次に，ベクトルと行列をデータファイルから与える SIMPLEモデルを示します． 

 

 

 

TPM.smp 

// 集合と添え字 

Set E; Element e(set=E); // 小学校 

Set J; Element j(set=J); // 中学校 

Set H; Element h(set=H); // 高等学校 

 

// 行列パラメータとベクトルパラメータ 

Matrix EJ((e,j)), JH((j,h)); 

Vector flow(e), Eup(e), Jup(j), Hup(h); 

 

// 変数 

Variable  x,y; 

0 <= x; 

0 <= y; 

x + y <= flow["E2"]; 

 

// 変数ベクトル 

Vector X(e); 

X["E1"] = flow["E1"] + x; 

X["E2"] = flow["E2"] - x - y; 

X["E3"] = flow["E3"] + y; 

 

// 制約 

X <= Eup; 

trans(trans(X) * EJ) <= Jup; 

trans(trans(X) * EJ * JH) <= Hup; 

 

// 目的関数値 

Objective obj(type=minimize); 

obj = x + y; 

// 求解 

solve(); 

 

// 出力 

x.val.print(); 

y.val.print(); 

データファイル（.csv形式）は以下のようになります． 

各小学校への入学人数と，各学校への進学の遷移確率行列 

     

各学校の入学上限 

     

このモデルを実行すると， 15.2907x  ， 24.7093y  が求まりますので，整数値に丸めて

小学校 2から小学校 1への調整人数 15人と小学校 2から小学校 3への調整人数 25人が得られ

ます．また，その結果，以下のように進学予定人数がわかります．  

 

Hupper.csv 

h,Hup 

H1,360 

H2,360 

 

Jupper.csv 

j,Jup 

J1,360 

J2,360 

 

Eupper.csv 

e,Eup 

E1,240 

E2,240 

E3,240 

 

JH.csv 

JH,H1,H2 

J1,0.6,0.4 

J2,0.45,0.55 

 

EJ.csv 

EJ,J1,J2 

E1,0.8,0.2 

E2,0.4,0.6 

E3,0.3,0.7 

 

flow.csv 

e,flow 

E1,200 

E2,280 

E3,200 

2.27 ムーア・ペンローズ一般逆行列 

 NUOPT V13 から新たに導入された Vector/Matrix クラスの使い方に慣れるため，最適

化問題ではありませんが，ムーア・ペンローズ一般逆行列を求めてみましょう．この問題は[8]

を参考文献としています． 

 

A  をm n  行列とする線型方程式 A x y  が解 x を持つような y  に対して， A 
x = y

がこの方程式の一つの解となる場合，n m  行列 A 
 を A  の一般逆行列といいます． 

A  の一般逆行列は，次の式から求めることができます． 

AA A A   

一般的に A 
 は一意に定まりません． 

行列 M  の任意の一般逆行列を M 
 と表すとき， A  のムーア・ペンローズ一般逆行列 

A 
 は 

 T T T TA A A A AA A A


   

と書くことができ，これは一意に定まります．ここで，
TA  は A  の転置行列です． 

よって，
T TA AA A  に対する一般逆行列を一つ求めればよいことがわかります． 

以上より，次のように定式化できます． 

集合 M  行 

N  列 

   

定数 , ,m nA m M n N   定数行列成分 

   

変数 , ,ijX i N j N   変数行列成分 

   

目的関数  この問題に目的関数は無い 

   

制約条件 T T T T T TA AA A X A AA A A AA A  X は
T TA AA Aの一般逆行列 

   

（例題） 

次の行列 Aの，ムーア・ペンローズ一般逆行列 A 
を求めよ． 

1 1 1

1 1 1

1 1 1

1 1 1

A

 
 

 
 
 
   



 200 

定式化した結果を SIMPLE で記述すると以下のようになります． 

 

csv 形式の入力ファイルは以下のようになります， 

 

MPpsedoinverse.smp 

Set M, N; 

Element m(set = M), i(set = N), j(set = N); 

 

// 定数行列 

Matrix A((m,i)), B((i,j)); 

B = trans(A)*A*trans(A)*A; 

 

// 変数 

Variable x(index = (i,j)); 

Matrix X((i,j)); 

X[i,j] = x[i,j];// 行列 X に変数 x を詰め込む 

 

// 制約条件 

B*X*B == B; 

 

solve(); 

 

// ムーア・ペンローズ一般逆行列 

Matrix A_plus((i,m)); 

A_plus = trans(A)*A*X*trans(A); 

 

for(int k = 1; k <= N.card(); k++){ 

simple_printf(“%5.4f “, A_plus[k,m].val); 

simple_printf(“¥n”); 

} 

A.csv 

A,1,2,3 

1,1,-1,1 

2,1,1,-1 

3,-1,1,1 

4,1,1,1 

このモデルを NUOPT で実行すると，最後に 

 

という表示がされて，この例題の答えが 

1 1 1 1
4 4 4 4

1 1 1 1
4 4 4 4

1 1 1 1
4 4 4 4

A 

 
 
  
 
 
 

 

であることがわかります． 

0.2500 0.2500 -0.2500 0.2500  

-0.2500 0.2500 0.2500 0.2500  

0.2500 -0.2500 0.2500 0.2500 



 202 

 

参考文献 

[1] George B. Dantzig, Mukund N.Thapa, Linear Programming 1:Introduction, 

Springer, 1997 

[2] 福島雅夫, 数理計画入門, 朝倉書店, 1996 

[3] 刀根薫, 数理計画, 朝倉書店, 2007 

[4] H.P.ウイリアムス, 数理計画モデルの作成法, 産業図書, 1995 

[5] Fisher,R.A., The Use of Multiple Measurements in Taxonomic Problems. 

Annals of Eugenics, 7, 179-188, 1936 

[6] Frank J.Fabozzi, Petter N.Kolm, Dessislava A. Pachamanova, Sergio M. 

Focardi, Robust Portfolio Optimization and Management, John Wiley & Sons, 

Inc., 2007  

[7] Christoph Helmberg, Semidefinite Programming for Combinatorial 

Optimization, Konrad-Zuse-Zentrum fur Informationstechnik Berlin,January 

2000 

[8] 柳井春夫・竹内啓，射影行列 一般逆行列 特異値分解，東京大学出版会 

[9] 枇々木規雄・田辺隆人，ポートフォリオ最適化と数理計画法，朝倉書店 

 



 203 

 

索引 

０ 

0-1整数計画問題, 157 

Ａ 

Activity, 189 

Ｃ 

csv, 28, 40, 46, 49, 57, 64, 70, 75, 95, 106, 111, 

119, 120, 128, 140, 146, 198, 200 

Ｄ 

dat, 11, 17, 23, 28, 32, 33, 34, 40, 57, 64, 70, 75, 

84, 115, 125, 140, 164 

DEA, 25 

DMU, 26 

Ｅ 

Excel連係, 136, 141, 165, 167, 171 

Expression, 119 

Ｌ 

LP, 4, 5 

Ｍ 

max関数, 114 

MILP, 4 

MIP, 4, 5 

Ｎ 

NLP, 4, 5 

NUOPT, 6, 12, 18, 33, 41, 69, 76, 153, 165, 167, 

171, 195 

Ｐ 

p センター問題, 5, 65, 71 

p メディアン問題, 5, 65 

print, 69 

Ｑ 

QP, 4, 5 

Ｒ 

RCPSP, 4, 5 

Ｓ 

SDP, 4, 5 

SIMPLE, 4, 10, 32, 35, 39, 41, 45, 62, 74, 90, 

119 

solve, 41, 97 

sum, 41, 119 

Ｖ 

Vector/Matrix, 47, 108, 126, 142, 153, 195, 

199 



 204 

Ｗ 

wcsp, 41, 102, 114 

WCSP, 4, 5 

い 

イールドカーブ, 5, 132, 136 

イールドカーブ推定問題, 5 

一般逆行列, 199 

え 

SDP緩和問題, 153, 154 

お 

オープンショップ問題, 174 

か 

格付け, 5, 137, 138 

格付け推移行列, 5, 137, 138 

格付け推移行列推定問題, 5 

ガントチャート, 165, 169, 172, 179, 184, 188, 194 

こ 

混合線形整数計画問題, 4 

さ 

最小カット問題, 50 

最小二乗問題, 5, 117 

最小費用流問題, 5, 52, 58, 59 

最大カット問題, 153 

最大流問題, 5, 42, 52, 58 

サブツアー, 77 

し 

資源制約付きスケジューリング問題, 4, 161, 175 

収益率, 122, 147 

集合, 33, 36 

集合被覆問題, 5, 36 

終了条件, 41 

巡回セールスマン問題, 76 

上下限制約, 42 

ジョブショップスケジューリング問題, 5, 174 

ジョブショップ問題, 174 

す 

推移確率行列, 138 

数理計画問題, 4 

スポットレート, 132, 133, 136 

せ 

整数計画問題, 4, 35, 37, 41 

制約式, 7, 26, 80, 114, 123 

制約充足問題, 4, 102, 114 

制約条件, 6, 13, 18, 26, 29, 37, 39, 40, 41, 42, 59, 

65, 71, 77, 86, 195 

接続行列, 47 

設備計画問題, 5, 112 

セミナー割当問題, 5 

遷移確率行列, 195 

線形計画問題, 4, 7, 26, 80 

先行制約, 169, 181, 183, 185, 189 



 205 

そ 

相関行列, 5, 143 

相関行列取得問題, 5 

双対変数, 50 

双対問題, 50 

た 

多期間計画問題, 5, 18 

多品種流問題, 5, 58 

ち 

中間変数, 77 

つ 

ツアー, 77 

て 

定数, 10, 22, 39, 41, 45, 55, 94, 104, 117, 118, 

119 

凸二次計画問題, 4 

な 

ナップサック問題, 4, 5, 29, 35, 39 

に 

二次割当問題, 108 

は 

配合問題, 5, 6 

パラメータ推定, 127, 128 

半正定値, 4, 143, 144, 149 

半正定値計画問題, 4, 144, 149 

ひ 

非線形計画問題, 4, 133, 138 

ふ 

フローショップ問題, 174 

分散共分散行列, 147 

へ 

閉路, 77 

変数, 6, 12, 13, 18, 26, 29, 36, 37, 42, 59, 65, 71, 

76, 80, 98, 117 

ほ 

包絡分析法, 25 

ポートフォリオ最適化問題, 5 

ま 

マルコビッツモデル, 122, 147 

み 

ミニマックス問題, 71 



 206 

む 

ムーア・ペンローズ一般逆行列, 199 

も 

モード, 166, 168, 169, 171, 175, 176, 190 

目的関数, 6, 7, 13, 18, 24, 26, 29, 36, 42, 52, 59, 

66, 71, 76, 80, 117, 119, 123, 128, 133, 138, 

157, 195 

モデル, 10, 16 

モデルファイル, 35, 41 

ゆ 

輸送問題, 5, 12 

ら 

ラプラス行列, 153 

り 

リスク, 122, 137, 147 

リスケジューリング, 189 

隣接行列, 153 

ろ 

ロジスティック回帰, 127 

ロジスティック関数, 128 

ロバストポートフォリオ最適化問題, 5 

わ 

割当問題, 5, 85, 86, 98, 102 

 


	1. はじめに
	2. 各種例題の紹介およびSIMPLEでの記述例
	2.1 配合問題（SAMPLES\例題集1.prj　参照）
	2.2 輸送問題（SAMPLES\例題集1.prj　参照）
	2.3 多期間計画問題（SAMPLES\例題集1.prj　参照）
	2.4 包絡分析法（DEA）モデル（SAMPLES\例題集1.prj　参照）
	2.5 ナップサック問題（SAMPLES\例題集1.prj　参照）
	2.6 集合被覆問題(SAMPLES\例題集1.prj  参照)
	2.7 最大流問題（SAMPLES\例題集1.prj　参照）
	2.7.1 最小カット問題

	2.8 最小費用流問題(SAMPLES\例題集1.prj　参照)
	2.9 多品種流問題（SAMPLES\例題集1.prj　参照）
	2.10  ｐ メディアン問題（SAMPLES\例題集2.prj　参照）
	2.11  ｐ センター問題（SAMPLES\例題集2.prj　参照）
	2.12 巡回セールスマン問題（SAMPLES\例題集2.prj　参照）
	2.13 割当問題（SAMPLES\例題集2.prj　参照）
	2.13.1  割当問題とは
	2.13.2  基礎的なマス埋め割当問題
	2.13.3 仕事割当問題
	2.13.4  施設配置問題

	2.14 二次割当問題（SAMPLES\例題集2.prj　参照）
	2.15 設備計画問題（SAMPLES\例題集2.prj　参照）
	2.16 最小二乗問題（SAMPLES\例題集2.prj　参照）
	2.17 ポートフォリオ最適化問題（SAMPLES\例題集2.prj参照））
	2.18 ロジスティック回帰モデル（SAMPLES\例題集2.prj参照）
	2.19 イールドカーブ推定問題（SAMPLES\例題集3.prj　参照）
	2.20 格付け推移行列推定問題(SAMPLES\例題集3.prj　参照)
	2.21 相関行列取得問題(SAMPLES\例題集3.prj　参照)
	2.22 ロバストポートフォリオ最適化問題（SAMPLES\例題集3.prj　参照）
	2.23 隣接行列（最大カット問題）（SAMPLES\例題集3.prj　参照）
	2.24 セミナー割当問題
	2.25 ジョブショップスケジューリング問題（SAMPLES\例題集3.prj　参照）
	2.25.1 オープンショップ問題
	2.25.2 フローショップ問題
	2.25.3 ジョブショップ問題
	2.25.4 リスケジューリング問題

	2.26 遷移確率行列
	2.27 ムーア・ペンローズ一般逆行列

	参考文献
	索引

